

> **betrachte Quellcode**

Du kannst mit dem Quellcode wenig anfangen.

> **schlag Hilfe im Handbuch nach**

Du blätterst ein wenig darin und findest nachstehendes Dokument.

M A R I E L

Interactive Fiction schreiben mit Inform 7 und GerX

basierend auf Mariel Release 5 und GerX 3/110528, Mai 2011,
geschrieben von Michael Baltes

Inhaltsverzeichnis

Interactive Fiction – eine Einführung.....	3
Vorbereitungen.....	6
Die Installation von "Extensions".....	7
Ein deutsches Wörterbuch und ein Webseiten Template.....	7
Das "Framework" von Inform 7.....	8
Ein Konzept für ein interaktives Spiel.....	9
Die Welt.....	11
Das Schreiben mit Inform 7.....	12
Include Anweisung.....	14
Use Optionen.....	16
Die Besonderheiten von GerX.....	17
Ein einleitender Text.....	19
Attribute.....	20
Understand Definitionen.....	21
Klassen	23
Beziehungen	24
Orte und Regionen.....	25
Der Spieler.....	29
Dinge.....	29
Navigation.....	34
Prinzip der Konversation.....	38
NPC's.....	41
Themen und Konversation.....	42
Gameplay.....	47
Innenbereich.....	47
Außenbereich.....	67
Der Schluss.....	75
Das Testen und Ent-buggen.....	79
Replay und Skein.....	79
Debug-Befehle.....	81
Die Musterlösung.....	83
Das Veröffentlichen.....	83
Eine Webseite.....	84
IFDB, Ifwiki und Ifwizz.....	85
Nachwort:.....	87

1. Interactive Fiction – eine Einführung

Interactive Fiction sind Spiele, die der Spieler wie ein Buch liest, oder Geschichten, die der Leser wie ein Spiel spielt – Interaktion mit dem Medium ist dabei unabdingbar, um die Geschichte voranzutreiben. Interactive Fiction entfaltet, wie ein gutes Buch, seinen Reiz erst auf den zweiten Blick – nämlich, wenn der Spieler oder Leser allmählich mit der Handlung und ihren Konventionen vertraut wird.

Interactive Fiction (kurz: IF) hat seine Wurzeln in den siebziger Jahren des 20. Jahrhunderts. Damals, als Rechner noch graue, klobige Maschinen waren, entwickelte sich IF – abseits von simplen Reaktionsspielen - als erste Möglichkeit, eine interaktive Geschichte mithilfe des Computers zu erzählen.

Charakteristisch an IF ist dabei, dass der Text nicht in seitenlangen "ermüdenden" Passagen erzählt wird, sondern in kurzen und knappen Abschnitten, die den Leser immer wieder zur Interaktion auffordern. Nach und nach lernt der Leser mehr über seine Umwelt, über den Hintergrund der Geschichte und über die Rolle, die er in der Geschichte selbst spielt.

Geprägt war die erste Epoche in der IF Historie von einfachen Texten, unbeschriebenen Objekten ("You see nothing special about the table.") und unzähligen Räumen (oft über siebzig verschiedene), die auf komplexe Weise untereinander verknüpft waren.

Das Zeichnen einer Umgebungskarte war unumgänglich, gespickt mit Notizen zu den einzelnen Gegenständen und den Charakteren, auf die der Spieler traf, um den Überblick über das komplexe System zu behalten.

Eine kurze Passage aus "Zork" von Infocom, erschienen 1980.

ZORK I: The Great Underground Empire
Copyright (c) 1981, 1982, 1983 Infocom, Inc. All rights reserved.
ZORK is a registered trademark of Infocom, Inc.
Revision 88 / Serial number 840726

West of House
You are standing in an open field west of a white house, with a boarded front door.
There is a small mailbox here.

West of House
> open mailbox
Opening the small mailbox reveals a leaflet.

West of House
> get leaflet
Taken.

West of House
> read it
"WELCOME TO ZORK!"

ZORK is a game of adventure, danger, and low cunning. In it you will explore some of the most amazing territory ever seen by mortals. No computer should be without one!"

Moderne IF hat sich in jeglicher Hinsicht weiterentwickelt.

Viele "Trends" und "Tendenzen" aktueller Computerspiele finden auch entsprechende Anwendung im Bereich der Interactive Fiction, bzw. teilweise sogar ihren Ursprung.

So gilt es heute als "gutes" Spieldesign, wenige Orte zu verwenden, so dass der Spieler zu jeder Zeit in der Lage ist, diese "im Kopf" zu visualisieren – Kartenzeichnen entfällt und auch blinde oder sehbehinderte Spieler sind in der Lage, die "Topographie" der Geschichte ohne Hilfsmittel zu überblicken.

Leere Räume sind passé und unbeschriebene Objekte verpönt. Die Texte sind komplexer und anspruchsvoller geworden und wandeln sich im Lauf der Geschichte, sind unterschiedlich je nach Tageszeit oder in Abhängigkeit von bestimmten Ereignissen und suggerieren dem Leser, dass sich die Textwelt "verändert", machen die Umgebung "lebendiger".

Nicht Rätsel sind im Vordergrund, sondern die Umstände, wie der Spieler in seine jetzige Situation gekommen ist, was in der Vergangenheit passiert ist, Beziehungen des Protagonisten zu NPC's, eine bestimmte Haltung, die der Spieler einnehmen muss oder kann, kurzum: "interactive storytelling" mit all seinen Stärken, aber auch Schwächen.

Dies erfordert natürlich auch mehr Zeit, um eine Geschichte zu schreiben, bzw. zu vollständig zu implementieren. Es gilt, alle Räume mit Sorgfalt auszustatten, alle Objekte, die im Text erwähnt sind, zu implementieren und die Geschichte so auszuarbeiten, dass sie konsistent und logisch erscheint.

Stärken von IF anhand eines kleinen Auszugs aus "Glowgrass" von Nate Cull, aus dem Jahr 1997:

Der Protagonist ist ein "Xenohistoriker", der in der fernen Zukunft eine alte Siedlung der Menschen untersucht.

Geschicht wird hier mit der Sichtweise des Lesers gespielt: Er sieht eine rechteckige Vertiefung, die sich dann als "swimming pool" herausstellt.

Die Bedeutung ist dem Historiker aber nicht mehr bekannt – Ein Stilmittel der Erzählung, das nur in dieser reinen Textform funktioniert.

> e

Lawn

You're standing in a square area of neatly trimmed green lawn, shimmering faintly in the afternoon light. To the west, the lawn gives way to courtyard, and beyond it to the west and north are buildings. An odd, paved rectangular depression in the ground leads down to the southeast; the encroaching jungle vegetation further back is kept at bay by a barrier of translucent sheeting.

Something about the scene gives you a sudden tingle of recognition, a surge of awe. As if you're walking into a legend.

> x depression

The depression leads down to the southeast.

> se

Pool

You're standing in a concrete rectangle set into the ground, paved with a grey substance, probably stressed concrete. It's recognisable as a swimming pool, a popular Ancient recreation fixture. Whatever water was once here has long since drained away, probably through the outlet grate set centrally in the pool's floor.

You see a bright orange object here.

Moderne IF spielt nicht selten sogar nur in einem einzigen Raum. Manche Geschichten sind linear aufgebaut, andere sind sehr weit verzweigt und führen zu unterschiedlichen Enden, je nachdem, wie der Leser sich entscheidet. Einige sind klassischen Genres zuzuordnen, andere sind eher experimentell, oder philosophisch.

Interactive Fiction bewegt sich heutzutage zwischen Literatur und Spiel, zwischen Realität und Fiktion, zwischen geschriebenem Wort und der eigenen Phantasie.

Die Grenzen sind noch lange nicht erforscht.

Der Unterschied von "Zork" zu einem aktuellen IF Titel, allein hinsichtlich Textlänge und Wortwahl, wird deutlich anhand folgender Passage aus "Klub Karisma", geschrieben 2008 von Christian Blümke:

KLUB KARISMA

Ein interaktives Rein-Raus-Spiel von Christian Blümke

Release 8 / Serial number 110419 (v8.11) / Inform v6.31 Library 6/11

Kein Plan, was hier läuft? Dann sage INFO.

Keine Lust mehr auf das, was hier läuft? Dann sage ENDE.

Reeperbahn, Hans-Albers-Platz

Der kleine Platz liegt direkt an der Reeperbahn, die sich von Ost nach West quer durch St. Pauli zieht. In seiner Mitte steht eine bronzene Hans-Albers-Statue, umgeben von Kneipen, Tingeltangel-Bars und Theatern.

In einem auffälligen Gebäude an der Südseite des Platzes befindet sich der Klub Karisma. Genau daneben führt eine dunkle Hofeinfahrt in Richtung Südwesten um das Gebäude herum.

Vor dem Portal des Klubs wacht ein als Tarzan verkleideter Türsteher, der eine Frau nach der anderen durchwinkt.

Vor wenigen Minuten sind Sandra und Starstripper Guido im Klub Karisma verschwunden. Das ist der Laden, in dem der Mistkerl jeden Abend auftritt. Es ist zwar sonst nicht deine Art, Leuten einfach so heimlich hinterher zu spionieren, aber das hier ist ja nun wirklich mal ein Notfall.

> brüll tarzan an

Lautstärke bringt hier nichts.

> b tarzan

Johnny Weissmüller wäre sicher neidisch auf diesen anabolikagestählten Körper gewesen. Der Karisma-Tarzan trägt einen Lendenschurz. Und sonst nichts. Während er eine Frau nach der nächsten in den Klub lotst, stellt er ein Grinsen zur Schau, das er wohl für verführerisch hält.

> b klub

Faszinierend, wie der Architekt des Klubs es geschafft hat, aus dem dreistöckigen Jahrhundertwende-Bau einen psychedelischen Retina-Reizer zu machen. Überall auf der Fassade geben sich wirre, unzusammenhängende Muster in den allergrellsten Farben ein Stelldichein. Hektische Spektral-Blinklichter runden das Bild ab.

Der Eingang des Klubs wird aus zwei fünf Meter hohen orangefarbenen Palmen gebildet, deren Wedel sich in der Mitte treffen. Aus den Kronen der Bäume ragen keck einige goldene Bananen (das sind doch Bananen, oder?) in den Nachthimmel. Neben dem Eingang ist ein Schild angebracht.

Aus dem Klub hörst du Frauenstimmen rhythmisch im Chor grölen und applaudieren: "GUI--DO! GUI--DO! GUI--DO!"


> _

2. Vorbereitungen


Bevor es losgehen kann, müsst Ihr Euch Inform 7 für Eure Plattform (WIN/OSX/LINUX) von der Inform 7 Seite herunterladen und installieren.

Ihr findet die Downloads unter: www.inform7.com

Nach der Installation und dem Starten der Applikation sollte der Bildschirm in etwa so aussehen:



Ihr klickt auf "Start a new project":



Hier tragt Ihr ein Verzeichnis ein, worin die Geschichten, die Ihr schreibt, gespeichert werden sollen, dann wählt Ihr einen Namen für die Geschichte und tragt Euren Namen oder ein Pseudonym in der Zeile "The author's name" ein – fertig.

Der Screen zeigt dann links das Quelltextfenster und rechts das Storyfenster. Dies könnt ihr nach Belieben auch ändern, später mehr dazu. Im Quelltextfenster steht jetzt folgender Text:

"Mariel" by Michael Baltes


Soweit so gut, der erste Schritt ist getan. Noch versteht Inform 7 aber kein Deutsch. Deshalb geht es auf der nächsten Seite weiter. [->]

1.1 Die Installation von "Extensions"

Jetzt fehlt uns zuallererst noch GerX, die "German Extension" für Inform 7. Dazu ladet Ihr die Erweiterung (Erweiterungen haben immer die Endung .i7x) unter

<http://ifiction.pageturner.de/inform7/>

herunter und legt die entpackte Datei einfach auf dem Desktop ab. Dann legt ihr Inform 7 wieder in den Vordergrund und klickt auf File – "Install Extension":



Ihr wählt den Desktop und dann die Datei aus und den Rest erledigt Inform 7 für Euch. Inform legt eine Liste aller installierten Erweiterungen an, die jederzeit abrufbar ist.

Technische Kurzinfo:

Inform 7 legt diese Erweiterungen in einen Bibliotheksordner auf der Festplatte unter Library/Inform (OSX), unter eigene Dateien/Inform (WIN), oder unter /usr/share/gnome-inform7/Extensions (LINUX). Dort befindet sich auch das Webseiten Template und die "standard rules", die auf jeden Fall einen Blick wert sind.

Und bevor es richtig losgeht, installieren wir noch ...


1.2 Ein deutsches Wörterbuch und ein Webseiten Template

Das deutsche Wörterbuch ist aus Open Office entnommen und kann relativ leicht eingebunden werden. Ihr findet es zusammen mit dem Template unter

<http://www.michaelbaltes.com/schreiben.html>


Ladet Euch die Dateien "de_DE.aff", "de_DE.dic" und das Template herunter und speichert die Wörterbuch-Dateien im Inform Programm Ordner unter Programme/Inform 7/dictionaries, bzw. das Template in den Bibliotheksordner (s.o.).

Damit steht das deutsche Wörterbuch nach dem Neustart von Inform unter Edit / Spelling als Auswahl zur Verfügung. Mit dem Template beschäftigen wir uns später.



1.3 Das "Framework" von Inform 7

Sehen wir uns das Framework von Inform 7 mal näher an. Der Screenshot zeigt jetzt:



Auf der linken Seite steht unser Quelltext. Rechts ist die Dokumentation zu Inform 7 geöffnet, unterteilt in 26 Kapitel. Daneben gibt es Dokumentationen zu den installierten Extensions ("Installed Extensions") sowie das "Inform Recipe Book", eine Art Rezeptbuch für immer wiederkehrende Fragen zur Implementation.

Die Dokumentation ist eher technischer Natur und erklärt die Grundprinzipien, während das Rezeptbuch das Implementieren eher aus Sichtweise einer konkreten Problemstellung angeht.

Hinweis: Der Ansatz von Inform 7 beruht auf dem Prinzip des "literate programming", worüber Donald Knuth 1992 eine Abhandlung in Buchform publiziert hat, erschienen im Cambridge University Verlag.

Nachstehendes Zitat, zu finden unter www.literateprogramming.com beschreibt kurz den Inhalt:

**"I believe that the time is ripe for significantly better documentation of programs, and that we can best achieve this by considering programs to be works of literature. Hence, my title: "Literate Programming."
Let us change our traditional attitude to the construction of programs:
Instead of imagining that our main task is to instruct a computer what to do,
let us concentrate rather on explaining to human beings what we want a computer to do."**

Bevor wir mit dem Codeschreiben beginnen, wollen wir uns aber erst ausführlich mit dem Konzept der Geschichte auseinandersetzen. Zunächst wollen wir aber einen Blick auf die Konzeption von "Mariel" werfen, also beginnen wir quasi mit der ersten Idee. [->]

1.4 Ein Konzept für ein interaktives Spiel

Hinweis: Der veröffentlichte und vollständig kommentierte Quellcode von Mariel soll einen begleiteten Einstieg in das System bieten. Quelltexte oder Zitate aus dem Spiel sind immer **grün** markiert, technische Hinweise **blau**.


Noch ein Hinweis: wer Mariel noch nicht gespielt hat, sollte nicht weiterlesen, damit er sich den Spielspaß nicht verdirbt. (S P O I L E R!)

Ein letzter Hinweis: Für Release 5 von Mariel wurde das Framework der ersten Fassung erheblich erweitert. Wer Release 1 gespielt hat, wird festgestellt haben, dass gerade in der ersten Szene (im Krankenhaus) viele Standards bewusst überschrieben worden sind. Zudem kommt der Tod des Spielers zu unerwartet, die Nichtspielercharaktere sind schwach charakterisiert, bzw. deren Verhalten nicht ganz konsistent. Nach der Analyse der Stärken und Schwächen der Geschichte kam ich zu dem Entschluss, folgende Änderungen zu realisieren:

- 1) Erweitertes Setting "innerhalb" des Krankenhauses
- 2) Mehr Informationen zu dem Ort, die nach und nach aufgedeckt werden
- 3) Verdeutlichung, warum Mariel die Schlüsselperson in dem Spiel ist und
- 4) vage Andeutungen, warum sie so handelt
- 5) Eine überraschende Wende zum Schluss
- 6) Mehr "Fairness" im Gameplay

Jetzt aber, "from the scratch":

Zeichnungen, Diagramme, Notizen und ähnliche Hilfsmitteln skizzieren einen groben Abriss der Handlung und des Settings.



In meiner Vorstellung befinden wir uns in einem kleinen Wäldchen, irgendwo auf dem Land.

Eine alte Kathedrale ist der Hauptschauplatz, vor vielen Jahren umgenutzt zu einem Hospital, wo Diphteriekranke langsam auf ihren Tod warteten. Dort arbeitete eine Krankenschwester viele Jahre lang.

Eine Krankenschwester, die durch das Leid, das sie mit ansehen musste, langsam verrückt geworden ist. Eine Krankenschwester und ihr Helfer, die geblieben sind, obwohl alle anderen das Hospital längst verlassen haben.

Eine Krankenschwester, die mit alten Medikamenten experimentiert hat, die ein Serum entwickelt hat, das Tote wieder auferweckt und die den Menschen jetzt mit ihren Serum helfen will, und ihre Augen vor den Nebenwirkungen gänzlich verschließt, kurzum: M A R I E L.


Die Kathedrale ist von einem Außenbreich umgeben, der durch eine Einzäunung abgegrenzt ist. Der einzige Ausgang befindet sich auf der Südseite des Geländes.

Der Leser soll nach und nach entdecken, worum es in dieser Geschichte geht, und soll herausfinden, wie er dem Desaster entkommt.

Dazu findet er folgende Gegenstände, bzw. begegnet folgenden NPC's (Nichtspielercharakteren):


- einen Notizzettel
- einen Monitor
- ein Tagebuch
- einen weiteren "Patienten"
- eine Chronik der Kathedrale
- natürlich Mariel selbst
- den "Hausmeister"
- den "Polizisten"

Folgendes Diagramm fasst die inhaltlichen und die zeitlichen Aspekte der Handlung zusammen:



1.5 Die Welt

Die kleine interaktive Welt soll folgendermaßen aussehen:



Zunächst etwas verwirrend, zeigt diese Karte die Räume und die Beziehungen untereinander:

Grün eingefärbt ist der Innenbereich der Kathedrale, blau der Außenbereich. Grau sind Räume, die keiner von beiden Kategorien zugeordnet sind, und eher eine Platzhalterfunktion einnehmen.

Die Richtungen UP/DOWN sind mit kleinen roten Doppelpfeilen dargestellt, Türen sind als Balken zwischen den Räumen markiert, grüne Blitze kennzeichnen die Richtungen INSIDE/OUT.

Das soll erst mal als Übersicht reichen. Bis dahin ist es noch ein langer Weg. Wir beginnen auf der nächsten Seite mit unseren ersten Codezeilen.

Technischer Hinweis:

Inform 7 ist ein Vorsatz zu Inform 6. Inform 7 "übersetzt" die natürliche Sprache in Inform 6 code, ist dabei aber wesentlich weniger effizient, als reiner Inform 6 code. Das bedeutet, dass eine in Inform 7 geschriebene Geschichte als reiner Inform 6 code weniger Speicherplatz beanspruchen würde.

Dafür liegt der Vorteil von Inform 7 klar auf der Hand. Der code ist auch für außenstehende schnell verständlich, da er in natürlichem Englisch (mit deutschen Ausgabetexten) formuliert wird. Auch für den Autor selbst hat dies große Vorteile. Eine gute Gliederung vorausgesetzt, behält auch er leichter den Überblick und findet auch nach einer größeren Pause im Schreiben viel einfacher wieder den Einstieg.

2. Das Schreiben mit Inform 7

Wir schauen uns vor Beginn die unterschiedlichen Elemente, aus denen ein Inform 7 Quellcode bestehen kann, kurz an. Anfängern empfehle ich, die beiden Seiten erst mal kurz zu überfliegen und später noch einmal genauer nachzuschlagen, ein Fortgeschrittener sollte die nachfolgenden Elemente bereits kennen:

- **"Objects" and "Kinds"**

Objekte sind Abbildungen des Typ "object". Das beinhaltet Orte (location), Dinge (thing), Personen (person), etc. Kinds sind selbst kein Objekt, definieren aber allgemeingültige Eigenschaften, die den davon instanziierten Objekten vererbt werden.

Beispiele:

The hospital is a neuter room, in the indoors.

- **"Phrases"**

Phrasen sind Anweisungsblocks, die verschiedene Aufgaben haben können, z.B. die Zuweisung eines Wertes zu einer Variablen, oder die Änderung eines Zustands eines Objekts, oder eine Entscheidung zu treffen.

Beispiele:

Let the destination be the room-or-door north from location.

**To say begin infusing:
now the player is doomed.**

**if the player can see the nurse:
try answering the nurse that.**

- **"Rules"**

Regeln, die das Geschehen beeinflussen. Inform 7 wickelt das Ganze Geschehen im Hintergrund über solche Regeln ab. Regeln fassen Phrasen zusammen.

Beispiele:

**Report searching the desk:
say "Dein Blick fällt zunächst auf eine kleine Schublade an der Seite des Tisches."**

**Before going somewhere (this is the leave when going rule):
if the current interlocutor is not no-body:
try leaving the current interlocutor.**

- **"Descriptions"**

Deskriptoren beziehen sich auf Objekte. Die einfachste Art ist der Bezug auf ein einzelnes Objekt, etwas komplexer sind Gruppen von Objekten, die eine bestimmte Eigenschaft haben.

Beispiele:

**Instead of drinking the pills:
try eating the pills.**

**After opening a closed container (called the goal):
try searching the goal.**

Instead of taking something which is carried by an other person (called the owner):

say "[Er owner] ist sicher nicht gewillt, [den noun] herzugeben."

– **"Actions"**

Aktionen sind Interaktionen mit der Welt, die der Leser normalerweise durch das Eingeben vom Kommandos auslöst. Sie können aber genauso gut durch den Quellcode ausgelöst werden, z.B. um eine Aktion auf eine andere umzuleiten.

Beispiele:

Instead of exiting when player is in ventilation system:
try going down.

Instead of opening the screws when player is holder of the screwdriver:
try unscrewing the screws with the screwdriver.

– **"Understanding"**

sagt Inform 7, wie es Eingaben des Lesers verstehen soll.

Beispiele:

Understand "dreh [something] um" as turning.

Understand "Schacht [m]" as the ventilation system.

Understand "frag ueber/zu/nach [text]" as vaguely asking.

– **"Relations"**

Beziehungen halten die Zustände von Objekten zueinander fest.

Beispiel:

Knowing about relates various people to various things. The verb to know about (he knows about, they know about, it is known about) implies the knowing about relation.

– **"Tables"**

Tabellen fassen komplexe Inhalte wie Strings, Variablen und Objekte zusammen.

Beispiel:

Table of Gwyndors Ask-Answers

init item response

1 Journey "[one of]'Wie kommen wir nach Bevenstein, Gwyndor?', frage ich.[paragraph break]'Werft einen Blick auf die Karte', erwidert Gwyndor, 'ihr haltet euch grob in Richtung Südwesten. Der wichtigste Wegpunkt ist Gendors Kamm, die einzige Passage über das Gebrige Anth'aril. Danach folgt ihr den Wegen der Menschen.'[or]'Wegen dem Weg ...', sage ich.[paragraph break]'Es ist alles gesagt', erwidert Gwyndor. [or]Ich glaube, es hat wenig Sinn, weiter über den Weg zu reden.[stopping]"

Das soll erst mal ausreichen, auf die einzelnen Elemente wird während des Schreibens noch wesentlich ausführlicher eingegangen, so dass sich dann klar wird, wann welches Element im Quelltext korrekt eingesetzt wird. [->]

2.1 Include Anweisung

Nach langem Hin und Her folgt jetzt endlich unser erste Codezeile, die wir im linken Fenster hinzufügen:

Include German by Team GerX.

Damit binden wir die GerX-Erweiterung in das Projekt ein. Diese Zeile sollte immer *ganz* am Anfang stehen.

Inform7 erlaubt es, den Quelltext entsprechend zu untergliedern. Dafür gibt es in absteigender Reihenfolge die Gliederungsebenen


- "volume"
- "book"
- "part"
- "chapter"
- "section"

Nachdem Mariel eine eher kurze Geschichte wird, beschränken wir uns hier auf die Ebenen PART und SECTION und bauen eine Gliederung wie folgt auf, indem wir die nachfolgenden Zeilen im Quellcode eingeben:

Part – The prelude
Section – Release Options
Section – General settings
Section – What things (and persons) can be
Section – Better standards
Section – Better Exit Lister (for use with Exit Lister by Eric Eve)
Section – Fluids
Section – Custom Messages
Section – Room descriptors
Section – Directives
Section – Caught rooms
Section – Conversation handling
Section – Knowing relationship
Section – Faraway

Wenn Ihr auf den CONTENTS Button über dem linken Quelltextfenster klickt, sollte jetzt rechts abgebildete Grafik erscheinen, deren einzelne Bestandteile wir jetzt Schritt für Schritt aufbauen wollen.

Hinweis: diese Eingaben haben bislang nur eine "Gliederung" erstellt, sie bewirken noch keinerlei Definitionen oder Anweisungen an Inform 7, sondern dienen lediglich der Übersichtlichkeit des Quelltexts.



Die Section "Release Options" füllen wir mit folgenden Zeilen:

Release along with cover art.


Release along with the source text.

Release along with a "german" website.

Release along with the "quixe" interpreter.

Damit sagen wir Inform 7, dass wir im späteren Release (auf der Webseite) den Quellcode eingebunden haben wollen, und dass Inform 7 automatisch eine mit dem "Quixe" Interpreter spielbare Webseite generieren soll, die auf unserem zuvor eingefügten "german" template basiert. Die Vorlage schreibt die Begleittexte auf Deutsch (im Gegensatz zum englischen "Standardtemplate").

Wir klicken auf "Release" und erhalten folgende Fehlermeldung:



Zum einen meckert Inform 7, weil wir keinen einzigen Raum in der Geschichte implementiert haben und es somit nicht in der Lage ist, den Startort des Spielers zu bestimmen.

Zum anderen fehlt natürlich noch das cover: dazu benötigen wir noch zwei Dateien, namens "cover.png" und "small cover.png", die wir in den unten beschriebenen "materials" Ordner hineinkopieren müssen. Als Format geht auch .jpg, die Größe sollte 960 x 960 px beim großen und 120 x 120 px beim kleinen Cover betragen.

Wer Lust hat, kann ja mal mit einem eigenen Cover experimentieren.
Alle anderen finden die beiden Dateien hier zum Download:

<http://www.michaelbaltes.com/IF%20Mariel/index.html>

Dort einfach auf das kleine Bild rechtsklicken -> speichern unter ... und linksklicken, um das große Cover aufzurufen und dann ebenfalls rechtsklicken -> speichern unter ...

Der Zielort ist der "Materials Ordner", den Inform 7 mit dem Release-Klick automatisch neben dem Projektordner (in diesem Fall: mariel.materials) angelegt hat.

Weiter geht es mit dem nächsten Abschnitt.

2.2 Use Optionen

Drei weitere wichtige Zeilen, einzufügen unter "General settings":

Section – General settings

Use no scoring.

Use explicit error messages.

Use DICT_WORD_SIZE of 15.

"No scoring" schaltet alle Score-Funktionen aus, d.h. In der Statuszeile werden rechts nur noch die Züge angezeigt, der "score" Befehl und die Mitteilung der Punktzahl nach dem Ende der Geschichte deaktiviert.

"Explicit error messages" ist ein Zusatzoption aus GerX. Die Fehlermeldungen bei nicht verstandenen Kommandos werden geringfügig abgeändert. Hier könnt ihr mal ausprobieren, welche Unterschiede sich ergeben, oder welches System Ihr besser findet.

"DICT_WORD_SIZE" ist eine von mehreren in Inform 6 definierten Konstanten, die die Länge des Wörterbucheintrags bestimmen. Standardmäßig sind das in z-code 9 Zeichen, was aber im Deutschen schnell überschritten wird. So wäre z.B. Das Wort "Schluessel" (10 Zeichen) und das Wört "Schluesseloch" (14 Zeichen) für den Parser nicht mehr unterscheidbar. Da in Mariel ein Generalschlüssel auftauchen soll, ändern wir die Eintraggröße mit dieser use-Option auf 15 Zeichen.

Technische Kurzinfo:

Die Größe des Eintrags kann nur mit dem Glulx-Format geändert werden.

Daher ist es notwendig, die Einstellung von "z-code" auf "Glulx" zu ändern.

Die deutsche Extension beansprucht aufgrund der zahlreichen Zusatzfunktionen (so z.B. Das Abschneiden der Endungen bei Verben (vgl. "spring/springe/springen"), die Deklinationen der Substantive und deren Adjektive, etc. wesentlich mehr Speicher als die reine englische Grundform. Daher ist das Glulx-Format sowieso ab einer gewissen Größe der Geschichte unumgänglich.

Auch die nächsten Zeilen definieren noch allgemeingültige Dinge:

The story genre is "Survival Horror".

The story headline is "Ein interaktiver Horror".

The story description is "Ein Krankenhaus, in dem nichts so ist, wie es sein sollte. Nach einem schweren Unfall deckst du allmählich die Geschichte dieses Ortes auf, bis schließlich die Wahrheit ans Licht kommt."

The release number is 5.

Das "Genre", die "Headline" und die "Description" sind wichtig für den Eintrag in den Datenbanken. Die Release Number hat sich seit Beginn des Schreibens an Mariel auf 5 geändert.

Ein kleiner Hinweis:


Inform7 vergibt zu jedem neu angelegten Projekt eine eigenständige Identifikationsnummer (in diesem Fall //A0DA17C1-6D51-4B5B-809C-EABEA403BB12//), über die das Werk auch in der Interactive Fiction Database (<http://www.ifdb.tads.org/>) identifiziert wird.

Daher ist es zu vermeiden, "alte" Projektdateien für ein neues Projekt zu kopieren, da die Nummer somit bestehen bleibt. Der korrekte Weg ist hier, ein neues Projekt anzulegen und dann den Quelltext des alten Projekts einfach in das neue hineinzukopieren.

2.3 Die Besonderheiten von GerX

Es empfiehlt sich in jedem Fall, die eingebaute Kurzdokumentation für GerX zu lesen. Klickt dafür im rechten Fenster auf "Installed Extensions" und dann auf Team GerX / German.

Screenshot:



Beim Schreiben auf Deutsch sind ein paar Besonderheiten zu beachten, die für alle nachfolgenden Codeabschnitte sinngemäß gelten.

Zuallererst gibt es zwei verschiedene Methoden der Definition: die "Denglisch" Variante und die "einsprachige" Variante.

Eine einfache Eingangshalle könnte in der englischen Grundversion so aussehen:

The Entrance is a room.

In der "Denglisch" Variante, auf Deutsch, so:

The Eingangsbereich is a room.

Und in der "einsprachigen" Variante so:

The entrance is a room.
The printed name is "Eingangsbereich[-s]".

In diesem Handbuch werde ich mich auf die "einsprachige" Variante beschränken. Diese "verursacht" zwar etwas mehr Text, ist aber in meinen Augen besser lesbar und erleichtert tatsächlich die Unterscheidung zwischen "Quelltext" und "Ausgabertext".

Eine weitere Besonderheit, die wir auch gleich zu Anfang benötigen, ist die Definition von korrekt

deklinierten Adjektiven. Dazu verwenden wir die Property "printed name", also eine Eigenschaft, die wir schon oben bei der Raumdefinition verwendet haben, und auch für alle Instanzen des Typ "object" sinngemäß gelten.

Wann auch immer Inform ein Objekt beim "Namen" nennt, wird der "printed name" ausgegeben.

Bei Adjektiven wird einfach ein "^" in eckigen Klammern hinten angestellt und GerX ergänzt automatisch die richtige Endung, je nach Kasus.

The rubbish is neuter scenery in the hospital. The printed name of the rubbish is "nutzlos[^] Gerümpel[-s]".

Eine zweite Besonderheit: falls der Gegenstand mit einem -s als Genitiv-Endung dekliniert wird, muss dies gesondert in eckige Klammern hintenangestellt werden.

Normalerweise wird der verwendete Bezeichner bei der Definition eines Gegenstands automatisch im Wörthuch als "reference" und "printed name" abgelegt. Dies wollen wir in der einsprachigen Schreibweise natürlich verhindern, da die Eingabe des Lesers "untersuche rubbish" ziemlich sinnlos ist. Daher verwenden wir die Phrase:

A thing is usually privately-named. A room is usually privately-named.

Hinweis für Programmierer:

Es kann natürlich auch – nach Belieben – mit Binde- oder Unterstrichen gearbeitet werden. Dies vermeidet in manchen Fällen sogar Verwechslungen, insbesondere dann, wenn Teile des Objektnamens bereits von Inform 7 verwendete Descriptoren enthält, kann es in seltenen Fällen zu Verwechslungen kommen.

(Beispiel: "Everybody is happy is a scene." ist ungültig, "Everybody_is_happy is a scene." ist gültig.)

Soweit, so gut. Wir machen jetzt nach diesem kurzen Exkurs direkt im Quelltext weiter. Also im linken Fenster auf "contents" klicken und dann auf "General settings", dann sind wir gleich im korrekten Abschnitt:

The screenshot shows the Inform 7 IDE interface. The main window is titled 'Mariette.S.inform - Inform' and has a menu bar with 'File', 'Edit', 'Format', 'Play', 'Replay', 'Release', 'Window', and 'Help'. Below the menu bar is a toolbar with icons for 'Go to', 'Run', 'Stop', and 'Help'. The main text area displays the 'Section - General settings' with the following content:

Use no scoring.
Use explicit error messages.
Use DICT_WORD_SIZE of 15.

The story genre is "Survival Horror".
The story headline is "Ein interaktiver Horror".
The story description is "Ein Krankenhaus, in dem nichts so ist, wie es sein sollte. Nach einem schweren Unfall deckst du allmählich die Geschichte dieses Ortes auf, bis schließlich die Wahrheit ans Licht kommt."

The release number is 4.

When play begins:
say "Ein leiser immer wiederkehrender Piepton ist zu hören - sonst nichts. Du schlägst langsam die Augen auf und schüttelst die Benommenheit so gut es geht ab. Was ist bloß passiert? Wie kommst du hierher?";[paragraph break]Das Einzige, woran du dich momentan erinnern kannst, ist ein blendendes Licht, ein Aufprall. Du hattest keine Chance mehr zu reagieren. Die Kurve war unübersichtlich und das entgegenkommende Auto hatte gerade an dieser Stelle zum Überholen angesetzt. Das Letzte, was du wahrgenommen hast, war das Knirschen von Metall und das Splittern der Windschutzscheibe.[paragraph break];
say "[bracket]Falls du mehr über diese Geschichte wissen möchtest oder Hilfestellung brauchst, gibt es die Befehle INFO, CREDITS oder HILFE.[close bracket]";[paragraph break]

A thing is usually privately-named. A room is usually privately-named.

The right sidebar is titled 'Extensions' and shows a list of extensions installed by 'Team GerX':

- Chapter 1: Willkommen
 - Section 1.1: Allgemeines
 - Section 1.2: Dankeschön
- Chapter 2: Die ersten Schritte
 - Section 2.1: Einbinden der German-Extension
 - Section 2.2: Bibliographische Daten
- Chapter 3: Vokabeln
 - Section 3.1: Was sind Vokabeln?
 - Section 3.2: Hauptwörter und Adjektive (Synonyme für Objekte)
 - Section 3.3: Verben und Satzmuster (Kommandos)
 - Section 3.4: Konflikte zwischen Vokabeln
- Chapter 4: Das Definieren von Dingen
 - Section 4.1: Zwei unterschiedliche Methoden
 - Section 4.2: Methode I: Die Denglisch-Variante (Standard)
 - Section 4.3: Methode II: Die einsprachige Variante
 - Section 4.4: Angabe des grammatischen Geschlechts (Genus)
 - Section 4.5: Angabe des Genus für Synonyme mit abweichendem Geschlecht (Changing Gender)
 - Section 4.6: Feststellen und Anzeigen des grammatischen Geschlechts (Genus)
 - Section 4.7: Sonderformen des unbestimmten Artikels: DEFINITE ARTICLE, YOURS, NO ARTICLE und PENDING
 - Section 4.8: Assemblies: Automatisch erzeugte Teile von Personen
- Chapter 5: Ausdrücke zur flexiblen Textausgabe
 - Section 5.1: Neue Textersetzungen müssen her
 - Section 5.2: Objektnamen
 - Section 5.3: Pronomen, Hilfsverben und Verb-Endungen
 - Section 5.4: Relativpronomen
 - Section 5.5: Wählbare Darstellung von "ß" und "ss"
 - Section 5.6: Exotisch: Wählbare Darstellung von "du"

2.4 Ein einleitender Text

Wir ergänzen folgende Zeilen im Abschnitt "General settings":

When play begins:

```
say "Ein leiser immer wiederkehrender Piepton ist zu hören - sonst  
nichts. Du schlägst langsam die Augen auf und schüttelst die  
Benommenheit so gut es geht ab. Was ist bloß passiert? Wie  
kommst du hierher?[paragraph break]Das Einzige, woran du  
dich momentan erinnern kannst, ist ein blendendes Licht, ein  
Aufprall. Du hattest keine Chance mehr zu reagieren. Die  
Kurve war unübersichtlich und das entgegenkommende Auto  
hatte gerade an dieser Stelle zum Überholen angesetzt. Das  
Letzte, was du wahrgenommen hast, war das Knirschen von  
Metall und das Splittern der Windschutzscheibe.[paragraph  
break]";  
say "[bracket]Falls du mehr über diese Geschichte wissen möchtest  
oder Hilfestellung brauchst, gibt es die Befehle INFO,  
CREDITS oder HILFE.[close bracket][paragraph break]".
```

A thing is usually privately-named. A room is usually privately-named.

"When play begins" ist wiederum eine Phrase, die immer am Anfang des "Abspielens" der Geschichte aufgerufen wird und kann auch zusätzliche Definitionen beinhalten. Wir verwenden dies "nur" für den Einleitungstext, der vor dem Beginn gedruckt werden soll. Ein "[paragraph break]" fügt einen Zeilenumbruch mit einer Leerezeile ein, [bracket] und [close bracket] die besagten eckigen Klammern, die für Textersetzungen vorbehalten sind, und daher nicht einfach so in den Augabetext geschrieben werden können.

Der Doppelpunkt markiert einen Anweisungsblock. Das Semikolon trennt einzelne Anweisungen innerhalb eines Anweisungsblocks und darf nicht vergessen werden. Auch die Einrückung mit Tabulatoren ist wesentlich. So stehen die beiden "say" Anweisungen untereinander in der gleichen Hierarchieebene, innerhalb des Ausführungsblocks "When play begins".

Korrekte Schreibweise einer verschachtelten Gruppe von Anweisungen:

Wenn die Szene Jagd beginnt:

mach dies;

mach das;

Falls der Spieler müde ist:

mach dies;

mach das;

Somit haben wir unsere erste "Rule" definiert, und wir könnten ihr, wenn wir wollten, auch einen Namen (in Klammern) geben:

When play begins (This is the print the introduction rule): ...usw

Damit könnten wir diese Regel auch mal deaktivieren, mit der Phrase:

The print the introduction rule is not listed in any rulebook.

Dies macht in der Regel aber nur in Erweiterungen Sinn, die es dem Autor ermöglichen sollten, die neuen Regeln dieser Erweiterung zu deaktivieren, ihre Reihenfolge zu ändern, andere Regeln vorher einzufügen, etc ... und sind hier und jetzt für uns nicht weiter von Bedeutung.

2.5 Attribute

Wir vergeben erst einmal ein paar "properties", also Attribute – sehr praktisch, denn über diese kann der Autor fast alles regeln, von einfachen Zuständen (open or closed z.B. sind standardmäßig bereits in Inform 7 definiert, etwa für Türen oder auch für Container, denen diese Eigenschaft ebenfalls zugewiesen werden kann) über die Auslösung von "Szenen" oder Ereignissen.

Section - What things (and persons) can be

Definition: a thing is other if it is not the player.

A thing can be moved.

A thing can be blatant.

A person can be asked.

A person can be doomed.

A person can be next-on-line.

A person can be nauseated.

A person can be manacled.

A person can be naïve.

Erwähnenswert: die Definition des Attributs "other", die an alle Dinge vergeben wird, die NICHT der Spieler selbst sind.

Kleines Detail am Rande: "Player" ist ein "object that varies", d.h. ein Platzhalter, der sich immer auf das Spielerobjekt bezieht. Das Spielerobjekt selbst ist "yourself". Daher kann der Autor dem "player" auch jederzeit eine andere Person zuweisen, um die Geschichte aus unterschiedlichen Perspektiven zu erzählen.

"Photopia" von Adam Cadre spielt sehr geschickt mit diesem Stilmittel.

Das soll uns jetzt nicht weiter beschäftigen, ebenso wie die folgenden Codeabschnitte, die noch als "Vorgeplänkel" zu werten sind und einige Grundlagen für den weiteren Verlauf schaffen.

Jetzt installieren wir noch folgende Erweiterungen:

- Exit Lister von Eric Eve (zu beziehen über die Inform Homepage)
- German Exit Lister von Team GerX (zu beziehen über die GerX Homepage)

und ergänzen dazu im Abschnitt der include Anweisungen folgende Zeilen:

Include Locksmith by Emily Short.

Include German by Team Gerx.

Include German Exit Lister by Team Gerx.

Include German Default Messages by Team Gerx.

Die zweite Zeile ist dabei schon vorhanden. Wichtig ist die Reihenfolge, da German bereits die Übersetzung von "Locksmith" beinhaltet und somit die Meldungen mit dem deutschen Text überschreibt.

Das "Wörterbuch" speichert alle Wörter, die Inform 7 versteht ab. In der "Grundform" der Extension sind dies über 700 verschiedene Formulierungen, die 66 unterschiedlichen Aktionen zugeordnet sind.

Anmerkung: hier ist Deutsch natürlich wesentlich umfangreicher als Englisch. So sind z.B. allein beim Verb "schlag" die unterschiedlichsten Aktionen zu berücksichtigen. (etwas nachschlagen = "consulting", etwas aufschlagen = "opening", jemanden schlagen = "attacking", etc ...)

Jetzt geht es ans Eingemachte: wir erweitern das "Wörterbuch" von Inform 7, also die

2.6 Understand Definitionen

In diesem Abschnitt werden gezielt Verbesserungen des Parsers für Aktionen, die in "Mariel" zur Verwendung kommen, vorgenommen.

Dabei wird der "Wortschatz" erweitert und das Verständnis der Spielereingabe verbessert.

Hinweis: Inform arbeitet mit "Satzschablonen", d.h. es prüft die Eingabe des Lesers auf vorgegebene Schemata – stimmt eines davon mit der Eingabe überein, wird dieses Schema "geparst". Einfache Befehle bestehen nur aus einem Verb ("spring"), komplexere beinhalten Verb und Nomen ("nimm Lampe"), oder gar Verb und zwei Nomen ("öffne Truhe mit Brechstange").

Wir ergänzen:

Section - Better standards

Understand "dreh [something] um" as turning.

Understand "kletter durch [something]" as climbing.

Understand "kletter durch [something] [hindurch]" as climbing.

Understand "brich [something] auf" as attacking.

Understand "brech [something] auf" as attacking.

Understand "brich [something] mit [something] auf" as attacking it with.

Understand "brech [something] mit [something] auf" as attacking it with.

Understand "brich [something] mit [something]" as attacking it with.

Understand "brech [something] mit [something]" as attacking it with.

Understand "stoss [something]" as pushing.

Understand "schubs [something]" as pushing.

Die Definitionen sind soweit selbsterklärend, [something] steht für ein Objekt, auf das sich der Leser bezieht. Das Objekt wird dann in dem Platzhalter "noun" abgelegt (bei zwei Objekten dann "second noun") und bis nach der Ausführung der Aktion festgehalten. Danach wird "noun" und "second noun" wieder gelöscht. Der Teil nach "as" bezeichnet eine Inform 7 bekannte Aktion.

[hindurch] ist ein "Understand Token", das durch GerX definiert wird. Hier wird "durch" oder "hindurch" als das [hindurch] Token verstanden. Das ist im Deutschen ziemlich hilfreich und erspart eigene Wörterbucheinträge für alle Fälle, wo "durch" gebraucht wird, also auch z.B. bei "geh durch Tür hindurch", was standardmäßig im Wortschatz enthalten ist.

Technischer Hinweis: Von diesen "Understand Token" gibt es einige vordefinierte, die ggf. verwendet werden können:

The Understand token hinein translates into I6 as "PREP_HINEIN_TOKEN".

The Understand token heraus translates into I6 as "PREP_HERAUS_TOKEN".

The Understand token weg translates into I6 as "PREP_WEG_TOKEN".

The Understand token ab translates into I6 as "PREP_WEG_TOKEN".

The Understand token darauf translates into I6 as "PREP_DARAUF_TOKEN".

The Understand token hinauf translates into I6 as "PREP_DARAUF_TOKEN".

The Understand token nach translates into I6 as "PREP_NACH_TOKEN".

The Understand token hindurch translates into I6 as "PREP_HINDURCH_TOKEN".

The Understand token hinüber translates into I6 as

"PREP_HINUEBER_TOKEN".

The Understand token hinueber translates into I6 as "PREP_HINUEBER_TOKEN".
The Understand token herunter translates into I6 as "PREP_HERUNTER_TOKEN".
The Understand token darunter translates into I6 as "PREP_DARUNTER_TOKEN".
The Understand token daran translates into I6 as "PREP_DARAN_TOKEN".
The Understand token dagegen translates into I6 as "PREP_DAGEGEN_TOKEN".

Sinnvoll ist hier ein Blick auf die bereits in GerX definierten Aktionen, die das Verwenden der Token anschaulich demonstrieren. Klickt dazu im rechten Fenster auf **Index / Actions** und scrollt runter bis zur Auflisten aller Kommandos nach dem Alphabet unter **"Commands available to the player"**.

Weiter geht es im Quelltext:

Instead of doing something other than examining to something which is part of a person:
say "Dazu bist du nicht in der Lage."

Ein weiterer Anweisungsblock, also eine neue "rule": (Wir erinnern uns: theoretisch könnten wir der Rule einen Namen geben, z.B. (This is the you can't do anything useful to a body part rule)
Eine "Instead" Anweisung fängt eine Aktion ab, bevor sie überhaupt begonnen hat. Hier wollen wir sinnlose Antworten bei Aktionen mit Körperteilen anderer Personen abfangen.

Eine Aktion läuft folgendermaßen ab:

Der Leser gibt das passende Kommando ein. Das Kommando wird "geparst", also die Eingabe in "Informesisch" übersetzt. Diese Routine heißt intern LTI (Language To Informese) und macht so nützliche Dinge wie Endungen abschneiden oder auch Umlaute durch 'oe/ae/ue' und 'ß' durch 'ss' zu ersetzen.

Daher ist es sehr wichtig, bei allen Understand Definitionen nur 'ss' und 'oe/ae/ue' zu verwenden, sonst wird das Wörterbuch nicht mehr ordnungsmäßig funktionieren und Fehler beim Parsen entstehen.

Jetzt aber zu Ablauf einer Aktion, als Beispiel dient die Aktion "Taking".

Inform setzt den "current action" Platzhalter auf die Aktion "Taking".
1. BEFORE:
Inform prüft auf "Before" rules: standardmäßig wird hier nichts ausgeführt.
2. INSTEAD:
Inform prüft auf "Instead" rules: standardmäßig wird hier nichts ausgeführt.
3. CHECK:
Inform führt die "Check" rule zu der Aktion aus. (Hier wird z.B. geprüft, ob der Gegenstand, den der Spieler nehmen will auch das Attribut "portable" hat, oder nicht). Wenn die "Check" rule erfolgreich passiert wird, gilt die Aktion als erfolgreich.
4. CARRY OUT:
Inform führt die "Carry-out" rule zu der Aktion aus. (Der Gegenstand wird dem Spieler ins Inventar gelegt)
5. AFTER:
Inform prüft auf "After" rules: standardmäßig wird hier nichts ausgeführt.
6. REPORT:
Inform führt die "Report" rule aus, d.h. In diesem Fall erhält der Spieler die Meldung "In Ordnung."

In obigem Schema ist ersichtlich, dass der Autor jederzeit mit einem "instead" den kompletten Regelablauf unterbrechen kann, bevor dieser überhaupt in Gang gesetzt wurde. Dasselbe gilt für "before", das sich als allererste Möglichkeit der Intervention einklinkt.

Die Aktion ist beendet, es wird wieder eine neue Eingabe des Spielers erwartet. Weiter im Quelltext:

Instead of taking something which is carried by an other person (called the owner):

say "[Er owner] ist sicher nicht gewillt, [den noun] herzugeben."

"Something which is carried ..." ist ein Descriptor, der für alle Objekte greift, die diese Eigenschaft besitzen, also von einer "anderen" (wir erinnern uns: other greift für alle Personen außer dem Spieler selbst) Person getragen werden. Simple Intention: die (unelegante) Standardantwort zu überschreiben. Praktisch: Die Textersetzung [Er owner] gibt nicht den "printed name" der betreffenden Person aus, sondern lediglich "Er/Sie/Es"

**After opening a closed container (called the goal):
try searching the goal.**

Diese Regel vereinfacht die Handhabung von geschlossenen Containern (also z.B. einem Schrank). Wenn der Spieler den Schrank öffnet, dann sicherlich mit dem Hintergedanken, hinein zu schauen. Inform 7 ist hier aber nicht so elegant. Es erwartet vom Spieler mit einem Befehl, den Schrank zu öffnen und mit dem nächsten hinein zu schauen. Stattdessen führen wir nach der Aktion "opening" (beachte: nur bei Erfolg) direkt die Aktion "searching" aus und ersparen dem Spieler Tipparbeit.

Hier übergeben wir den referenzierten Container in die temporäre Variable "goal". Das wäre zwar nicht unbedingt notwendig ("noun" würde auch gehen), ist aber guter Programmier"stil" und hilft bei komplexen Aktionen, Verwechslungen zu vermeiden.

Hinweis: "try" startet den kompletten Ablauf der genannten Aktion, so als ob es der Spieler direkt eingetippt hätte.

Folgende Regel verursacht einen Absturz der VM durch eine Endlosschleife:

**After opening a closed container (called the goal):
try opening the goal.**

Zwei letzte Regeln bleiben noch übrig, vor dem neuen Abschnitt:

**Before of eating something which is not edible:
say "Du kannst [den noun] nicht essen." instead.**

**Instead of attacking a door:
say "Du hämmerst wie wild auf [den noun] ein. Aber [er] [ist] zu stabil."**

Praktisch: das zuletzt genannte Objekt ist abgelegt in "noun". [er] übernimmt das zuletzt genannte Objekt und reicht es quasi weiter an [ist] . [ist] gibt dann automatisch "sind" aus, falls [er] das Attribut plural-named hat. Und:

**Instead of going down when the player is on something:
try exiting.**

Diese Regel besagt, dass "runter" die Aktion verlassen meint, wenn der Spieler z.B. auf dem Bett ist.

2.7 Klassen

"Mariel" soll zwei neue "kinds" von "thing" verwenden: Fluids und screwables. Der code dazu:

Section - Fluids

A fluid is a kind of thing.

**Instead of taking a fluid (called the item):
say "[Der item] rinn[t] durch deine Finger."**

Section - Screwables

A screwable is a kind of thing. A screwable is either tight or loose. A screwable is usually tight.

Fluid soll für alle Flüssigkeiten im Spiel gelten. Eine Flüssigkeit kann man natürlich nicht mitnehmen, also wird die Aktion "taking" dafür mit einem instead abgefangen.

Screwables deutet schon darauf hin: es wird was "zu schrauben" geben. Der Code hier ist kurz und weist lediglich das entweder-oder-Attribut "tight" oder "loose" zu.

Diese Art Attribute sind ebenfalls praktisch: man deklariert ein screwable als loose und das Attribut tight wird automatisch gelöscht – umgekehrt ebenso.

Die Phrase "is usually" weist diese Art Attribut allen Objekten dieses Kinds zu, ohne dass der Autor dies überall manuell eintragen muss.

Die Abschnitte "custom messages", "Room descriptors", "Directives", "Caught rooms" und "conversation handling" werden ebenfalls zurückgestellt.

Wir definieren abschließen noch die:

2.8 Beziehungen

Relations haben die Beziehung zwischen Objekten zum Ziel, also keine "eindimensionale" Definition von Attributen, sondern das Schaffen von Abhängigkeiten. Hier werden wir verfolgen, was der Spieler schon weiß, und was nicht.

Section - Knowing relationship

Knowing about relates various people to various things. The verb to know about (he knows about, they know about, it is known about) implies the knowing about relation.

Der Abschnitt ist sehr kurz. Wir sagen Inform das Verb, welches die Beziehung regelt. Dieses deklinieren wir kurz durch und schon versteht Inform Sätze wie:

Now the player knows about the nurse.

Vorausgesetzt natürlich, dass das Objekt "nurse" definiert wurde.

Ein letztes Attribut fehlt uns noch, um das man eigentlich in jeder Geschichte irgendwann braucht:

FARAWAY

A thing can be faraway.


Before doing something to a thing which is faraway:

if the current action is trying rubbing something or squeezing something or burning something or cutting something or tying something to something or swinging something or taking something or touching something or smelling something or pulling something or pushing something or turning something or opening something or closing something or unlocking something with something:

**say "[Der noun] [ist] zu weit entfernt.";
stop the action.**

Ein einfaches Beispiel: der Himmel, der in einem schönen schattigen Park zu sehen ist. Realistischerweise kann man diesen nicht berühren, schieben, schwingen, nehmen, etc. Daher werden alle diese Aktionen durch eine neue Regel mit "before" abgefangen. Ein "stop the action" muss die Regel abschliessen, sonst wird danach mit der Standardaktion weitergemacht. Jetzt haben wir die Vorarbeit erst mal abgeschlossen, wenden wir uns also dem "MAIN ACT" und den wirklich wichtigen Dingen zu.

Vorher ändern wir noch – wie am Anfang beschrieben – das Ausgabeformat auf GLULX.



Dies passiert im rechten Fenster im "Settings-Panel". Der Haken "bind up into a blorb ..." stellt sicher, dass Inform 7 das Cover an das Storyfile dranhängt. Jetzt aber los!

2.9 Orte und Regionen

Es ist Zeit für unseren ersten Ort: Ein Zimmer, in dem der Spieler aufwacht, ohne dass er weiß warum und wieso.

Wir definieren zunächst eine Region. Regionen fassen einen oder mehrere Orte zusammen.

Part - The main act

Chapter - World inside

The indoors is a region.

The wall is female plural-named backdrop in the indoors. The printed name of the wall is "massiv[^] Wände". Understand "massiv" and "Wand" and "Waende [p]" as the wall. The description of the wall is "Massive Wände umgeben dich."

Instead of attacking the wall:

say "Du hämmerst wie wild auf die Wand ein – nutzlos."

Dazu definieren wir ein sog. "Backdrop". Backdrops sind automatisch "scenery" objects (scenery meint, dass diese Objekte nicht extra in der Raumbeschreibung erwähnt werden) und sind in allen Räumen der Region vorhanden.

In Innenbereich also: die Wände.

Technischer Hinweis:

Backdrops sind Einzelobjekte, die nur einmal im Spiel vorkommen. Nach einem Wechsel der "location" des Spielers, wird dieser Gegenstand einfach zum neuen Ort bewegt, so dass er hier angesprochen werden kann, in diesem Fall sind die Wände also in jeden Raum verfügbar, der der Region "indoors" angehört.

Ein "instead" greift, falls der Spieler verzweifelt versucht, auf die Wände einzuschlagen.

Section - The first room

The hospital is a neuter room, in the indoors. "[if unvisited]Du hast keine Ahnung, wo du bist. Benommen schaust du dich um. Der Raum sieht aus wie eine düstere, unaufgeräumte Abstellkammer.[otherwise]Dieser Raum ist eng und bedrückend, verwahrlost und schmutzig. Gerümpel ist quer über den ganzen Raum verstreut. Die Möblierung besteht aus einem Bett, besser gesagt einer unbequemen, harten Liege und einem rostigen Metallschrank. [paragraph break]Eine Tür befindet sich im Süden und ein kleines vergittertes Fenster im Norden.[end if]".

The printed name of the hospital is "[if corridor is visited]Zimmer 126[else]Unbekanntes Zimmer[end if]".

Obige Phrase erschafft einen Raum, der Bezeichner dafür ist einfach "hospital". Der Raum befindet sich in der "indoors" Region, die "description" ist einfach in "" hintenangestellt. Das ist eine verkürzte Schreibweise und erspart das sonst übliche: The description of the hospital is ""...

Eckige Klammern können nicht nur Textersetzungen sein, sie können auch eine Phrase mit einer IF-Anweisung beinhalten. [if unvisited] ist wiederum eine Abkürzung von [if the hospital is unvisited] und wird mit einem [end if] abgeschlossen. Dazwischen ist noch [otherwise] plazierte. Stattdessen geht auch [else]. Das Resultat: Als "erste Raumbeschreibung", also wenn der Raum hospital noch das Attribut "unvisited" besitzt, wird nur ein kurzer Text ausgegeben, wenn der Spieler sich noch einmal bewusst mit dem Kommando "schau" umsieht, folgt eine ausführlichere Beschreibung.

Eine weitere Besonderheit sehen wir bei der Zuweisung des "printed name": Der Name variiert, je nachdem ob die Bedingung "corridor is visited" zutrifft oder nicht.

Denn der Spieler soll am Anfang seine "Zimmernummer" noch nicht wissen, erst wenn er den Flur betreten hat, sieht er die Nummer an der Tür, dann ändert sich automatisch der printed name von "Unbekanntes Zimmer" auf "Zimmer 126".

Jetzt ergänzen wir noch eine Zeile, pro forma, denn nur wenn ein Raum mit Namen "corridor" existiert, kann Inform 7 die vorstehende Phrase auch richtig zuordnen:

The corridor is a room.

Fertig. Wir sind soweit. Wir klicken auf "GO" und schauen uns das Ergebnis an:

Unbekanntes Zimmer
2

Ausgänge: Keine

Ein leiser immer wiederkehrender Piepton ist zu hören - sonst nichts. Du schlägst langsam die Augen auf und schüttelst die Benommenheit so gut es geht ab. Was ist bloß passiert? Wie kommst du hierher?

Das Einzige, woran du dich momentan erinnern kannst, ist ein blendendes Licht, ein Aufprall. Du hattest keine Chance mehr zu reagieren. Die Kurve war unübersichtlich und das entgegenkommende Auto hatte gerade an dieser Stelle zum Überholen angesetzt. Das Letzte, was du wahrgenommen hast, war das Knirschen von Metall und das Splittern der Windschutzscheibe.

[Falls du mehr über diese Geschichte wissen möchtest oder Hilfestellung brauchst, gibt es die Befehle INFO, CREDITS oder HILFE.]

Mariel
Ein interaktiver Horror von Michael Baltes
Release 4 / Serial number 110524 / Inform 7 build 6G60 (I6/v6.32 lib 6/12N) SD

Unbekanntes Zimmer
Du hast keine Ahnung, wo du bist. Benommen schaut du dich um. Der Raum sieht aus wie eine düstere, unaufgeräumte Abstellkammer.

>schau
Unbekanntes Zimmer
Dieser Raum ist eng und bedrückend, verwahrlost und schmutzig. Gerümpel ist quer über den ganzen Raum verstreut. Die Möblierung besteht aus einem Bett, besser gesagt einer unbequemen, harten Liege und einem rostigen Metallschrank.

Eine Tür befindet sich im Süden und ein kleines vergittertes Fenster im Norden.

>

Das sieht ganz ordentlich aus, unser Einleitungstext ist da, die vollständige Raumbeschreibung erscheint erst nach "schau" und wir können schon mal die Standard-Aktionen ausprobieren, wie singen, springen, nachdenken, winken, usw.

Natürlich fehlen jetzt noch die entsprechenden Objekte, es gibt noch kein Bett, keinen Schrank, kein Gerümpel, keine Tür und auch kein Fenster. Und keinen Ausgang, wie uns der "Exit Lister" in der Statuszeile korrekt mitteilt.


Werfen wir aber vorher noch einen Blick auf die Auswertung, die Inform 7 für uns bereits festgehalten hat: Nacheinander klicken wir die Reiter des rechten Fensters durch und erhalten wertvolle Informationen:

Das Errors-Panel:

Source
Errors
Index
Skein
Transcript
Game
Documentation
Settings

Progress
Problems

Translating the Source



This is the report produced by Inform 7 (build 6G60) on its most recent run through:

The 785-word source text has successfully been translated into a world with 2 rooms and 2 things, and the index has been brought up to date.

Das Index / Contents / Rules / World Index Panel:

Das sind die interessantesten Tools in Inform 7, da hier alle wesentlichen Dinge übersichtlich erfasst sind und jederzeit kontrolliert werden können:



Kinds Index

Table of all the kinds; how arithmetic affects them; and details of each kind in turn.
[What are kinds?](#); [More about kinds of object](#); [And kinds of value](#)

value	default value	repeat	props	under
object [17]	north	✗	✓	✗
room [2]	hospital	✓	✓	✓
thing [2]	yourself	✓	✓	✓
door	nothing	✓	✓	✓
container	nothing	✓	✓	✓
vehicle	nothing	✓	✓	✓
player's holdall	nothing	✓	✓	✓
supporter	nothing	✓	✓	✓
keychain	nothing	✓	✓	✓
backdrop [1]	wall	✓	✓	✓
person [1]	yourself	✓	✓	✓
man	nothing	✓	✓	✓
woman	nothing	✓	✓	✓
animal	nothing	✓	✓	✓
device	nothing	✓	✓	✓
passkey	nothing	✓	✓	✓
fluid	nothing	✓	✓	✓
screwable	nothing	✓	✓	✓
direction [12]	north	✓	✓	✓
region [1]	indoors	✓	✓	✓




Contents Index

Headings and extensions; named values, Tables and so on; the Library Card; the story file.
[How do headings work?](#); [What's the Library Card?](#)

Maribel by Michael Baltes

- Part - The prelude
 - Section - Release Options
 - Section - General settings
 - Section - What things (and persons) can be
 - Section - Better standards
 - Section - Fluids
 - fluid
 - Section - Screwables
 - screwable
 - Section - Knowing relationship
 - Section - Faraway
- Part - The main act
 - Chapter - World inside
 - indoors, wall
 - Section - The first room
 - hospital, corridor



Rules Index

Rules not directly tied to actions (see the Actions Index) or scenes (the Scenes Index).
[What are rulebooks?](#); [What are activities?](#); [Moving or abolishing rules](#)

Rules governing actions

These rules are the ones which tell Inform how actions work, and which affect how they happen in particular cases.

- persuasion ... rulebook (0 rules)
- unsuccessful attempt by ... rulebook (0 rules)
- before ... rulebook (10 rules)
- instead ... rulebook (7 rules)
- check ... rulebook (0 rules)
- carry out ... rulebook (0 rules)
- after ... rulebook (2 rules)
- report ... rulebook (0 rules)

Rules added to the sequence of play

These rulebooks are the best places to put rules timed to happen at the start, at the end, or once each turn. (Each is run through at a carefully chosen moment in the relevant top-level rulebook.) It is also possible to have rules take effect at specific times of day or when certain events happen. Those are listed in the Scenes index, alongside rules taking place when scenes begin or end.

- when play begins ... rulebook (2 rules)
- every turn ... rulebook (0 rules)
- when play ends ... rulebook (0 rules)



World Index

A map of the geographical layout; contents of each room; and an A to Z of rooms and things.
[What's the map?](#); [Using regions](#); [If the map looks wrong...](#); [Exporting to EPS](#)

Map



Hs indoors: hospital

Cr *Not in any region*: corridor

The indoors region

Hs **hospital** - room where play begins

- yourself - person

Cr **corridor**

Offstage (that is, initially not in any room):

- wall - backdrop

Weiter geht es in Quelltext mit einigen Definitionen zum Spieler selbst. [->]

2.9 Der Spieler

The player is nauseated. The player is male.

Messagecount is a number that varies. Messagecount is 1.

Every turn when the player is nauseated:

increase messagecount by 1;

if messagecount is 2:

say "[One of]Ein stechender Schmerz wütet in deinem Kopf.

[or]Alles um dich dreht sich und du kannst dich kaum

aufrecht halten.[or]Kalter Schweiß tropft von deiner

Stirn.[or]Dir ist schwindlig und speiübel.[or]Du musst

vor Schmerz kurz die Augen schließen.[or]Dein Kopf

pocht wie wild.[or]Du krümmst dich vor Schmerzen.

[cycling]";

now messagecount is 0.

The description of the player is "Dein Körper ist über und über mit kleinen Schnittwunden übersät, die vermutlich von dem Unfall stammen. Du fühlst dich wie gerädert."

Understand "schnittwunden" or "wunden" as yourself.

Am Anfang hat der Spieler Kopfhweh ohne Ende – er kann kaum klar denken. Dazu erhält er das Attribut "nauseated". Eine neue "Every turn" rule greift jetzt und lässt alle 3 Züge Text ausgeben. Die Ausgabe wird von einer Variablen gehandelt, die einfach von 0 bis 2 durchzählt. Die Say-Phrase benutzt [ONE OF] [OR] [CYCLING].

Das ist eine der Möglichkeiten der flexiblen Textausgabe. Jedes mal, wenn die Say Phrase aufgerufen wird, wird der Text um einen [OR]-Block weitergeschaltet. Das Ende [cycling] sagt Inform 7, dass es nun wieder am Anfang beginnen soll.

Der Spieler bekommt eine "description", so dass der Leser sich selbst mit "betrachte mich" ansehen kann. Auch "Wunden" oder "Schnittwunden" werden der Einfachheit halber als das Spielerobjekt verstanden. Damit sind wir auch schon fertig und fangen an, den Raum auszustatten.

Der Spieler ist am Anfang immer "neuter". Daher setzen wir korrekterweise das Attribut "male".

2.10 Dinge

The rubbish is neuter scenery in the hospital. The printed name of the rubbish is "nutzlos[^] Gerümpel[-s]". Understand "nutzlos" and "geruempel" and "sachen [p]" as the rubbish. The description of the rubbish is "Lauter komplett nutzlose Sachen, die kreuz und quer im Raum verteilt sind."

Das Gerümpel ist ein Neutrum, also bekommt es das Attribut "neuter". "Scenery" definiert, dass dieser Gegenstand nicht bewegt werden kann, und dass beim Aufruf der Raumbeschreibung nicht extra erwähnt wird, da dies ja bereits in der Beschreibung selbst passiert.

Das Synonym "Sachen" ist Plural, also schreiben wir ein "[p]" dahinter. Das ist wichtig, da der Spieler sich mit Angabe eines Artikels auf den zuletzt genannten Gegenstand beziehen kann:

>betrachte die sachen

Lauter komplett nutzlose Sachen, die kreuz und quer im Raum verteilt sind.

>nimm sie

Es macht wenig Sinn, sich mit nutzlosem Gerümpel zu beschäftigen.

Außerdem ist das Gerümpel kein Gegenstand, den man zählen kann, also soll der unbestimmte Artikel wegfallen:

The special indefinite article of the rubbish is NO ARTICLE.

Bei der Understand-Definition berücksichtigen wir das Adjektiv "nutzlos", um auch "betrachte die nutzlosen Sachen" zu verstehen.

Falls der Spieler auf die Idee kommt, sich ausgiebig mit dem Gerümpel zu beschäftigen, weisen wir ihn darauf hin, dass das zu nichts führt. Wir fangen alle Aktionen außer "betrachte" ab:

**Instead of doing anything other than examining to the rubbish:
say "Es macht wenig Sinn, sich mit nutzlosem Gerümpel zu beschäftigen."**

Next step - Die Decke, und ein Loch darin:

The ceiling is female scenery in the hospital. The printed name of the ceiling is "Kassettendecke". Understand "weiss" and "kassetten [p]" and "Decke" and "paneele [p]" and "deckenpaneele [p]" and "kassettendecke" as the ceiling. The description of the ceiling is "Eine Kassettendecke, vergilbt und staubig, mit einem Loch, das in einen Lüftungsschacht führt."

The hole is a male fixed in place. "In der Decke klafft ein Loch, das in einen kleinen Lüftungsschacht führt.". The printed name of the hole is "dunkl[^] Schacht[-s]". Understand "dunkel" and "Loch [n]" and "Schacht" as the hole. The description of the hole is "Der Schacht sieht uralt aus."

Definition: The hole is faraway if the player is not on the bed.

Definition: The ceiling is faraway if the player is not on the bed.

Die letzte Zeile verhindert das "Erreichen" des Lochs oder der Decke, wenn der Spieler nicht auf dem Bett steht. Dann soll die Decke und das Loch nämlich unerreichbar hoch sein.

**At 09:00 am:
move hole to hospital.**

Eine "At" rule, die ein "timed event" auslöst. Nämlich das Loch in den Raum zu befördern, wenn ein Zug vergangen ist. Das ist insofern wichtig, da das Loch nämlich nicht scenery, sondern "fixed in place" ist. Das bedeutet, dass das Loch sehr wohl extra erwähnt wird. Das ist ja auch unsere Intention, den Spieler gleich darauf hinzuweisen, dass hier etwas Besonderes ist. Aber eben noch nicht im ersten Zug, sondern erst danach. Aber – Ätsch zu früh gefreut, das Loch kann nur vom Bett aus erreicht werden:

**Instead of entering the hole when the player is not on the bed:
say "Von hier kannst du den Schacht nicht erreichen."**

Und zudem in diesem "nauseated" Zustand schon gar nicht:

**Instead of entering the hole when the player is nauseated:
say "In deinem Zustand kannst du unmöglich klettern."**

Dann schon, wir müssen "manuell" eingreifen, damit der Spieler in den Schacht gelangt.

**Instead of entering the hole when the player is on the bed:
say "Du hängst dich hoch, hängst einen Augenblick lang in der Luft und ziehst dich mit letzter Kraft in den Schacht hinein.";
move the player to the ventilation system.**

Auch im ersten Zug geht das noch nicht – und wir verraten noch nichts vom Loch!

**Instead of going up from hospital when hole is not in hospital:
say "Du bist zu benommen, um zu klettern."**

Ansonsten: versuch nach oben zu gehen:

**Instead of going up from hospital when hole is in hospital:
try entering the hole.**

Das wars! Ein einfaches Loch verursacht doch einiges an code, wie ihr seht. Tja, "Mariel" ist richtig fies, denn als nächstes kommt das Bett:

The bed is a neuter scenery enterable supporter in the hospital. The printed name of the bed is "alt[^] Bett[-s]". Understand "alt" and "Bett" and "Liege [f]" and "Sperrholz [n]" and "kante [f]" and "kanten [p]" as the bed. The description of the bed is "Das Bett ist ein dünne Liege aus Sperrholz, mit scharfen Kanten. [if crossways]Es nimmt fast den gesamten Raum zwischen Schrank und gegenüberliegender Wand ein, so dass die Schranktür unmöglich zu öffnen ist[otherwise]Es nimmt fast den gesamten Raum zwischen Tür und Fenster ein[end if].[if screws are tight] Es ist mit Schrauben fest am Boden verankert.[end if]".

The bed is either crossways or lengthways. The bed is crossways.

Das Bett ist der erste komplexere Gegenstand in "Mariel".

Das Bett ist ein "supporter", also eine Art Ablage, auf der etwas liegen kann. "Enterable" definiert, dass auch der Spieler diesen Gegenstand "betreten" kann. In der Understand-Definition verwenden wir abweichen vom Genus die Synonyme "Liege" und "Sperrholz" die korrekt ein "[f]" für feminin und "[n]" für neutrum hintenangestellt bekommen. Das ist das sog. "changing gender" also die Änderung des Genus in Abhängigkeit davon, welches Synonym der Spieler einsetzt.

Also funktioniert auch:

>b liege

Das Bett ist ein dünne Liege aus Sperrholz, mit scharfen Kanten. Es nimmt fast den gesamten Raum zwischen Schrank und gegenüberliegender Wand ein, so dass die Schranktür unmöglich zu öffnen ist. Es ist mit Schrauben fest am Boden verankert.

Dieses Bett weckt unangenehme Erinnerungen. Erinnerungen an lange Korridore mit vielen Kammern wie dieser, mit leblosen Gestalten auf diesen Betten.

Kalter Schweiß tropft von deiner Stirn.

>nimm sie

Dazu müsstest du zunächst von dem alten Bett herunter.

"Sie" bezieht sich hier korrekt auf die Liege, obwohl das Attribut des Betts "neuter" ist.

Das Bett hat zwei verschiedene Ausrichtungen, lengthways und crossways. Zudem ist es am Boden festgeschraubt. Die Beschreibungen variieren den Text je nach Zustand.

**Instead of looking under the bed when scribbled-note is in dummy room:
say "Unter dem Bett findest du einen Zettel."
move the scribbled-note to hospital;
now the initial appearance of the scribbled-note is "Unter dem Bett
liegt ein Zettel."**

Wir fangen das Schauen unters Bett ab und lassen den Spieler einen kleinen Zettel finden. Die Regel greift nur, wenn der Zettel sich in dem "Dummy Room" befindet, der noch implementiert werden muss. Die Property "initial appearance" legt den Text fest, der vor dem Aufnehmen des Gegenstands immer nach der Raumbeschreibung ausgegeben wird.

Nun zu den Schrauben:

The screws are plural-named scenery screwable in the hospital. Understand "Schrauben" or "Schraube [f]" as the screws. The printed name is "Schrauben". The description is "[if screws are loose]Die Schrauben sind locker.[otherwise]Die Schrauben sind fest angezogen und arretieren das Bett in dieser Position.[end if]"

Der Spieler befindet sich am Anfang auf dem Bett:

The player is on the bed.

Und das Bett kann gedreht werden, aber nur wenn vorher die Schrauben gelockert sind. Pushing und Pulling geben hier einen dezenten Hinweis, wenn der Spieler es ein zweites Mal versucht:

**Instead of pushing or pulling the bed:
if the screws are tight:
say "Das Bett ist mit Schrauben fest am Boden verankert."
instead;
say "[One of]Es ist so eng hier, dass das Bett nur wenige Zentimeter verschoben werden kann - es passt nahezu exakt in diesen fast quadratischen Raum hinein.[or]Es ist zwar sehr eng hier, aber vielleicht könntest du das Bett drehen, es scheint nicht allzu schwer zu sein.[stopping]"**

Aber wir wollen, dass der Spieler darauf kommen muss, das Bett zu drehen:

**Instead of turning the bed:
if the screws are tight:
say "Es ist mit Schrauben fest am Boden verankert." instead;
if the player is on the bed:
say "Das geht nicht, solange du auf dem Bett liegst.";
stop the action;
if the white door is open:
say "Die offene Tür verhindert jegliches Bewegen des Betts.";
stop the action;
if the cabinet is open:
say "Der offene Schrank verhindert jegliches Bewegen des Betts.";
stop the action;
say "Mit großer Mühe gelingt es dir, das Bett in eine Schräglage zu bringen, zu drehen und es wieder abzustellen. [if the bed is crossways]Es blockiert jetzt die Tür[otherwise]Es blockiert jetzt den Schrank[end if].";
now the bed is moved;
if the bed is lengthways:
now the bed is crossways;
otherwise:
now the bed is lengthways.**

Das Bett regelt zwei Öffnungsmechanismen, je nach Ausrichtung "lengthways" oder "crossways". Es blockiert damit entweder den Schrank oder die Zimmertür.

Dies verwenden wir später für das finale Rätsel zum Verlassen des Zimmers.

Jetzt kommt erst der Schrank:

The cabinet is a male enterable scenery closed and openable container in the hospital. The printed name of the cabinet is "Metallschrank[-s]". Understand "Schrank" or "Schrantuer [f]" or "Metallschrank" or "Beulen [p]" or "Rost [m]" or "Roststellen [p]" as the cabinet. The description of the cabinet is "Der Schrank sieht ziemlich heruntergekommen aus, überall Roststellen und Beulen. [if closed]Der Schrank ist im Moment geschlossen[otherwise]Der Schrank ist geöffnet[end if].".

Der Schrank ist ein "container", also ein Behälter, der andere Sachen umschließen kann, und wird mit "enterable" als betretbar definiert. In Abhängigkeit ob er offen oder geschlossen ist, variieren wir die Beschreibung etwas.

Und die Regel für das Abfangen des Öffnens:

**Instead of opening the cabinet when the bed is crossways:
say "Das alte Bett blockiert den Schrank. Er kann momentan nicht
geöffnet werden.".**

Als Ausgang aus dem Zimmer definieren wir zwei "Türen". Türen befinden sich immer zwischen zwei Räumen, sind offen (open), geschlossen (closed) oder gar verschlossen (locked).

The white door is a female scenery door, south of the hospital and north of the corridor. Understand "tuer" and "zu" and "zimmer" and "126" as the white door. The printed name is "Tür". The white door is closed. The description of the white door is "Eine weiße, stabile Tür, die sich nach innen öffnet.".

Und wieder die Regel für das Abfangen des Öffnens:

**Instead of opening the white door when the bed is lengthways:
say "Das alte Bett blockiert die Tür. Sie kann momentan nicht
geöffnet werden.".**

Die Tür braucht noch weitere Regeln, die das Öffnen und Schliessen handhaben:

**Instead of opening the white door when the player is on the bed:
say "Dazu musst du erst aufstehen.".**

Diese Regel soll bewirken, dass die Tür *nicht* automatisch geöffnet wird, wenn der Spieler (wie zu Beginn) auf dem Bett liegt.

**Instead of closing the white door when the nurse is in the hospital:
say "[The nurse] blockiert die Tür. Du kannst sie nicht schliessen.".**

Wenn die Krankenschwester in der Tür steht, kann sie ebenfalls nicht geschlossen werden. Gemein, gemein ...

**Instead of unlocking the white door with the crowbar when the white door is locked:
say "Du solltest eigentlich froh sein, dass sie im Moment
verschlossen ist.".**

**Instead of unlocking the white door with the crowbar when the white door is not locked:
say "Das ist unnötig, die Tür ist nicht verschlossen.".**

Die Brechstange kommt später ins Spiel.

Hinweis: Attribute müssen immer definiert werden, bevor sie einem Gegenstand im Quelltext zugewiesen werden können. Andernfalls führt das zu einer Fehlermeldung. Also immer "A thing can be blatant." vor "The backseat is a blatant thing ..." schreiben.

Und zuletzt das Fenster und die eben erwähnte Brechstange:

The window is a neuter scenery door, north of the hospital. The printed name of the window is "Fenster[-s]". Understand "klein" and "vergittert" and "Fenster" and "Gitter [n]" as the window. The window is closed. The description of the window is "Ein kleines, vergittertes Fenster in der Außenwand."

The crowbar is a female thing in the cabinet. The printed name of the crowbar is "schwer[^] Brechstange". Understand "schwer" and "Stange" and "Brechstange" and "Brecheisen" as the crowbar. The description of the crowbar is "Eine schwere Brechstange aus massivem Stahl."

Für das nächste Kapitel müssen wir etwas weiter ausholen und gehen zurück in den allgemeinen Teil.

2.11 Navigation

Die Navigation mittels Himmelsrichtungen ist ein "Hauptproblem" der modernen Interactive Fiction. Einerseits klingt es einfach nicht so elegant, im Text auf alle möglichen Himmelsrichtungen hinzuweisen. Gerade in Gebäuden ist das ziemlich seltsam. ("Im Norden befindet sich das Wohnzimmer").

Aber auf der anderen Seite ist die Navigation mit den Tastaturkürzeln N,S,O,W für erfahrene IF Spieler ein vertrauter Begleiter und sie erleichtern – mit etwas Übung und gutem Willen – die Vorstellung der Welt, die im Kopf des Lesers entsteht. Wenn jemand mit N einen Ort erreicht hat, ist es logisch, dass es mit S wieder zurück an den Ausgangsort geht. Kurz und gut.

Daher werden wir die konventionellen Richtungen beibehalten, den "Exit Lister" aber ein wenig modifizieren: Hier wollen wir statt der Richtungen nur die Kürzel verwenden und dahinter den Raumnamen schreiben, z.B.

N:(FLUR)

Daneben soll der Spieler die Möglichkeit haben, einfach

>geh in den Flur
>geh Flur
>Flur

eingzugeben, um eben dahin zu gelangen. Als dritte Besonderheit wollen wir den Befehl

>geh zum Schrank

den ein unerfahrener Spieler möglicherweise des öfteren probieren wird (eine ganz eigene Logik: ich geh zum Schrank, wenn ich ihn näher untersuchen will) abfangen und ihn darauf hinweisen, dass das nicht notwendig ist.

Passende Antwort:

"Du musst nicht zu dem Metallschrank hingehen, da er sich schon hier befindet."

Also, beginnen wir mit erst dem Exit Lister:

Section - Better exit lister (for use with Exit Lister by Eric Eve)

```
rule for printing the name of north while constructing the status line:
    say "n:";
    Let the destination be the room-or-door north from location;
    say "([destination]);"
```

Das war es schon. Eine Rule, die den Ausgabenamen der Richtung "Norden" ändert, während die Statuszeile aufgebaut wird, und dann das Ziel hintenanstellt. "destination" nimmt als lokale Variable das Objekt "Room-or-door" an, das entweder einen Raum oder eine Tür übergibt.

Das gleiche muss jetzt für alle Richtungen, einschl. Der Diagonalen (southwest, etc.), oben und unten und hinein hinaus erfolgen.

Aufgabe: vervollständigt den code selbst mit allen Richtungen und vergleicht Euer Ergebnis dann mit dem Quelltext.

Jetzt kommt das betreten eines naheliegenden Zimmers. Dazu muss jeder Raum eine Understand Definition erhalten. Wir beginnen mit dem Hospital. Im Zuge der Implementierung neuer Räume muss das entsprechend erweitert werden.

Section - Room-descriptors

Understand "Zimmer 126 [n]" or "Zimmer [n]" or "Kammer [f]" or "Abstellkammer [f]" as the hospital.

Wir achten auf die korrekte Zuordnung des Genus in Klammern. Jetzt kann sich der Spieler damit auf den aktuellen Aufenthaltsort beziehen.

Wenn der aktuelle Raum nur einen einzigen offensichtlichen Ausgang hat, wird ein Gehen in die entsprechende Richtung automatisch versucht – das übernimmt die Aktion "location-leaving", die in GerX definiert wird. Gibt es mehrere Richtungen, in die man gehen kann, fragt das Spiel nach und bittet um genauere Richtungsangaben.

Für jeden Raum sind die vier Synonyme "Ort [m]", "Platz [m]", "Raum [m]", "Standort [m]" und "Umgebung [f]" vordefiniert und werden durch das Understand-Token "[basic-location-synonyms]" repräsentiert. Durch unsere Understand-Definitionen wird dies erweitert.

Jetzt wenden wir uns dem Gehen in einen benachbarten Raum zu. Dazu gehört folgende neue Aktion "going by name":

Part - Actions

Section - Go to adjacent rooms

Going by name is an action applying to one thing.

Understand "geh zu [any room]" or "geh nach [any room]" or "geh in [any room]" or "geh [any room]" as going by name.

Check going by name:

```
if the noun is the location, say "Aber du bist schon hier." instead;
if the noun is not adjacent and the noun is unvisited, say "Dieses
    Hauptwort macht in diesem Zusammenhang keinen Sinn."
    instead;
```

```
if the noun is not adjacent and the noun is visited, say "[The noun] ist
    kein Ort, den du von hier erreichen kannst." instead.
```

Carry out going by name:

let aim be the best route from the location to the noun, using doors;
if aim is not a direction, say "Du hast keine Ahnung, wie du von hier
dorthin gelangen kannst." instead;
try going aim.

Die Understand Definition sollte mittlerweile selbsterklärend sein. Neu ist hier [any room], was soviel bedeutet wie "wende diese Satzschablone auf ALLE RÄUME an." Also gilt hier die Eingabe eines beliebigen Raums, ohne dass dieser in "Reichweite" des Spielers ist. Erinnerung: "something" greift nur für sichtbare Dinge innerhalb des aktuellen Standorts.

Würde das bei einer Standard Aktion wie berühren implementiert werden (understand "beruehr [any thing] as touching), könnte sich der Spieler mit diesem Kommando auf alle beliebigen in diesem Spiel vorkommenden Gegenstände beziehen. Probiert es ruhig aus, die check rules unterbinden diese Aktion aber trotzdem ("... gibt es hier nicht.")

In der Praxis wird so etwas z.B. bei dem Verb "sich erinnern" eine sinnvolle Anwendung finden, wenn der Spieler sich damit auf jedes mögliche "thing" beziehen können soll, ohne es unbedingt sehen zu müssen.

Das Attribut "adjacent" wird von Inform 7 bereits abgehandelt. Dieses Attribut bekommt jeder Ort, der unmittelbar an den Standort des Spielers angrenzt. Unmittelbar heisst, ohne dass eine Tür dazwischen ist. Und das ist auch der Grund, für nachstehenden Code:

Section - Directives

A directive is kind of thing. A directive is always scenery.
Instead of doing something other than entering or going near to a directive:
say "Betrete [den noun], um dorthin zu gelangen."

Wir erschaffen ein neues "kind", genannt "directive". Diese "directives" sind "fake"-Objekte, die eigentlich nur dazu dienen, dass der Spieler auch vom Zimmer aus in den Flur gehen kann, obwohl eine Tür dazwischen ist.

Alle Aktionen außer "going near" werden abgefangen, jetzt müssen wir noch "going near" definieren, eine neue Aktion, die , die ausschließlich bei "directives" getriggert werden soll.

Section - Going to near locations

Going near is an action applying to one thing.

Understand "geh zu [directive]" or "geh nach [directive]" or "geh in [directive]" or "geh [directive]" or "[directive]" as going near.

Ihr seht, das ist nicht viel code, erleichtert dem Spieler aber die "Orientierung" ungemein. Dazu fehlen aber noch die entsprechenden Objekte in den Räumen.

The foreseen-corridor is a male directive, here. Understand "Korridor" or "Flur" as the foreseen-corridor. The printed name is "Korridor".

Eine weitere Besonderheit: "here" meint den zuletzt definierten Raum. Also könnten wir für alle Ausstattungsgegenstände, die direkt nach dem Raum genannt sind auch "here" verwenden. Und zur letzten Aktion: wir unterbinden das gehen zu Dingen im Raum.

Section - Going to nearby things

Going nearby is an action applying to one thing.

Understand "geh zu [something]" as going nearby.

Check going nearby:

if the noun is the player, say "Du bist schon hier." instead;

if the noun is alive, say "Du musst nicht zu [dem noun] hingehen."
instead;

say "Du musst nicht zu [dem noun] hingehen, da [er] sich schon hier
befind[et]." instead

Drei einfache Check-rules runden das Ganze ab.

Damit Mariel kompiliert, erschaffen wir noch die zusätzlichen Dinge, auf die wir uns im Quelltext bezogen haben. Richtig ausarbeiten werden wir diese später.

The ventilation system is a room.

The nurse is a woman.

The dummy room is a room.

The scribbled-note is a male thing in the dummy room. Understand "zettel"
or "blatt [m]" or "papier [n]" as the scribbled-note. The printed name is
"Zettel[-s]". The description is "Mit einer dunklen Farbe hat jemand in
krakeliger Schrift geschrieben: FLIEH!"

Und schon ist alles soweit implementiert. Wir wagen einen "Restart". Schaut die Indizes durch. Ihr werdet sehen, dass sich die Zahl der Dinge und Aktionen schon etwas erhöht hat.

The screenshot shows the Inform 7 IDE interface. The main window displays the source code for a game titled "Mariel" by Michael Baltes. The code includes comments in German and is organized into sections: "Part - The prelude", "Section - Release Options", and "Section - General settings". The "Release Options" section contains several lines of code, some of which are commented out with square brackets. The "General settings" section includes settings for the story genre, headline, and description. A status bar at the bottom indicates "The game has been successfully compiled".

The right-hand window, titled "Translating the Source", shows a report generated by Inform 7 (build 6G60). The report states: "This is the report produced by Inform 7 (build 6G60) on its most recent run through: The 2565-word source text has successfully been translated into a world with 4 rooms and 16 things, and the index has been brought up to date." The report is accompanied by a small image of a textured surface.

Hinweis: die Release Optionen sind hier der Einfachheit halber deaktiviert.
Eckige Klammern im Quellcode umschließen immer einen Kommentar, d.h.
dieser Teil wird einfach wortwörtlich "ausgeklammert" und kann später durch
das Entfernen der Klammern wieder "aktiviert" werden.

2.11 Prinzip der Konversation

Eine weitere Besonderheit, mit der sich ein Autor beschäftigen muss, wenn er auch sogenannte "Nichtspielercharaktere" oder NPC's (also weitere Personen, mit denen der Spieler interagieren kann) einbauen möchte, ist das Konversationssystem. Konversation mit anderen Charakteren ist eine der kompliziertesten Aufgaben, die zu bewältigen sind.

Inform7 verwendet standardmäßig ein System, das auf sogenannten TOPICS oder einfach Textabschnitten basiert.

Daneben gibt es zahlreiche Erweiterungen für Inform7, die es ermöglichen, nach konkreten Gegenständen zu fragen, andere, die im Auge behalten, was der Spieler bereits gesehen hat und was nicht oder sogar spezielle Gesprächsthemen vorschlagen. Einige steuern Gespräche auch einfach über ein Menü.

In Mariel wollen wir auf dem Grundsystem aufbauen und Textabschnitte definieren, die den Gegenständen, die im Spiel vorkommen, entsprechen.

Section - Conversation handling

Instead of answering someone that "nein":
try saying no.

Instead of answering someone that "ja":
try saying yes.

Das war recht einfach. Statt nur "ja" zu tippen, kann der Spieler jetzt auch "sag ja" eingeben, um die Aktion "saying yes" auszuführen.

Instead of telling someone about something, try asking the noun about it.

Obiges "instead" leitet die Aktion tell nach ask um. Nachdem Mariel kein Krimi ist, benötigen wir die Unterscheidung zwischen "frage nach" und "erzähle von" nicht, wir fassen diese stattdessen zusammen.

Dann leiten wir die Aktion "vaguely communicating" (definiert durch GerX) um. Ein Blick auf den Actions Index sagt uns, wann diese Aktion greift:

Instead of vaguely communicating:
if the player can see the nurse:
try answering the nurse that;
stop the action;
if the player can see the caretaker:
try answering the caretaker that;
stop the action;
if the player can see the policeman:
try answering the policeman that;
stop the action;
say "Du bist momentan sprachlos."

Bei Mariel ist der Gesprächspartner immer eindeutig, da wir das Spiel so aufbauen, dass maximal eine weitere Person gleichzeitig mit dem Spieler in einem Raum sein kann. Der Ausdruck "can see" prüft, ob eine weitere Person (entweder die Krankenschwester oder der Hausmeister oder der Polizist) für den Spieler sichtbar und damit ansprechbar ist. Dann wird die Aktion auf diese Person umgeleitet. Falls keine Person anwesend ist wird die letzte Zeile ausgeführt und die Antwort "Du bist sprachlos." ausgegeben.

Mit dem folgenden Codeabschnitt schaffen wir eine neue Aktion "vaguely asking":

Vaguely asking is an action applying to one topic.

Jetzt benötigen wir noch das gesamte Vokabular, das für die Standardaktionen Ask/Tell greift, auch für unsere neue Aktion:

Understand "frag ueber/zu/nach [text]" as vaguely asking.
Understand "frag ueber/zu/nach [text]" as vaguely asking.
Understand "erzaehl ueber/von [text]" as vaguely asking.
Understand "red ueber/von [text]" as vaguely asking.
Understand "schwatz ueber/von [text]" as vaguely asking.
Understand "schwätz ueber/von [text]" as vaguely asking.
Understand "sprich ueber/von [text]" as vaguely asking.
Understand "sprech ueber/von [text]" as vaguely asking.
Understand "bericht ueber/von [text]" as vaguely asking.

Die neue Aktion "vaguely asking" bekommt wie gewohnt abschließend noch einen "check" Teil, der prüft, ob denn wirklich jemand zu sehen ist. Dieser Teil fängt den Aufruf der obigen Routine "Du bist momentan sprachlos." vorher ab.

Check vaguely asking:
if the player can not see a person:
say "Du redest ein wenig mit dir selbst. Das hat dich schon
immer beruhigt.";
stop the action.

Und einen "carry out" Teil, der die Aktion dann schließlich auf "ask" umleitet:

Carry out vaguely asking:
now the noun is a random other person in the location;
try asking the noun about.

Damit ist die Definition unserer neuen Aktion abgeschlossen. Inform7 weiß, was vor der Ausführung der Aktion zu prüfen ist und was zu tun ist, um diese auszuführen.
Am Ende fehlt noch:


The caretaker is a person.
The policeman is a person.

Der Zustand aller Aktionen kann nach dem ersten Kompilieren im rechten Fenster unter Index/Actions abgerufen werden. Hier sieht man übrigens auch, woher die Definition der Aktionen kommt - von Inform7 selbst, von verwendeten Erweiterungen oder vom Spiel.

Wir klicken jetzt auf "vaguely communicating" in der Liste der durch GerX definierten Aktionen und erhalten nebenstehendes Detailfenster:

Das gleiche Detailfenster gibt es auch für unsere neue Aktion "vaguely asking", zu finden unter "New actions defined in the source".

Hier sind zudem alle Regeln aufgelistet, die auf diese Aktion angewendet werden, die letzte in der Liste ist unsere neue "instead" Regel. Ein Klick auf den Pfeil daneben führt uns direkt zu der betreffenden Stelle im Quelltext – praktisch!

**Actions: Detail view** [Back to full view](#)

A single action in detail.
[About the action rulebooks](#)

Actions defined by the extension German by Team Gerx
Wide taking inventory, Tall taking inventory, Dressing, Undressing, Exiting from, Closing it with, Rubbing it with, Attacking it with, Burning it with, Going into, Vaguely going from, **Vaguely communicating**, Thinking about it, Location-leaving, Declining, Library checking

Vaguely communicating (past tense Vaguely communicated)
This action fires when the player enters communication-related commands missing the interlocutor and/or the topic, e.g. FRAG, REDE, ANTWORTE.

Typed commands leading to this action
"antwort", "red", "erzaehl", "frag"

Rules controlling this action
before doing something to a thing which is faraway
instead of doing anything other than examining to the rubbish
instead of doing something other than examining to something which is part of a person
instead of doing something other than entering or going near to a directive
instead of vaguely communicating

Wir sind mit dem "Gehen" noch nicht ganz fertig. Wir ergänzen im ersten Raum eine Regel, die hier ausnahmsweise einschreitet und den Spieler nicht einfach rausstürmen lässt, sondern dazu zwingt, Schritt für Schritt vorzugehen. Das brauchen wir, damit dann die "Überraschung" Wirkung zeigt.

```
before going south when location is hospital:
  if the player is on the bed:
    say "(Dazu steigst du erst von dem alten Bett.)[line break]";
    try silently getting off the bed;
  if the white door is closed:
    say "Dazu musst du erst die Tür öffnen.";
    stop the action;
  if the player does not know about nurse:
    move the nurse to the hospital;
    now the player knows about the nurse;
    stop the action.
```

Der letzte Abschnitt setzt ein Relation zwischen Spieler und der Krankenschwester. Was dann daraufhin passiert, sehen wir später. Auch "entering" muss abgefangen werden:

```
before entering the door:
  if the player is on the bed:
    say "Dazu musst du erst aufstehen.";
    stop the action;
  if the white door is closed:
    say "Dazu musst du erst die Tür öffnen.";
    stop the action;
  if the player does not know about nurse:
    move the nurse to the hospital;
    now the player knows about the nurse;
    stop the action.
```

Auch nach Norden geht es (noch) nicht raus.

```
Instead of going north when location is hospital:
  if the player is on the bed:
    say "Dazu musst du erst aufstehen.";
    stop the action;
  if the window is closed:
    say "Das Fenster ist vergittert. Keine Chance.";
    stop the action;
  say "Du kletterst durch das Fenster hinaus.";
  move the player to the backyard.
```

Es folgen noch ein paar "Kleinigkeiten". "Leg dich hin" wird als Aktion "sleeping" verstanden. Stattdessen soll sich der Spieler eben erst mal hinlegen (und wenn er will dann beim zweiten Mal versuchen, zu schlafen). Das Schlafen ist auch schwierig, jedoch naheliegend mit Kopfweh ...

```
Instead of sleeping when the player is not on the bed:
  try entering the bed.

Instead of sleeping:
  if player is nauseated:
    say "Du döst eine Weile vor dich hin und nickst schließlich
    doch kurz ein. Schweißgebadet wirst du wieder wach. Dein Kopfweh hat sich
    eher verschlimmert.";
  otherwise:
    say "Das Letzte, woran du jetzt denken solltest, ist Schlaf."
```

Das Herumhüpfen auf der Matratze, bitte, wenn der Spieler es so will:

**Instead of jumping when the player is on the bed:
say "Du springst ein wenig auf der Matratze."**

Und erwähnt wird vor dem Verlassen durch die Tür die Ecke. Daher erschaffen wir auch ein Objekt, das mit "schau ecke" angesprochen werden kann, natürlich nur, wenn die Tür offen ist.

The door-edge is a female scenery, here. Understand "link/recht/ander" and "Ecke" as the door-edge. The printed name is "Ecke". The description is "Du siehst dich noch genauer um - niemand zu sehen."

Ein kleiner Hinweis zum "Understand" Teil: Es ist unbedingt darauf zu achten, die Adjektive ohne Endungen zu deklarieren. Obwohl das komsich kling "link" wird dies als "links" oder "linkes" verstanden, da GerX die Endungen abschneidet. Fügt der Autor die Endungen an, wird das Adjektiv nicht mehr korrekt dekliniert. Unbedingt beachten!

**Before doing something to the door-edge when the white door is closed:
say "Du musst erst die Tür öffnen, um um die Ecke zu schauen."
instead.**

**Before doing something to the door-edge when player can see the nurse:
say "Das bringt jetzt nichts." instead.**

Wir handeln noch unsere "directive" ab, und das war' dann auch schon.

**Instead of entering the foreseen-corridor:
try going south.**

**Instead of going near the foreseen-corridor:
try going south.**

Jetzt komm aber unser erster NPC ins Spiel. [->]

2.12 NPC's

Part - The actors

Section - Mariel

The nurse is a woman. The printed name of the nurse is "Krankenschwester". Understand "Krankenschwester" or "Schwester" or "Mariel" or "Frau" as the nurse. The description of the nurse is "[The nurse] hat blonde Haare und trägt einen weißen Kittel. Ihre Gesichtszüge wirken hart und ihre Haut ist faltig und blättrig und scheint sich langsam von ihrem Gesicht abzulösen. Ihre Augen sind leblos und starr."

Soweit ist alles Standard. Mariel ist natürlich eine Frau, in der description wir der printed name von Mariel als Platzhalter mit [The nurse] eingesetzt. Das ist nachher wichtig, weil wir, nachdem sich Mariel vorgestellt hat, den "printed name" auf "Mariel" ändern werden.

**Rule for writing a paragraph about the nurse:
say "[one of][The nurse] wartet ungeduldig, mit strengem Blick, dass du dich wieder hinlegst.[or][The nurse] tritt ungeduldig von einem Fuß auf den anderen.[or][The nurse] schaut dich mit durchdringendem Blick an.[or][The nurse] macht dir**

unmissverständlich Zeichen, dich wieder hinzulegen.
[or]Streng deutet [the nurse] auf das Bett.[at random]".

Inform 7 gibt Objekte (also auch Personen) im Raum immer mit einem Standardsatz wie "Du siehst hier auch eine Krankenschwester" aus. Dies wollen wir aber etwas eleganter haben: Daher lassen wir den Text wie angegeben "per Zufall" ("at random") durchwechseln.

Die Krankenschwester hat auch einige Details, die entdeckt und näher betrachtet werden können:

The nurse-hair is plural-named part of the nurse. The printed name of the nurse-hair is "Haare". Understand "blond" and "Haare" as the nurse-hair. the description of the nurse-hair is "Die blonden Haare sehen ungepflegt und verfilzt aus und hängen in langen Strähnen herunter.".

The nurse-face is neuter part of the nurse. The printed name of the nurse-face is "Gesicht[-s]". Understand "Gesicht" or "Gesichtszuege [p]" or "Haut [f]" or "Hautfetzen [p]" as the nurse-face. the description of the nurse-face is "An vielen Stellen lösen sich Hautfetzen ab. Ihr Lächeln ist eine Farce.".

The nurse-eyes is plural-named part of the nurse. The printed name of the nurse-eyes is "Augen". Understand "Augen" as the nurse-eyes. the description of the nurse-eyes is "Ihre Augen sind starr, von dir abgewendet, so als schaut sie an dir vorbei.".

The gown is a male thing, worn by the nurse. The printed name of the gown is "Kittel[-s]". Understand "weiss" and "Kittel" and "Flecken [p]" as the gown. the description of the gown is "Der Kittel war vielleicht mal weiß. Jetzt ist er vor Dreck eher grau und mit seltsamen dunklen Flecken überzogen.".

Und zwei Aktionen werden wir noch abfangen:

Instead of taking the gown:
say "Vergiss es."

Instead of going south from hospital when player can see the nurse:
say "Du versuchst, an ihr vorbei zu kommen, aber ohne Erfolg."

Bevor jetzt der nächste Teil folgt, die

2.13 Themen und Konversation

Zuallerst ein paar Grundlagen:

Wir beginnen ganz unverfänglich; wenn der Spieler z.B. "sag Hallo" eingibt, stellt sich Mariel erst mal vor. Der "printed name" wird dann zu "Mariel" geändert und sie wird "proper-named" was den Artikel davor wegfallen lässt.

Section - Mariel conversation

Instead of answering the nurse that when the player is asked:
say "'Ähm, also ... ich ...'[paragraph break]Die Krankenschwester unterbricht dich. 'Ich bin Mariel und werde mich um Sie kümmern.' sagt sie und ein unterkühltes Lächeln umspielt ihre Lippen. 'Legen Sie sich erstmal wieder hin.'";
change the printed name of the nurse to "Mariel";
now the nurse is proper-named;
now the player is not asked.

Dann fangen wir das Ja- oder Nein-sagen ab, wenn das Attribut "is asked" für den Spieler zutrifft:

Instead of saying yes in the presence of the nurse when the player is asked:

say "Ja gerne, also ... ich ..."[paragraph break]Die Krankenschwester unterbricht dich. 'Ich bin Mariel und ich werde Ihnen nachher einen Tee bringen', sagt sie. Ein unterkühltes Lächeln umspielt dabei ihre Lippen. 'Legen Sie sich erstmal wieder hin.'";
change the printed name of the nurse to "Mariel";
now the nurse is proper-named;
now the player is not asked.

instead of answering the nurse that "nein" when the player is asked:

try saying no

instead of answering the nurse that "ja" when the player is asked:

try saying yes

Instead of saying no in the presence of the nurse when the player is asked:

say "Nein danke, also ... ich ..."[paragraph break]Die Krankenschwester unterbricht dich. 'Ich bin Mariel und sie sollten auf jeden Fall viel trinken', sagt sie. Ein unterkühltes Lächeln umspielt dabei ihre Lippen. 'Legen Sie sich erstmal wieder hin, ich bringe Ihnen nachher einen Tee.'";
change the printed name of the nurse to "Mariel";
now the nurse is proper-named;
now the player is not asked.

Das Attribut wird wieder gelöscht, falls die Regel abgearbeitet wurde. Mariel ist nun "proper-named", also fällt der Artikel davor weg, und der printed name wird auf "Mariel" geändert.

Folgende Regel fängt alle "sag [text]" Aktionen ab, die sonst zu einem unzulänglichen "Keine Antwort." führen würden. Die Illusion, dass der Spieler sagen kann, was er will, soll aufrecht erhalten werden.

Instead of answering the nurse that:

say "[one of]'Ähm, also ... ich ...'[or]'Ich wollte nur ...'[or]'Vielleicht könnte ich ...'[or]'Meinen sie ich ...'[or]'Ach bitte lassen Sie ...'[or]'Könnten sie bitte ...'[at random][line break][The nurse] unterbricht dich. [one of]'Legen sie sich einfach wieder hin, ich kümmere mich gleich um sie.'[or]'Sie sind doch ein braver Patient, oder?'[or]'Ich befehle ihnen, sich augenblicklich hinzulegen.'[or]'In ihrem Zustand herumzulaufen, ist unverantwortlich.'[or]'Ich hole etwas, das ihre Schmerzen lindert, aber legen sie sich erst wieder hin.'[at random]".

"Asking someone to try doing something" behandelt Aufforderungen des Spielers an Mariel, eine Aktion auszuführen. Dies wird standardmäßig in I7 als "persuasion rule fails" deklariert. Falls das anders sein soll, kann man eine entsprechende "persuasion rule" selbst definieren. Für Mariel benötigen wir das nicht, wir fangen die Aktionen gezielt mit "instead" ab.

Instead of asking the nurse to try exiting:

say "'Aber, aber', erwidert die Krankenschwester und schenkt dir einen mitleidigen Blick. 'Ich möchte Ihnen doch helfen, bitte legen Sie sich wieder hin.'".

Instead of asking the nurse to try going south:

say "'Ich werde ja gehen', sagt [the nurse], 'aber vorher gehen Sie wieder in ihr Bett zurück.'".

Wir belohnen den Spieler, der verrückterweise Mariel bittet, ihn oder sich selbst anzugreifen, mit sehr konkreten Hinweisen:

Instead of asking the nurse to try attacking the player:

say "'Genau das wollte ich tun', sagt [the nurse], 'aber vorher gehen Sie wieder in ihr Bett zurück.'".

Instead of asking the nurse to try attacking the nurse:

say "'Ich bin schon tot' sagt [the nurse] und ein diabolisches Grinsen umspielt ihr Gesicht, 'machen Sie sich daher um mich keine Sorgen.'".

Der Rest wird hier abgefangen, schließlich läßt sich Mariel nichts befehlen:

Instead of asking the nurse to try doing something:

say "'Sie wird dir sicher nicht gehorchen.'".

Falls der Spieler etwas ganz anderes von Mariel will, blitzt er selbstverständlich ab:

Instead of asking the nurse to try doing something:

say "'Sie wird dir sicher nicht gehorchen.'".

Auch irgendwelche Gegenfragen akzeptiert Mariel nicht:

Instead of asking the nurse about when the player is asked:

say "'Die Krankenschwester unterbricht dich. 'Beantworten Sie mir doch bitte erst meine Frage.'".

Den Hauptteil der Konversation handeln wir über "ASK-Topics" ab. Die Weichen dazu haben wir vorher schon gestellt, indem wir die TELL AKTION auf ASK umgeleitet haben. Der allgemeine Teil gibt dem Spieler den Hinweis, das richtige Thema noch nicht gefunden zu haben und was anderes zu probieren:

Instead of asking the nurse about:

say "'Dir fehlen irgendwie die Worte. Vielleicht solltest du versuchen, bekannte Themen anzusprechen.'".

Jetzt folgen der Reihe nach die verschiedensten Themen, die sich fast alle auf Gegenstände aus dem Spiel beziehen. Wir fassen diese in einer Tabelle zusammen. Die Tabelle soll 3 Spalten bekommen, in der ersten steht das "Topic", das auf Übereinstimmung mit der Spielereingabe geprüft wird. In der zweiten halten wir in der Variable "Done" fest, ob der Spieler das schon einmal gefragt hat. In der dritten wird in den String "response" der Antworttext geschrieben. Dar war's schon – unglaublich, aber wahr.

Folgende Regel checkt die Übereinstimmung, setzt ggf. Done auf "1" und gibt den Antworttext aus:

Instead of asking the nurse about a topic listed in the Table of nurse answers:

if done entry is 0:

say "[response entry][line break]";
now done entry is 1;

otherwise:

annoy the nurse.

Jetzt noch die Tabelle selbst, etwas schlecht lesbar, aber sehr komplakt:

Table of nurse answers

Topic Done Response

"brechstange/stange" 0 "'Hätten Sie zufällig eine Brechstange für mich?', fragst du.[paragraph break]'Was für eine dumme Idee', sagt [the nurse]. 'Sie brauchen jetzt keine Brechstange, glauben Sie mir, sie müssen sich ausruhen.'"

"tuer/ausgang" 0 ""Wegen der Tür ...', sagst du.[paragraph break]'Sie sollten die Tür immer geschlossen halten', sagt [the nurse]. 'Sonst bekommen Sie noch einen Zug und erkälten sich. Legen Sie sich wieder auf das Bett und ich bin gleich wieder bei Ihnen.'"

"bett/liege" 0 ""Wegen dieser Liege ...', sagst du.[paragraph break]'Davon rede ich doch die ganze Zeit', sagt [the nurse]. 'Sie brauchen jetzt viel Ruhe.'"

"schrank/schrantuer" 0 ""Ich kriege den Schrank nicht auf', sagst du.[paragraph break]'Der Schrank ist vollkommen unwichtig', sagt [the nurse]. 'Konzentrieren Sie sich jetzt auf ihre Genesung.'"

"fenster/gitter/fenstergitter" 0 ""Warum sind die Fenster vergittert?', fragst du.[paragraph break]'Das Gitter ist zu Ihrer eigenen Sicherheit da', sagt [the nurse]. 'Sie sehen ja selbst, wie unruhig Sie sind. Begeben Sie sich sofort zurück ins Bett.'"

"unfall/nacht/auto/splitter/strasse" 0 ""Wegen dem Unfall ...', sagst du.[paragraph break]'Sie waren sehr leichtsinnig', sagt [the nurse]. 'Sie hatten viel Glück, Sie könnten jetzt tot sein.'"

"mich/mir/dich/dir" 0 ""Was ist eigentlich mit mir los?', fragst du.[paragraph break]'Gott sei Dank haben wir Sie gefunden', sagt [the nurse]. 'Sie hätten verbluten können dort draußen. Machen Sie sich keine Sorgen, ich Sorge für Sie.'"

"ihr/mariel/krankenschwester" 0 ""Ähm, wegen Ihnen ...', sagst du.[paragraph break]'Ich bin hier völlig unwichtig', sagt [the nurse]. 'Das wichtigste ist, dass Sie schnell wieder gesund werden.'"

"zimmer/raum/ort/kammer/geruempel/ausgang/abstellkammer" 0 ""Warum bin ich in dieser Abstellkammer?', fragst du.[paragraph break]'Der Raum ist nicht der schönste, ich weiß', sagt [the nurse]. 'Sie bekommen einen anderen, wenn die Belegung es zulässt.'"

"krankenhaus/hospital/gebaeude/klinik" 0 ""Was ist das für eine Klinik?', fragst du.[paragraph break]'Wir sind eine sehr moderne Klinik mit einer enormen Erfolgsquote', sagt [the nurse]. 'Bei uns beträgt die Genesungsrate fast hundert Prozent. Sie sehen, bei uns sind Sie in besten Händen.'"

"inform/gerx/inform7/if/i7" 0 ""Kennen Sie Inform 7?', fragst du.[paragraph break]'Davon habe ich bisher gar nichts gehört', sagt [the nurse]. 'Aber das macht nichts. Legen Sie sich einfach wieder hin.'"

"textadventure/textadventures" 0 ""Kennen Sie Textadventures?', fragst du.[paragraph break]'Ja, ich habe damals 'The Lurking Horror' gespielt', sagt [the nurse]. 'Aber das ist im Moment zu aufregend für Sie. Legen Sie sich einfach wieder hin.'"

"spritze/infusion/injektion/fluessigkeit/medikament/medikamente/hilfe/schmerzen/schmerz/wunden/wunde/schnittwunden/schnittwunde/verletzung/verletzungen" 0 ""Ich habe diese Verletzungen ...', sagst du.[paragraph break]'Ja, ja', sagt [the nurse] ungeduldig. 'Ich hole ja gleich etwas, wenn Sie sich nur wieder brav hinlegen.'"

"haut/hautfetzen/gesicht" 0 ""Wegen Ihrem Gesicht ...', sagst du.[paragraph break]'Ach das', sagt [the nurse]. 'Das ist nur ein leichter Sonnenbrand. Nichts, worüber Sie sich Gedanken machen sollten.'"

"zombie/zombies/untote/horror/horrorfilm" 0 ""Ich muss irgendwie an diesen Horrorfilm mit Zombies gestern im Fernsehen denken', sagst du.[paragraph break]'Ich weiß nicht genau, was Sie meinen', sagt [the nurse] ausweichend. 'Ich habe ihn leider nicht gesehen.'"

"korridor/flur/flure/korridore" 0 ""Was ist hinter dieser Tür?', fragst du.[paragraph break]'Glauben Sie mir', sagt [the nurse]. 'Sie wollen sicher nicht wissen, was da draußen so alles passiert. Es ist das Beste, Sie ruhen sich erstmal aus.'"

"dunkelheit/nacht" 0 ""Ist es Nacht?', fragst du.[paragraph break]'Sie haben recht, mein Bester', sagt [the nurse]. 'Ich mag die Dunkelheit, denn sie legt ihre Schatten über alle Ereignisse des Tages.'"

"wetter" 0 ""Was haben wir für Wetter?', fragst du.[paragraph break]'Mit dem Wetter', sagt [the nurse], 'können Sie sich morgen noch

beschäftigen. Sie müssen jetzt schlafen."
 "arzt/aerzte/chefarzt/doktor" 0 ""Kann ich einen Arzt sehen?', fragst du.
 [paragraph break]'Die Ärzte sind sehr beschäftigt heute', sagt [the nurse].
 'Wir hatten mehrere schwere Verkehrsunfälle in der Gegend. Sie können
 sicher sein, dass der Arzt morgen zu Ihnen kommt.'
 "augen/auge" 0 ""Ähm, wegen Ihrem Auge', sagst du.[paragraph
 break]'Machen Sie sich keine Sorgen um mich', sagt [the nurse] mit einer
 monotonen Stimme, 'sondern eher um sich selbst. Ruhen Sie sich aus.'"

Wichtig ist nur, die Tabulatoren korrekt einzusetzen: also immer zwischen den Tabelleneinträgen.

Oben haben wir noch eine neue Phrase definiert, "Annoy the nurse". Was dann zu tun ist, sagen wir Inform 7 auf folgende Weise:

To annoy the nurse:

say "[one of]'Da sehen Sie selbst, wie verwirrt Sie sind', merkt [the
 nurse] an, 'Dieses Thema hatten wir doch vorhin
 schon.'[or]'Sie wiederholen sich ständig', sagt [the nurse]
 genervt.[or]'Sie sind ja komplett durcheinander', bemerkt [the
 nurse]. 'Wir haben doch bereits darüber gesprochen.'[or]'Sie
 brauchen dringend Ruhe', sagt [the nurse]. 'Sie erzählen mir ja
 immer dasselbe.'[or]'Legen Sie sich doch bitte hin', sagt [the
 nurse]. 'Ich hole gleich was für Sie.'[at random]".

Damit ist dieser Abschnitt auch schon abgeschlossen.

Hinweis: Auch die beiden anderen NPC's, der Polizist und der Hausmeister
 "funktionieren" nach dem selben Schema, daher wird hier jetzt nicht näher
 darauf eingegangen, sondern die eigentlich spielrelevanten Mechanismen,
 näher untersucht.

Genauso könnten wir auch jeden Raum und jedes scenery-object
 auseinandernehmen, das würde aber für das Verständnis des Spiels wenig
 "Neues" bringen, daher werden im folgenden nur noch die wirklich wichtigen
 Sachen erwähnt.

Also gilt im Folgenden der Fokus dem [->]

3. Gameplay

3.1 Innenbereich

"Scenes" sind im Hintergrund ablaufende Ereignisse, die entweder nur als "Merker" fungieren, oder tatsächlich z.B. zeitbedingte Ereignisse auslösen. Wir schauen uns hier die erste Szene an, auf die der Spieler trifft, ab dem Moment, als Mariel das erste Mal "auftaucht".

Blocking is a scene. Blocking begins when the player can see the nurse.

Hinweis: Eine Scene wird nur einmal abgearbeitet, es sei denn sie besitzt das Attribut "recurring". Das heißt in unserem Fall, dass die Szene nur einmal in Gang gesetzt wird, nämlich wenn der Spieler Mariel das erste Mal sieht.

Die Definition, was passiert, wenn die Szene beginnt, hier wird jetzt auch das zuvor erwähnte Attribut "asked" für den Spieler gesetzt:

When blocking begins:

**say "Gerade als du den Raum verlassen willst, steht urplötzlich eine Krankenschwester in der Tür. 'Aber, aber, mein Bester. Wer wird denn schon aufstehen?', sagt sie mit einem kühlen Unterton und betritt das Zimmer. 'Soll ich Ihnen etwas zu trinken bringen?';
now the player is asked.**

Blocking ends naturally when the player is on the bed.

Blocking ends abruptly when the player can see the nurse for the 16th turn.

Wir unterscheiden zwei Arten von Ende – ein "natürliches" Ende und ein "abruptes". Das eine tritt ein, wenn sich der Spieler "brav" wieder hinlegt, das andere wenn er sich weigert (nach 16 Zügen). Jetzt fehlt noch der Part, wo wir Inform 7 sagen, was bei den Enden zu tun ist:

When blocking ends naturally:

**if nurse is not proper-named:
change the printed name of the nurse to "Mariel";
now the nurse is proper-named;
say "'Ich bin übrigens Mariel', stellt sie sich vor.[paragraph
break]";
say "'So ist es brav, sie brauchen jetzt viel Ruhe', sagt [the nurse],
'ich komme später wieder zu Ihnen.' Die Tür wird abrupt
zugeschlagen.";
now the white door is closed;
move the nurse to the backyard.**

When blocking ends abruptly:

**if nurse is not proper-named:
change the printed name of the nurse to "Mariel";
now the nurse is proper-named;
say "'Ich bin übrigens Mariel', stellt sie sich vor.[paragraph
break]";
say "Mit einer kraftvollen Bewegung stößt dich [the nurse] auf das
Bett. 'So ist es brav, sie brauchen jetzt viel Ruhe, ich komme
später wieder zu Ihnen.' Die Tür wird abrupt zugeschlagen.";
move player to bed;
now the white door is closed;
move the nurse to the backyard.**

Damit ist der erste "spielrelevante" Abschnitt abgearbeitet. Wir erinnern uns: Der Spieler wacht in dem unbekannten Zimmer auf, öffnet die Tür und versucht nach Süden zu gehen. Mariel erscheint und wartet darauf, dass der Spieler sich wieder hinlegt. Konversation ist ziemlich aussichtslos, denn Mariel ist stur, wiegt den Spieler "in Sicherheit" und lässt ihn nicht vorbei.

Nachdem der Spieler sich wieder hingelegt hat, rappelt er sich wieder auf – es bleibt ihm nur der Weg nach Süden, denn der Lüftungsschacht ist in seinem derzeitigen Zustand (Attribut "nauseated") unerreichbar.

Jetzt ist es an der Zeit, Stück für Stück die fehlenden Räume zu implementieren. Das wären zunächst die drei Teile des Flurs. Hier passiert nicht viel Aufregendes, außer dem bemerkenswerten kleinen Zettel an der Tür des Stationszimmers, dem wir Aufmerksamkeit schenken:

Section - The corridor east

The note is a female fixed in place in the corridor-east. "Am Stationszimmer hängt eine Liste aus." Understand "liste" or "patientenliste" or "namen [p]" as the note. The printed name is "Liste". The description is "Eine Patientenliste, auf der verschiedene Namen zu lesen sind. Alle Namen bis auf deinen und einen weiteren sind durchgestrichen."

Der übrige code kann einfach aus dem Quelldokument kopiert werden, als nächstes wenden wir uns dem "Stationszimmer" zu.

Section - Mariel's chamber

Mariels chamber is a room, in the indoors, south from corridor-east. "Der Raum ist etwa so groß, wie der Raum, in dem du aufgewacht bist und genauso dreckig. Hier befindet sich ein Tisch und ein Schrank mit einem roten Kreuz darauf.[paragraph break]Der einzige Ausgang im Norden führt zum Korridor."

The printed name is "Stationszimmer".

Hier gibt es den Medizinschrank:

The redcross-cabinet is a male enterable scenery closed and openable container in the mariels chamber. The printed name is "Schrank[-s]". Understand "Schrank" and "Schranktuer [f]" and "rot" as the redcross-cabinet. The description is "Der Schrank sieht schäbig und alt aus, das rote Kreuz darauf ist abgeblättert und kaum mehr erkennbar. [if closed]Der Schrank ist im Moment geschlossen[otherwise]Der Schrank ist geöffnet[end if].".

Und – Oh Wunder – eine Pillendose, mit Pillen drin:

The can is a female closed and openable container in the redcross-cabinet. Understand "unleserlich" and "Pillendose" and "Dose" and "Aufschrift" as the can. The printed name is "Pillendose". The description is "Eine kleine Dose mit einer unleserlichen Aufschrift."

The pills are a plural-named edible in the can. Understand "klein" and "weiss" and "rund" and "Tabletten/Pillen" and "Tablette/Pille [f]" as the pills. the printed name is "Tabletten". the description is "Kleine, runde, weiße Tabletten."

"Schluck Tabletten" läuft unter der Aktion "eating", daher:

Instead of drinking the pills:
try eating the pills.

Und eine abschließende Regel für das Essen der Pillen, dies sich diesmal in "report" einklinkt, d.h. die Pillen sind hier schon geschluckt:

Report eating the pills:
now the player is not nauseated;
say "Du schluckst eine und wartest ... das Kopfweg wird nicht besser ... Du schluckst noch eine ... eine kleine Linderung ... noch eine ... noch etwas besser, aber es geht einfach zu langsam ... Der Gedanke, das Kopfweg loszuwerden wird übermächtig und plötzlich schluckst du alle auf einmal. Dein Kopfschmerz nimmt endlich ab und die Übelkeit verschwindet, bis du dich fast wieder normal fühlst." instead.

Das Attribut des Spielers "nauseated" wird damit nach Einnahme der Pillen gelöscht.

Erwähnenswert ist hier noch der Tisch:

The desk is a male scenery supporter in the mariels chamber. Understand "weiss" and "Tisch" and "flecken [p]" as the desk. The printed name is "weiß[^] Tisch[-s]". Understand "weiss" and "Tisch" and "flecken [p]" as the table. The description of the desk is "Ein weißer Tisch mit dunklen Flecken darauf. An einer Seite ist eine Schublade angebracht."

Report searching the desk:
say "Dein Blick fällt zunächst auf eine kleine Schublade an der Seite des Tisches."

Und der Monitor darauf, der "Verursacher" des leisen Pieptons, der hier ständig zu hören ist.

The monitor is a male fixed in place on the desk. Understand "Monitor" or "Bildschirm" or "Ueberwachungsmonitor" as the monitor. The printed name is "Monitor". The description is "Das Piepen kommt von diesem Monitor -scheint ein Überwachungsmonitor zu sein. Er zeichnet mit einer Kurve den wöchentlichen Mengenverlauf einer Flüssigkeit auf. Die Werte schwanken zwischen fünf und fünfundzwanzig Liter. Der Zeitbalken geht bis an den Jahresanfang zurück."

Hier benötigen wir eine Regel, die den Monitor elegant beschreibt. Nachdem der Tisch ein "scenery-object" ist, beschreibt Inform 7 den Monitor, da dieser sich auf dem Tisch befindet mit einem Standard Satz. Das ist übrigens eine der umstrittensten Regeln in Inform7 und heißt "describe what's on scenery supporters rule".

Daher unsere eigene Regel:

Rule for printing a locale paragraph about the desk when the number of things on the desk is 1:
say "Auf dem Tisch steht ein Monitor, von dem dieses ständige Piepen ausgeht.";
now desk is mentioned.

Diese Regel greift nur, wenn der Monitor sich als einziges auf dem Tisch befindet. Hat der Spieler was dazugelegt, werden die Gegenstände "stadardmäßig" aufgezählt – für unseren Zweck so ausreichend. Das Attribut "mentioned" sagt Inform7, dass keine weitere Erwähnung des jeweiligen Objekts mehr notwendig ist.

Die Schublade, mit einem Zeitungsartikel und einem Schlüssel, leider erst mal verschlossen:

The drawer is a female scenery closed and locked container in the mariels chamber. Understand "Schub [m]" or "Schublade" or "Lade" as the drawer. The printed name is "Schublade". The description is "Eine kleine Schublade,

im Moment [if open]geöffnet[else]geschlossen[end if]."

The scrapped article is a male thing in the drawer. Understand "Artikel" and "Zeitungsartikel" and "Blatt [n]" and "Papier [n]" as the scrapped article. The printed name is "Zeitungsartikel". the description is "Der Artikel ist Ende letzten Jahres datiert. Darin steht, dass sich der 'Orden der Hundert Seelen' dazu entschlossen hat, einen Vertrag mit einem Pharmakonzern zu unterzeichnen. Damit hat sich der Orden verpflichtet, monatlich eine feste Menge an Blut an den Konzern zu liefern."

Instead of examining the scrapped article when the player is not holder of the scrapped article:

say "Dazu musst du den Zeitungsartikel erst in der Hand halten."

The master key is a male thing in the drawer. Understand "schwer" and "zweit" and "Hauptschluessel/Generalschluessel/Schluessel/Aufdruck/Zweitschluessel" as the master key. The printed name is "schwer[^] Schlüssel". The description is "Ein kleiner Aufdruck an der Seite deklariert diesen Schlüssel als Zweitschlüssel."

Der Schlüssel schließt praktischerweise alle Türen im Krankenhaus auf.

The master key unlocks the first-room-door, the second-room-door, the third-room-door, the fourth-room-door, the cathedral-door.

Ein wenig Disambiguierung fehlt noch, da wir ja mit zwei Schlüsseln (Generalschlüssel und Schubladenschlüssel operieren):

Does the player mean unlocking the first-room-door with the master key: it is very likely.
Does the player mean unlocking the second-room-door with the master key: it is very likely.
Does the player mean unlocking the third-room-door with the master key: it is very likely.
Does the player mean unlocking the fourth-room-door with the master key: it is very likely.
Does the player mean unlocking the cathedral-door with the master key: it is very likely.

Eine "Does the player mean" rule leitet eine unvollständige oder zweifelhafte Eingabe, z.B.

>schliess 128 mit schlüssel auf

Was meinst du, den winzigen Schlüssel oder den schweren Schlüssel?

Wird jetzt zu:

>schliess 128 mit schlüssel auf

(dem schweren Schlüssel auf)

Du schließt die Tür auf.

So, jetzt kann der Spieler zurück in Zimmer 126 (dessen Name ja jetzt bekannt ist, da der Flur das Attribut "visited" bekommen hat) und in den Schacht klettern.

Wir machen mit dem Raum über dem Zimmer weiter:

Section - Ventilation system

Ventilation system is a male room, above the hospital, in the indoors. "Hier ist es sehr eng und stickig und es ist so gut wie nichts erkennbar. Etwas weiter östlich endet der Schacht vor einem Gitter, durch das von unten Licht herein fällt.[paragraph break]Der einzige Ausgang ist das Loch, das zurück in dein Zimmer führt."

The printed name of the ventilation system is "Schacht[-s]".

Wir erschaffen ein Dummy Objekt namens "trap":

The trap is a thing. The trap can be set.

Altbekanntes, aber wir müssen jetzt das Verlassen manuell abhandeln, da der Spieler ja logischerweise wieder auf dem Bett landen soll, außerdem läuft eine Art "cut-scene" ab:

Instead of going down from ventilation system:

say "Vorsichtig handelst du dich wieder nach unten auf das Bett.";

if the trap is set:

now the trap is not set;

now the player is manacled;

say "Als du gerade wieder heruntergeklettert bist, spürst du, wie dich eine eisige Hand am Nacken packt. 'Was haben sie da oben zu suchen?', fragt Mariel und wirft dich mit einer geschickten Handbewegung auf das Bett.[paragraph break]'Ich, äh ...', stammelst du und versuchst dich zur Wehr zu setzen. Im Gerangel bemerkst du, dass ein kleiner Schlüssel aus Mariels Kittel auf den Boden fällt.[paragraph break]'Wer nicht brav ist, wird bestraft', sagt Mariel, die anscheinend nichts davon bemerkt hat und fesselt dich mit einem Strick ans Bett. 'Jetzt bleiben sie endlich liegen, sobald wie möglich werde ich ihre Schmerzen lindern.'[paragraph break]Mit diesen Worten lässt sie dich allein und schließt die Tür hinter sich.";

now the white door is closed;

move the manacles to the player;

move the tiny key to the hospital;

now the initial appearance of the tiny key is "Ein kleiner Schlüssel liegt auf dem Boden.";

move the player to the bed.

Also, wenn die Falle "gestellt" ist, wird der Spieler erwischt und ans Bett gefesselt: Mariel verliert dabei einen kleinen Schlüssel, der die Schublade aufschließt.

The tiny key is a male thing. Understand "winzig/klein" and

"Schluessel/Silberschluessel" as the tiny key. the printed name is "winzig[^] Schlüssel". the description is "Ein kleiner Schlüssel aus Silber."

The tiny key unlocks the drawer.

Does the player mean unlocking the drawer with the tiny key: it is very likely.

Bevor es weitergeht, vervollständigen wir erst den nächsten Raum, das Schachtende.

Section - Eastend

East end is a neuter room, east from the ventilation system, in the indoors.

"Durch ein kleines Gitter im Boden fällt etwas Licht in den Schacht. Unter dir befindet sich ein weiteres Zimmer, ähnlich dem, in dem du aufgewacht bist. Der Raum ist mit einem Tisch und einem Schrank möbliert, die Tür ist zugemauert worden.[paragraph break]Die einzige Richtung, in die du dich von hier bewegen kannst, ist zurück nach Westen."

The printed name is "Schachtende".

Die Idee ist, dass sich hier ein Gitter befindet, das zu klein ist, um hindurchzupassen, aber geöffnet

werden kann, um den Schrank in dem Zimmer darunter öffnen zu können.

The grate is a neuter scenery openable and closed container in the east end. The printed name of the grate is "Lüftungsgitter". Understand "Gitter" or "Lueftungsgitter" as the grate. the description of the grate is "Ein kleines Lüftungsgitter, das im Moment [if open]geöffnet[otherwise]geschlossen[end if] ist."

Instead of inserting something into the grate:
say "Du willst [den noun] sicher nicht verlieren."

Das Öffnen soll mit "brich Gitter auf" ebenfalls möglich sein:

Instead of attacking the grate:
if the grate is open:
say "Keine weitere Gewalt notwendig." instead;
say "Du schlägst mit aller Kraft auf das Gitter ein und es schwingt nach unten klappend auf.";
now the grate is open.

Instead of dropping something when location is east end:
say "Du willst [den noun] jetzt keinesfalls hier ablegen.".

before taking something when location is east end:
if the grate is closed:
say "Dazu musst du erst das Lüftungsgitter öffnen.";
stop the action.

Der Schrank ist dem in Zimmer 126 ähnlich:

The station-cabinet is a male enterable faraway scenery closed and openable container in the east end. The printed name of the station-cabinet is "Schrack[-s]". Understand "rostig" and "Schrack" and "Schracktüer [f]" and "Metallschrack" and "Beulen [p]" and "Rost [m]" and "Roststellen [p]" as the station-cabinet. The description of the station-cabinet is "Der Schrack schaut genauso aus, wie der in deinem Zimmer. [if closed]Der Schrack ist im Moment geschlossen[otherwise]Der Schrack ist geöffnet[end if].".

Nach dem Öffnen und Schließen des Gitters wird manuell interveniert, da der Schrack dann entweder "faraway" (unerreichbar) sein soll, oder eben nicht:

After opening the grate:
say "Du öffnest das Lüftungsgitter.";
now the grate is open;
now station-cabinet is not faraway.

After closing the grate:
say "Du schließt das Lüftungsgitter wieder.";
now the grate is closed;
now station-cabinet is faraway.

Das Tagebuch des Pflegers verstecken wir im Schrack, darin erfährt der Spieler mehr über die Geschichte des Krankenhauses, sofern er sich die Mühe macht, und es mehrmals liest.

The Diary is a thing in the station-cabinet. Understand "buch" or "tagebuch" as the diary. The printed name is "alt[^] Buch". the description is "[One of]Du überfliegst die Seiten hastig. Eine Art Tagebuch eines gewissen John Rezznar, der hier wohl vor über vierzig Jahren als Krankenpflieger gearbeitet hat. Die ersten Einträge besagen, dass die Klinik offenbar Ende der siebziger Jahre mit dem Fortschritt nicht mehr Schritt halten konnte und als

hoffnungslos veraltet still gelegt wurde.[or]Du blätterst hastig weiter. Die Oberschwester Mary Higgins und dieser John sind als einzige hier geblieben. Sie haben mit alten Medikamenten experimentiert. Am Ende ihrer Versuche, konnten sie tote Tiere, die an Versuchen gestorben waren, mit der Injektion einer Flüssigkeit wieder zum Leben erwecken.[or]Die letzten Zeilen sind mit zitteriger Hand geschrieben: 'Mir wird die Ehre zuteil, als erster Mensch die Wirkung des Serums zu erfahren. Mary sagt, es sei der erste Schritt der Menschheit in die Unsterblichkeit. Ich bin ruhelos. Ich warte auf den Tagesanbruch.'[stopping]"

After examining the diary:
now the player knows about the diary.

Instead of examining the diary when the player is not holder of the diary:
say "Du musst das alte Buch erst in die Hand nehmen."

Das Nehmen des Buchs reicht aus, um die Falle in Gang zu setzen:

Instead of taking the diary when diary is in station-cabinet:
say "Du nimmst das Buch vorsichtig aus dem Schrank.";
move diary to player;
now the trap is set.

Instead of searching the station-cabinet when station-cabinet is open:
say "Im obersten Fach des Schranks siehst du [if nothing is in station-cabinet]nichts[else][a list of things in station-cabinet with accusative] [end if]."

Die obige say-Phrase listed alles im Schrank so auf, als läge es im obersten Fach. Der Spieler kann ja sicher nicht die unteren Fächer erreichen, wenn er sich im Schacht befindet.

The dummy room is below the east end.

Diesen Raum benötigen wir lediglich, damit der Spieler versuchen kann, nach unten zu gehen, was aber niemals möglich sein wird:

Instead of going down from east end:
say "Selbst wenn das Lüftungsgitter geöffnet ist - du passt niemals durch das enge Loch hindurch."

Und zum Schluss noch ein paar Geh-Konventionen:

Instead of going nowhere from the east end:
say "Die Schachtwand ist im Weg."

Instead of exiting when location is east end:
try going down.

Instead of going south from east end:
say "Unmöglich, die Tür von hier aus zu erreichen."

Instead of opening the boarded-door when location is east end:
say "Unmöglich, die Tür von hier aus zu erreichen."

Inform 7 gibt in Klammern hinter der Raumbeschreibung eine Mitteilung aus, falls der Spieler sich auf etwas befindet. Hier wollen wir eingreifen, wenn er gefesselt ist und stattdessen schreiben "(an das Bett gefesselt)". Dazu hatten wir ja die "custom messages" eingebunden, was uns jetzt zugute kommt:

Section - Custom Messages

Table of custom library messages (continued)

library-action	library-message-id	library-message-text
Looking action	8	"[if the player is manacled] (an [end if])[if the library message object is a supporter and player is not manacled] (auf [end if])[if the library message object is a container and player is not manacled] (in [end if])[if player is manacled][den library message object][else][dem library message object][end if][if the player is manacled] gefesselt[end if])[ignore library line break]"

Soweit so gut, jetzt machen wir mit dem Zimmer weiter:

The manacles are a plural-named thing. Understand "dick" and "Fesseln [p]" and "Strick [m]" and "Seil [n]" as the manacles. The printed name is "Fesseln". the description is "Ein dicker Strick, der dich an das Bett fesselt."

Bitte sehr, die Fesseln. Und natürlich das Blocken von allen möglichen Aktionen, wenn gefesselt:

Instead of getting off something (called the item) when player is manacled:
say "Du bist an [den item] gefesselt."

before entering something when player is manacled:
say "Du bist an das Bett gefesselt." instead.

Instead of cutting the manacles:
say "Ja klar, bloß wie?"

Instead of doing something other than examining or rubbing or looking or taking inventory when the player is manacled:
say "Du bist gefesselt."

Befreien geht nur, indem der Spieler die Fesseln an der "scharfen" Bettkante (siehe Beschreibung Bett) reibt – reiben ist ein bereits intergriertes Standard-Verb. Wir brauchen es aber mit einem zweiten Nomen.

Section - Rubbing it on

Rubbing it on is an action applying to two things.

Understand "reib [something] an/auf [something]" as rubbing it on.

check rubbing it on:
if the player is not holder of the noun:
say "Du musst [den noun] erst in der Hand halten." instead.

Carry out rubbing it on:
say "Das bringt nichts."

Aber bitte dreimal reiben – man braucht halt etwas Geduld:

rubcount is a number that varies. rubcount is 0.

Instead of rubbing the manacles:
say "Wenn du nur eine Idee hättest, an was!"

Instead of unscrewing the manacles with something (called the goal):
try rubbing the manacles on the goal.

Instead of rubbing the manacles on something (called the edge):

```

if edge is not bed:
    say "Wenn du nur einen scharfen Gegenstand hättest!"
    instead;
otherwise:
    increase rubcount by 1;
    if rubcount is 1:
        say "Du reibst mit den Fesseln an der scharfen
            Bettkante. Nach einiger Zeit hat sich der Strick
            etwas gelockert.";
    if rubcount is 2:
        say "Du reibst weiter mit den Fesseln an der scharfen
            Bettkante. Sie lockern sich mehr und mehr.";
    if rubcount is 3:
        say "Du reibst und reibst mit den Fesseln an der
            scharfen Bettkante. Schließlich trennst du den
            Strick durch. Du bist frei. Vor Erleichterung
            schleuderst du deine Fesseln aus dem
            Fenster.";
        now the player is not manacled;
        remove the manacles from play.

```

Den Abschluss soweit machen noch das Loch (vom Schacht aus) und paar Gehen-Konventionen, wie gehabt:

The ventilation-hole is a neuter scenery in the ventilation system. The printed name of the ventilation-hole is "Loch[-s]". Understand "Loch" as the ventilation-hole. The description of the ventilation-hole is "Das Loch ist gerade groß genug, um hindurch zu passen."

Instead of going nowhere from the ventilation system:
say "Die Schachtwand ist im Weg."

Instead of entering the ventilation-hole:
try going down.

Instead of exiting when location is ventilation system:
try going down.

Somit ist auch die zweite Begegnung mit Mariel abgeschlossen, der Spieler weiß jetzt mehr über die Geschichte und konnte der Krankenschwester ein zweites Mal entkommen. Jetzt folgt der dritte Anlauf.

Von den Fesseln befreit, steht der Spieler wieder auf und nimmt natürlich den kleinen Schlüssel an sich. Dieser sperrt die Schublade im Stationszimmer und der Spieler bekommt den Generalschlüssel und den Zeitungsbericht, der etwas Hintergrund zum Handeln von Mariel aufdeckt.

Mit dem Generalschlüssel kann er alle weiteren Räume aufsperrn. Wir fahren mit dem "second room" fort, bzw. mit der dazugehörenden Tür:

The second-room-door is a scenery female locked door, south from corridor-west, north from second-room. Understand "weiss" and "tuer" and "zu" and "zimmer" and "129" as the second-room-door. The printed name is "Tür". The description is "Eine weiße Tür, wie die anderen auch."

The second-room is a room, in the indoors. "Dieser Raum ist etwas größer, aber genauso schmutzig und besitzt statt eines Fensters nur einen schmalen Lüftungsschlitz in der Wand. Ein altes Bett und ein Schrank stehen verlassen da. [paragraph break]Eine Tür befindet sich im Norden."

The printed name is "Zimmer 129".

Hier gibt es wieder einige Objekte, die wir schon von anderen Räumen kennen, bemerkenswert jedoch:

The gap is a male scenery in the second-room. Understand "schmal" and "Schlitz/Lueftungsschlitz" as the gap. The printed name is "Lüftungsschlitz[-s]". The description is "Der Schlitz ist gerade groß genug, um hinaus sehen zu können."

Instead of searching the gap:

say "Du siehst Mariel mit einem Mann auf dem Eingangshof stehen.
Sie sind beide damit beschäftigt, Sachen in eine große
Mülltonne zu stopfen.";
move the thing-you-see-outside to the second-room.

The thing-you-see-outside is a faraway neuter scenery. Understand "Mann [m]" or "Mariel [f]" or "Tonne [f]" or "Muelltonne [f]" as the thing-you-see-outside. The printed name is "was draußen passiert,". The description is "Es ist nicht mehr zu erkennen, so sehr du dich auch verrenkst."

Das gibt etwas andere Standard Antworten wie:

>**schau aus schlitz**

Du siehst Mariel mit einem Mann auf dem Eingangshof stehen. Sie sind beide damit beschäftigt, Sachen in eine große Mülltonne zu stopfen.

Ein stechender Schmerz wütet in deinem Kopf.

>**schieb mariel**

Das was draußen passiert, ist zu weit entfernt.

Wir machen mit dem vierten Raum weiter:

Section - The fourth room

The fourth-room-door is a female locked scenery door, south from corridor, north from fourth-room. Understand "weiss" and "tuer" and "zu" and "zimmer" and "127" as the fourth-room-door. The printed name is "Tür". The description is "Eine weiße Tür, wie die anderen auch."

The fourth-room is a room, in the indoors. "Dieser Raum ist etwas größer, aber genauso schmutzig und besitzt kein Fenster. Auch hier ist ein Bett und einer dieser alten Metallschränke zu finden.[paragraph break]Eine Tür befindet sich im Norden."

The printed name is "Zimmer 127".

Auf dem Bett hier findet der Spieler einen weiteren Patienten:

The bed-127 is a neuter scenery enterable supporter in the fourth-room. Understand "alt" and "Bett" and "Liege [f]" and "Sperrholz [n]" and "kante [f]" and "kanten [p]" as the bed-127. The printed name is "Bett[-s]". The description is "Das Bett ist ein dünne Liege aus Sperrholz."

Instead of pushing or pulling the bed-127:

say "Du schiebst das Bett etwas hin und her."

Instead of pushing the bed-127 to north:

say "Es passt nicht durch die Tür."

Instead of entering the bed-127 when the helpless man is on the bed-127:
say "Es hat nur einen Platz auf dem Bett."

The helpless man is a man on the bed-127. Understand "patient/mann" as the helpless man. The printed name is "Patient[-en]". The description is "Der Mann ist ohne Bewusstsein. Ein Schlauch verbindet ihn mit einem Infusionsständer."

Ein paar Aktionen werden noch abgefangen:

Instead of waking the helpless man:
say "Er reagiert nicht."

Instead of taking the helpless man:
say "Er ist zu schwer für dich."

Instead of pushing or pulling the helpless man:
say "Du schiebst und drückst, bekommst ihn aber nicht von der Liege."

Instead of opening the helpless man:
say "Du hast deine Chirurgie-Ausrüstung nicht dabei."

Nun kommt aber der Clue: der Spieler läuft Gefahr, bei seinen Eskapaden entdeckt zu werden:

Danger is a thing. Danger is either obvious or over. Danger is over.

Die Gefahr wird "obvious" wenn der Spieler den Raum betritt:

After going to fourth-room when fourth-room is unvisited:
try looking;
now danger is obvious

Try looking muss hier extra ausgeführt werden, da ein "After going to [room]" dies unterbindet.

Der Schrank hat auch eine besondere Bedeutung:

The cabinet-127 is a male enterable scenery closed and openable container in the fourth-room. Understand "rostig" and "Schrank" and "Schranktür [f]" and "Metallschrank" and "Beulen [p]" and "Rost [m]" and "Roststellen [p]" as the cabinet-127. The printed name is "Schrank[-s]". The description is "Der Schrank ist alt und verbeult. [if closed]Er ist im Moment geschlossen[otherwise]Er ist geöffnet[end if]."

Wenn er betreten wird, ist die Gefahr vorüber.

After entering the cabinet-127 when danger is obvious:
say "Schnell verschwindest du im Schrank und ziehst die Schranktür hinter dir zu. Jemand kommt herein. Es poltert. Mit einem schnappenden Geräusch schließt sich die Zimmertür. Du wartest noch ein paar Herzschläge lang. Dann öffnest du die Schranktür wieder.[paragraph break]Dein Blick fällt auf eine kleine Handlampe im Schrank.";
now danger is over.

Die Ausstattung überspringen wir, die könnt ihr nach Belieben abtippen oder nicht, die Lampe ist aber wichtig:

The little lamp is a female device in the cabinet-127. Understand "Lampe" or "Leuchte" or "handlampe" as the little lamp. The printed name is "klein[^] Handlampe".

After switching on the little lamp:
say "Du schaltest die Handlampe an - sie leuchtet schwach auf.";
now little lamp is lit.

After switching off the little lamp:
say "Du schaltest die Handlampe wieder aus.";
now little lamp is not lit.

Hinweis:

Die Lampe bekommt nach dem Einschalten das Attribut "lit". Das bedeutet, sich ist eine Lichtquelle. Inform 7 kennt das Attribut "dark" für Räume, diese Räume werden dann so behandelt, dass der Spieler dort keine Raumbeschreibung bekommt und nur über sein Inventar verfügen kann. Trägt er aber eine Lichtquelle bei sich, kann er wieder agieren wie gewohnt.

Dazu gehört die Scene "Hiding", denn der Spieler soll ja dazu gebracht werden, sich zu verstecken:

Section - Hiding

Hiding is a scene.
Hiding begins when the danger is obvious.

Instead of going north from fourth-room when danger is obvious:
say "Sicher keine gute Idee, jetzt auf den Flur zu gehen."

Eine "every turn" rule regelt die Textausgabe während "hiding" abläuft:

Every turn during hiding:
say "[One of]Draußen sind plötzlich leise Schritte zu hören.[or]Die Schritte werden langsam lauter.[or]Die Schritte verstummen jetzt, direkt vor der Tür.[or]Du hörst leise Stimmen![or]Du hörst wenige Satzketzen. 'Lass uns keine Zeit verlieren ... Ausruhen kannst du später ... Erst noch die beiden, dann ... Ich bringe ihn gleich raus.'[paragraph break]Die Stimmen sind verstummt.[or]Schweißperlen auf deiner Stirn, gleich kommt jemand herein.[or]Schnell, gleich kommt jemand herein! [stopping]"

Das Ende:

Hiding ends when the danger is over.

When hiding ends:
now the fourth-room-door is open;
remove helpless man from play;
remove pole from play;
change the description of the note to "Eine Patientenliste, auf der verschiedene Namen zu lesen sind. Der Inhalt hat sich geändert. Alle Namen bis auf deinen sind durchgestrichen."

Instead of listening during hiding:
say "[One of]Du hörst Stimmen![or]Geräusche vor der Tür!
[or]Jemand ist da draußen![cycling]"

So, wir sind jetzt hier fertig. Der Patient ist mitgenommen worden, die Liste an der Tür des Stationszimmers hat sich geändert, und der Spieler ist noch einmal davongekommen.

Jetzt kann er mit der Lampe den dunklen Chorraum erforschen:

Section - The Cathedral

The cathedral-door is a female locked scenery door, east from corridor-east, west from cathedral. Understand "schwer" and "Eichentuer/Tuer" as the cathedral-door. The printed name is "Eichentür". The description is "Im Gegensatz zu den anderen Türen sieht diese sehr alt aus. Schwer und massiv und mit zahlreichen Ornamenten verziert."

The carvings are plural-named part of the cathedral-door. Understand "Ornamente" or "Verzierungen" or "Figuren" or "Muender" or "Augen" or "Haende" or "Darstellungen" as the carvings. The printed name is "Ornamente". the description is "Verschiedene Figuren, die auf den ersten Blick an Darstellungen in alten Kirchen erinnern. Auf den zweiten Blick erkennst du offenstehende Münder, Hände, die schützend über die Köpfe gehalten werden und vor Schreck geweitete Augen."

The cathedral is a dark room, in the indoors. "Im Schein der Lampe wird der Chorraum einer alten Kathedrale sichtbar. Ein aus Stein errichtetes Monument, alt und erhaben. Im Halbrund des Chors ruht der Altar, umgeben von altem Chorgestühl.[paragraph break]Der einzige Ausgang führt nach Westen, zurück auf den Flur."

The printed name is "Kathedrale, Presbyterium"

Das Chorgestühl hat eine Besonderheit, nämlich einen "lockeren" Stuhl, den man mitnehmen kann:

The choir is a neuter enterable scenery supporter in the cathedral. Understand "Gestuehl" or "Chorgestuehl" or "Stuehle [p]" or "Sitze [p]" or "Stuhl [m]" as the choir. the printed name is "Chorgestühl[-s]". the description is "Sie sind aus Eichenholz und sehen sehr alt aus.[if single-seat is not in indoors] Einer der Stühle wackelt.[end if]"

Instead of taking the choir:

if single-seat is in indoors:

say "Die anderen sind alle fest." instead;

otherwise:

say "Du nimmst den lockeren Stuhl mit.";

move single-seat to player.

The single-seat is a male enterable portable supporter. Understand "Stuhl" or "Chorstuhl" as the single-seat. The printed name is "Chorstuhl[-s]". The description is "Einer der Stühle aus dem Chorraum."

Der Stuhl kann aber auch wieder zurückgestellt werden:

Instead of dropping the single-seat when location is the cathedral:

say "Du stellst den Stuhl wieder in die Reihe.";

remove single-seat from play.

Und zu guter Letzt, der Altar mit der Chronik darauf:

The Altar is a male scenery supporter in the cathedral. Understand "Altar" or "Stein" as the altar. The printed name is "Altar[-s]". The description is "Kalter, massiver Stein."

The chronicles is a female thing on the altar. Understand "uralt" and "Buch" and "Chronik [f]" and "Chroniken [p]" as the chronicles. The printed name is "Chronik". The description is "Die Chronik sieht uralt aus. Vorsichtig blätterst du darin und stellst fest, dass verschiedene Seiten eingemerkt sind."

Die Idee ist hier, gezielt einzelne Seiten nachzuschlagen, mit wichtigen Ereignissen rund um die Kathedrale, äh das Krankenhaus:

The sites are a plural-named part of the chronicles. Understand "eingemerkt" and "Seiten/Merker/Einmerker" as the sites. The printed name is "eingemerkt[^] Seiten". The description is "Folgende Jahreszahlen sind eingemerkt: 1882, 1923, 1978, 1984."

"Schlag 1882 in chronik nach" soll ebenso funktionieren wie "betrachte 1882". Daher erschaffen wir 4 fake-objekte:

The very old entry is part of the sites. Understand "1882" as the very old entry.

Instead of doing something other than examining to the very old entry:
say "Es ist nur ein Einmerker."

Instead of examining the very old entry:
try consulting the chronicles about "1882".

The old-entry is part of the sites. Understand "1923" as the old-entry.

Instead of doing something other than examining to the old-entry:
say "Es ist nur ein Einmerker."

Instead of examining the old-entry:
try consulting the chronicles about "1923".

The new-entry is part of the sites. Understand "1978" as the new-entry.

Instead of doing something other than examining to the new-entry:
say "Es ist nur ein Einmerker."

Instead of examining the new-entry:
try consulting the chronicles about "1978".

The very new entry is part of the sites. Understand "1984" as the very new entry.

Instead of doing something other than examining to the very new entry:
say "Es ist nur ein Einmerker."

Instead of examining the very new entry:
try consulting the chronicles about "1984".

Und der Part mit dem Nachschlagen:

Does the player mean consulting the chronicles about: it is very likely.

Instead of consulting the sites about:
try consulting the chronicles about.

instead of consulting the chronicles about:
say "Du suchst eine Weile, findest darüber aber nichts in der Chronik."

instead of consulting the chronicles about "1882":
say "Im Jahre 1882 wurde der Bau der Kathedrale abgeschlossen. Bei der feierlichen Eröffnung gab es einen bedauerlichen Todesfall: Der Bischof wurde von einem herabfallenden Dachziegel getötet. Seitdem war man überzeugt, dass die Kathedrale unter keinem guten Stern steht."

instead of consulting the chronicles about "1923":
say "Im Jahre 1923 brach eine schwere Diphtherie-Epidemie aus. Man benötigte dringend Räume für die Kranken und so baute man die Kathedrale um. Trennwände wurden eingezogen und teilten den Kirchenraum in viele kleine, zum Großteil dunkle

und unbelüftete Zimmer.[paragraph break]Wo es die schweren Außenmauern zuließen, brach man kleine Lüftungsöffnungen in das Mauerwerk. Abgeschottet von der Außenwelt starben hier viele Menschen an der schweren Krankheit."

instead of consulting the chronicles about "1978":

say "Im Jahre 1978 wurde das Krankenhaus endgültig geschlossen. Die Einrichtung war veraltet und das Krankenhaus konnte mit moderner Medizin nicht mehr schritthalten. Bis auf Mary Higgins und John Rezznar verließen alle Menschen das Krankenhaus. Wenige Jahre später wurde die Kathedrale der Treffpunkt einer Sekte namens 'Der Orden der Hundert Seelen'. Die Gründerin wird einige Male unter dem Namen 'Mariel' erwähnt.[paragraph break]Die Zahl der Anhänger nahm im Lauf der Zeit immer mehr zu."

instead of consulting the chronicles about "1984":

say "In jüngerer Zeit sind immer öfter Zeitungsartikel in die Chronik eingeklebt. Schilderungen von Einheimischen, die angeblich Menschen mit blasser und blättriger Haut in der Gegend gesehen haben wollen. Predigten des hiesigen Pastors, der vor dem 'Orden der Hundert Seelen' warnt.[paragraph break]Berichte über vermisste Personen, die in dem Wäldchen um die Kathedrale unterwegs waren."

Jetzt sind wir schon fertig und wir machen mit dem dritten Raum weiter:

Section - The third room

The third-room-door is a female locked scenery door, north from corridor-west, south from third-room. Understand "weiss" and "tuer" and "zu" and "zimmer" and "128" as the third-room-door. The printed name is "Tür". The description is "Eine weiße Tür, wie die anderen auch."

The third-room is a room, in the indoors. "Dieser Raum ist sehr klein und hat überhaupt kein Fenster. Nur ein hohes Regal lehnt an der Wand.[paragraph break]Eine Tür befindet sich im Süden."

The printed name is "Zimmer 128".

Hier ist ein Regal, auf dessen oberstem Fach ein Schraubendreher liegt, unerreichbar für den Spieler:

The shelves is a neuter scenery supporter in the third-room. Understand "Regal" or "regalboden [m]" or "regalboeden [p]" as the shelves. The printed name is "Regal[-s]". The description is "Das Regal sieht ziemlich heruntergekommen aus."

Instead of inserting something into the shelves:

try putting the noun on the shelves.

Instead of climbing the shelves:

say "Du steigst auf einen Regalboden und es knackt laut. Du lässt davon ab, das Regal wird dein Gewicht nicht tragen."

Instead of entering the shelves:

say "Du steigst auf einen Regalboden und es knackt laut. Du lässt es besser bleiben, das Regal wird dein Gewicht nicht tragen."

The screwdriver is a male faraway thing on the shelves. Understand "Schraubenschlüssel" or "Schraubendreher" or "Dreher" or

"Schraubenzieher" or "Zieher" as the screwdriver. The printed name is "Schraubendreher". The description is "Ein glänzender Schraubendreher."

The screwdriver can be unreachable. The screwdriver is unreachable.

Es sei denn, der Spieler stellt sich auf den Stuhl aus dem Chorraum:

Definition: the screwdriver is faraway if the screwdriver is unreachable and the player is not on the single-seat.

Report taking the screwdriver when the screwdriver is unreachable:
now screwdriver is not unreachable;
say "Du streckst dich soweit wie möglich und erreichst den
Schraubendreher knapp mit den Fingerspitzen."

Das "leftover" brauchen wir später:

Definition: a thing is leftover if it is not the screwdriver.

Jetzt wird es komplizierter, falls der Spieler etwas ins Regal legt, soll der Schraubendreher in seiner ursprünglichen Position "im obersten Fach" beschrieben werden, der Rest soll natürlich weiter unten liegen:

Das Ergebnis soll etwa so aussehen:

Zimmer 128

Dieser Raum ist sehr klein und hat überhaupt kein Fenster. Nur ein hohes Regal lehnt an der Wand.

Eine Tür befindet sich im Süden.

Im obersten Fach des Regals liegt ein glänzender Schraubendreher. Darunter siehst du einen schweren Schlüssel.

>_

Der code dazu ist wie folgt:

For printing a locale paragraph about the shelves:
set the locale priority of the shelves to 0;
Let N be the number of things on the shelves;
if the screwdriver is on the shelves and screwdriver is faraway and N
is greater than 1:
say "Im obersten Fach liegt ein glänzender Schraubendreher.
Darunter siehst du [a list of leftover things on the
shelves with accusative].";
repeat with item running through things on the shelves:
now item is mentioned;
stop the action;
otherwise if screwdriver is on the shelves and screwdriver is faraway:
say "Im obersten Fach liegt ein glänzender
Schraubendreher.";
stop the action;
Repeat with item running through things on the shelves:
say "Auf dem Regal siehst du [a list of things on the shelves
with accusative]."

Listen können in den vier relevanten Fällen ausgegeben werden ("with nominative, genitive, accusative, dative").

Bis auf Kleinigkeiten sind wir hier fertig, und lassen den Spieler jetzt endgültig in die Falle laufen:

Section - The last try

Definition: A thing is unimportant if it is not the screwdriver.

After going to hospital when player is holder of the screwdriver:

say "Als du den Raum betrittst, stehst du plötzlich vor der Krankenschwester. 'Aha, hier sind Sie', ruft Mariel mit vor Zorn bebenden Lippen, 'ich sagte Ihnen doch, Sie sollen liegen bleiben!'[paragraph break]Mit einer Stärke, die du einer Krankenschwester nie zugetraut hättest, befördert sie dich zurück ins Bett und entreißt dir [the list of unimportant things held by player with accusative].[paragraph break]Den Schraubendreher hat sie dabei wohl übersehen. 'Bleiben sie liegen', sagt sie mit einem scharfen Unterton, 'ich bereite eine Infusion für sie vor, damit sie wieder zu Kräften kommen.'[paragraph break]Sie verschließt die Tür von außen und ihre Schritte entfernen sich.";

move player to bed;

repeat with item running through unimportant things held by player:

remove item from play;

now the white door is closed;

now the white door is locked;

now the player is next-on-line.

Jetzt geht es Schlag auf Schlag, denn Mariel bereitet den "Cocktail" vor:

Section - Scene preparing

Preparing is a scene. Preparing begins when the player is next-on-line.

When preparing begins:

move nurse to corridor.

Instead of going south from hospital when preparing is happening:

say "Das willst du jetzt auf keinen Fall!"

Every turn during preparing:

say "[One of]Türen werden auf- und zugeschlagen.[or]Stille. Kein Laut ist zu hören.[or]Stille. Kein Laut ist zu hören.[or]Stille. Kein Laut ist zu hören.[or]Wieder Türgeräusche.[or]Etwas schweres wird über den Boden geschleift.[or]Schritte kommen näher.[or]Die Schritte enden vor der Tür.[or]Lautes Schnaufen.[or]Die Atemgeräusche werden allmählich ruhiger.[or]Eine Hand legt sich auf die Klinke.[begin infusing][stopping]"

Hinweis: Praktischerweise kann man mit solchen "hooks" (ist zugegebenermaßen etwas "hacky") Auslöser für Aktionen oder Anweisungsblöcke einfach innerhalb der say-Phrasen einbauen. Dies ist trotzdem ein elegantes System, um in den Spielverlauf einzugreifen. Hier nutzen wir [begin infusing] als Textersetzung, die aber keinen Text ausgibt, sondern ein Attribut setzt: NOW THE PLAYER IS DOOMED.

To say begin infusing:

now the player is doomed.

Preparing ends sadly when infusing begins.

Preparing ends happy when player is in backyard.

Der Spieler hat jetzt wenige Züge, um das Offensichtliche zu tun: Das Bett loszuschrauben, zu drehen, den Schrank zu öffnen, die Brechstange zu nehmen, das Fenster aufzubrechen und abzuhaufen.

Beginnen wir mit dem Schrauben:

Wir definieren eine neue Aktion für das Lösen der Schrauben:

Section - Unscrewing it with

Unscrewing it with is an action applying to two things.

Understand "loes [something] mit [dativ] [something]" as unscrewing it with.
Understand "locker [something] mit [dativ] [something]" as unscrewing it with.

Understand "dreh [something] mit [dativ] [something] auf/locker" as unscrewing it with.

Understand "loes mit [dativ] [something] [something]" as unscrewing it with (with nouns reversed).

Understand "locker mit [dativ] [something] [something]" as unscrewing it with (with nouns reversed).

Understand "dreh mit [dativ] [something] [something] auf/locker" as unscrewing it with (with nouns reversed).

Soweit zum "Verständnis", bzw. dem Wörtbucheintrag. Dann kommt die Check-rule:

Check unscrewing it with:

if the second noun is not the screwdriver:

say "[Der second noun] eignet sich dafür nicht." instead;

if noun is not a screwable:

say "[Der noun] [if noun is plural-named]können[else]kann[end if] nicht gelockert werden." instead;

if noun is loose:

say "[Der noun] [ist] schon locker." instead;

Und die Carry-out rule:

Carry out unscrewing it with:

now the noun is loose.

Und die Report-rule:

Report unscrewing it with:

say "Du lockerst [den noun] mit [dem second noun]."

Und eine für das Festziehen:

Section - Screwing it with

Screwing it with is an action applying to two things.

Understand "befestig [something] mit [dativ] [something]" as screwing it with.

Understand "zieh [something] mit [dativ] [something] an/fest" as screwing it with.

Understand "dreh [something] mit [dativ] [something] fest" as screwing it with.

Understand "befestig mit [dativ] [something] [something]" as screwing it with (with nouns reversed).

Understand "zieh mit [dativ] [something] [something] an/fest" as screwing it with (with nouns reversed).

Understand "dreh mit [dativ] [something] [something] fest" as screwing it with (with nouns reversed).

Check screwing it with:

if the second noun is not the screwdriver:

say "[Der second noun] eignet sich dafür nicht." instead;

if noun is not a screwable:

say "[Der noun] [if noun is plural-named]können[else]kann[end if] nicht festgezogen werden." instead;

if noun is tight:

say "[Der noun] [ist] schon festgezogen." instead;

Carry out screwing it with:

now the noun is tight.

Report screwing it with:

say "Du ziehst [den noun] mit [dem second noun] fest."

Somit sind wir erst mal durch, das Attribut loose/tight wird entsprechend gesetzt. In der "Praxis" gibt es aber noch viele Möglichkeiten, das Lösen der Schrauben zu formulieren, z.B.

>öffne Schrauben mit Schraubendreher

Also werden wir weitere Aktionen auf das Lösen/Festziehen umleiten.

Section - Screwing the screws

Instead of opening the screws when player is holder of the screwdriver:

try unscrewing the screws with the screwdriver.

Instead of closing the screws when player is holder of the screwdriver:

try screwing the screws with the screwdriver.

Instead of closing the screws with the screwdriver:

try screwing the screws with the screwdriver instead.

Instead of unlocking the screws with the screwdriver:

try unscrewing the screws with the screwdriver.

Instead of turning the screws when screws are tight:

if player is holder of screwdriver:

try unscrewing the screws with the screwdriver;

otherwise:

say "Dir fehlt das geeignete Werkzeug."

Instead of turning the screws when screws are loose:

if player is holder of screwdriver:

try screwing the screws with the screwdriver;

otherwise:

say "Dir fehlt das geeignete Werkzeug."

Instead of pulling the screws when screws are tight:

if player is holder of screwdriver:

try unscrewing the screws with the screwdriver;

otherwise:

say "Dir fehlt das geeignete Werkzeug."

Instead of pulling the screws when screws are loose:

if player is holder of screwdriver:

try screwing the screws with the screwdriver;

otherwise:
say "Dir fehlt das geeignete Werkzeug."

Natürlich ist das Festziehen nicht möglich, wenn das Bett bereits gedreht wurde.

Instead of screwing the screws with the screwdriver when the bed is lengthways:
say "Das ist in dieser Position des Betts nicht möglich."

Wenn der Spieler zu lange fackelt, ist jetzt aber wirklich Schluss:

Section - Scene infusing

Infusing is a scene. Infusing begins when the player is doomed.

When infusing begins:

if location is not hospital:
if bed is lengthways:
say "Aus deinem Zimmer kommen seltsame Geräusche.";
otherwise:
say "[if blocking has happened][Der nurse][otherwise] [A nurse][end if] steckt ihren Kopf durch das Loch in der Decke. 'Aha, da sind sie' sagt sie und zerrt dich an deinen Füßen mit eiserner Kraft zurück nach unten auf das Bett. [paragraph break]";
kill the player;
stop the action;
if the white door is locked:
say "Mit einem Klicken wird die Tür wieder aufgeschlossen. [paragraph break]";
if the bed is crossways:
if blocking has happened:
say "[The nurse] betritt die Kammer wieder.[paragraph break]";
otherwise:
say "[A nurse] öffnet die Tür und betritt die Kammer. [paragraph break]";
kill the player;
otherwise:
say "Jemand versucht, die Tür zu öffnen, aber das Bett blockiert die Tür."

Every turn during infusing:

if the bed is crossways:
if blocking has happened:
say "[The nurse] betritt die Kammer wieder.[paragraph break]";
otherwise:
say "[A nurse] öffnet die Tür und betritt die Kammer. [paragraph break]";
kill the player;
otherwise:
if location is hospital:
say "[one of]Jemand rüttelt heftig an der Tür.[or]Du hörst ein lautes Fluchen von draußen.[or]Ein ungeduldiges Klopfen hallt durch die Kammer. [or]Die Tür erbebt unter heftigen Schlägen.[at random]".

Hinweis: Falls der Spieler versucht hat, in den Schacht zu steigen, greift obige Phrase: if location is not hospital, da "location" immer der aktuelle Ort ist, an dem sich der Spieler aufhält.

Oben haben wir noch eine neue Phrase definiert: KILL THE PLAYER.

Also, kurz und schmerzlos:

Section - To kill the player

To kill the player:

say "Sie hält eine Spritze mit einer grünen Flüssigkeit in der Hand. Du willst dich wehren, aber eisige Finger halten dich mit einem unmenschlich starken Griff fest. Du spürst einen brennenden Schmerz, als sich die grüne Flüssigkeit langsam in deinen Venen ausbreitet.";
end the story finally saying "Du bist verloren."

Die im Abschnitt "Actions" folgenden Sections Despairing / Cringing / Dancing / Praying / Crying / Spewing / Counting / Tilting / Commenting erweitern den Befehlswortschatz, ohne direkten Einfluss auf das Gameplay. Daher überspringen wir diese Abschnitte und wenden uns dem Außenbereich zu.

Wir erinnern uns: Der Spieler ist jetzt Mariel dreimal entkommen, hat einiges über die Kathedrale, äh das Krankenhaus gelernt und sollte jetzt eigentlich zweifelsfrei bemüht sein, diesem unwirtlichen Ort schnellstmöglich zu entkommen.

3.2 Außenbereich

Der Außenbereich legt sich hufeisenförmig um die Kathedrale, wobei nur eines der Enden wirklich nach draußen führt, das andere endet in einer Sackgasse.

Ach ja, da war ja noch der Typ, den wir vom Fenster aus gesehen haben, und die ominöse Mülltonne.

Section - The caretaker

The caretaker is a man in the dumping ground. The printed name of the caretaker is "Koloss[-es]". understand "Riese" or "Koloss" or "Mann" or "Ungeheuer [n]" or "Huene" or "knochen [p]" as the caretaker. the description of the caretaker is "Der Mann ist ein Hüne von gut zwei Meter achtzig Größe. Breit und muskulös, wie ein Berg steht er vor dir. Große Teile seiner Haut fehlen. Und dann sein Gesicht: offenes Fleisch, Knochen und leere, weiße, starre Augen."

Distance is a number that varies. Distance is 3.

After going somewhere when the location is in outdoors:
change distance to 3;
if hunting is happening:
move the caretaker to the location;
continue the action.

Wir verwenden einen einfachen Zähler namens "distance". Der prüft, ob uns der Typ erwischt hat oder nicht. Durch die Flucht nach hinten gewinnen wir wieder etwas mehr Abstand.

Das Verhalten des Typen regelt die Szene hunting:

Section - Hunting

Hunting is a recurring scene.

Hunting begins when the caretaker can see the player and the player is holder of the flesh and location is dumping ground and distance is 3.

Rule for writing a paragraph about the caretaker:
now the caretaker is mentioned.

When hunting begins:
now the player knows about the caretaker.

Hunting verursacht viel Text, abhängig von der momentanen location:

Every turn during hunting:
decrease distance by 1;
if location is the dumping ground:
if distance is 2:
say "Der Mann bemerkt etwas und dreht langsam den Kopf. Er rümpft seine Nase, als ob er etwas wittert und beginnt langsam, mit wankenden Schritten auf dich zuzugehen.";
if distance is 1:
say "Der Mann wankt weiter langsam aber stetig auf dich zu, Panik ergreift dich. Sein Blick ist leer und kalt. Noch wenige Schritte und er erreicht dich.";
if location is the right path:
if distance is 2:
say "Der Mann wankt weiter hinter dir her. Er bewegt sich langsam den Weg entlang, direkt in deine Richtung. Wieder dreht er den Kopf hin und her und rümpft seine Nase, wie ein Tier, das eine Witterung aufnimmt.";
if distance is 1:
say "Der Mann kommt dir bedrohlich nahe, deine Beine beginnen zu zittern. Die Augen des Mannes sind leer und weiß. Noch wenige Schritte und er erreicht dich.";
if location is the backyard:
if distance is 2:
say "Der Mann hält kurz inne - er gibt ein lautes Gurren von sich. Wieder dreht der Mann sich in deine Richtung. Es scheint kein Entkommen zu geben. Langsam, unablässig kommt er mit wankenden Schritten auf dich zu.";
if distance is 1:
say "Wiederum trennen dich nur noch wenige Schritte von dem Mann. Deine Beine fühlen sich müde und schwer an, als wollen sie ihren Dienst versagen.";
if location is the left path:
if distance is 2:
say "Der Mann wankt weiter hinter dir her und folgt langsam dem Weg. Wieder dreht er den Kopf hin und her und hält einen Moment inne - dann bewegt er sich zielstrebig weiter auf dich zu.";
if distance is 1:
say "Wieder kommt der Mann dir bedrohlich nahe. Dein Herz pocht wie wild. Die Wangen des Mannes sind eingefallen und er ist blaß im Gesicht. Noch wenige Schritte und er erreicht dich.";

```

if location is the dying forest:
    if distance is 2:
        say "Raschelnd bewegt sich der Mann über den
            laubbedeckten Boden. Mit wankenden Schritten
            kommt er auf dich zu, sein Gesicht ist vor Wut
            verzerrt.";
    if distance is 1:
        say "Der Mann scheint übermenschliche Kräfte zu
            haben, Panik ergeißt dich. Noch wenige Schritte
            und er erreicht dich.";
if location is the clearing:
    if distance is 2:
        say "Jetzt gibt es keinen Ausweg mehr. Der Koloss
            kommt unaufhaltsam auf dich zu und diese
            Lichtung ist von undurchdringlichen dicht
            stehenden Bäumen umgeben - kein Weg vor
            und keiner zurück.";
    if distance is 1:
        if the flesh is not on the wooden trapdoor:
            say "Der Mann wankt weiter auf dich zu und
                umrundet den Holzdeckel. In Panik
                hältst du deine Hände schützend vor
                deine Augen. Er hat dich fast erreicht.";
        otherwise:
            say "Der Mann wankt weiter auf dich zu. Jetzt
                steigt er auf den Holzdeckel und bleibt
                einen Moment lang stehen, greift sich
                das Stück Fleisch und macht sich über
                das Fleisch her.";
        move the caretaker to the wooden trapdoor;

```

Instead of taking the flesh when caretaker is on the pit:
 say "Da fasst du besser nicht hin."

Hier steckt bereits des Rätsels Lösung drin: Der Mann muss auf den Deckel auf der Lichtung gelockt werden und dann ... aber erstmal brauchen wir noch den entsprechenden Ausgangs-"room":

Section - Dumping ground

The dumping ground is a male room, south from the right path, in the outdoors. "Der Eingangshof ist genauso verdreckt wie der Hinterhof. Am Rand des Geländes steht eine große Mülltonne. Auf dieser Seite ragen zwei mächtige schwarze Türme in den Himmel, dazwischen die immens große Westrosette der Kathedrale.[paragraph break]Der Weg in die Freiheit ist das offene Eingangstor im Süden[if the caretaker is in the dumping ground], wenn nicht ein Koloss von einem Mann direkt davor stehen würde[end if]. [paragraph break]Im Norden liegt der verdreckte Pfad zum Hinterhof."

The printed name of the dumping ground is "Eingangshof".

Und die Mülltonne:

The dustbin is a female scenery closed and openable container in the dumping ground. The printed name of the dustbin is "Mülltonne". Understand "riesig" and "immens" and "gross" and "Muell" and "Tonne" and "Muelltonne" as the dustbin. the description of the dustbin is "Eine immens große Mülltonne steht hier am Zaun."

Does the player mean doing something to the flesh: it is very likely.

The flesh is a neuter thing in the dummy room. Understand "Fleisch" or "Stueck" or "Fleischstueck" or "Stueck Fleisch" as the flesh. The printed name is "Fleisch". the description is "Roh und blutig, wie die anderen auch."

The flesh-pieces are a plural-named thing in the dustbin. Understand "Fleischstuecke" or "Stuecke" or "Fleischstueck [n]" or "Fleisch [n]" as the flesh-pieces. The printed name is "Fleischstücke". The description is "Fleischstücke, roh und blutig."

Wir verwenden das gleiche Prinzip wie beim Chorstuhl.

Instead of taking the flesh-pieces:

if flesh is in outdoors:

say "Du hast schon eines genommen, das reicht erstmal."

instead;

otherwise:

say "Du nimmst eines der Fleischstücke heraus.";

move flesh to player.

Report dropping the flesh:

say "Du legst das Stück Fleisch vor dir ab." instead.

Der Koloss ist ganz wild auf das Fleisch. Während hunting abläuft werden die Richtungen am Koloss vorbei geblockt, also no way!

Section - Blocking directions during hunting

Instead of going south from dumping ground when hunting is happening:

say "Du siehst keinen Weg, wie du an dem Koloss von Mann vorbei kommen kannst.".

Instead of going south from right path when hunting is happening:

say "Der Pfad ist viel zu eng, um an dem Mann vorbeizukommen.".

Instead of going east from backyard when hunting is happening:

say "Du siehst keinen Weg, wie du an dem Koloss von Mann vorbei kommen kannst.".

Instead of going east from left path when hunting is happening:

say "Der Pfad ist viel zu eng, um an dem Mann vorbeizukommen.".

Instead of going north from dying forest when hunting is happening:

say "Du siehst keinen Weg, wie du an dem Koloss von Mann vorbei kommen kannst.".

Instead of going north from clearing when hunting is happening:

say "Du siehst keinen Weg, wie du an dem Koloss von Mann vorbei kommen kannst.".

Die Enden der Szene:

Hunting ends early when player is not holder of the flesh and flesh is not on the wooden trapdoor and flesh is not in the pit.

Hunting ends pretty cruel when player is not holder of the flesh and flesh is not on the wooden trapdoor and flesh is in the pit.

Hunting ends unsuccessfully when flesh is on wooden trapdoor and leaves are not on the wooden trapdoor and distance is 2.

Hunting ends sadly when distance is 0.

Hunting ends happily when the caretaker is in the pit.

Und was dann so passiert:

When hunting ends happily:

say "Du atmest tief durch. Das war wirklich knapp."
move flesh to dummy room.

When hunting ends sadly:

say "Der Mann packt dich am Arm. Er schleift dich mit sich, langsam, aber unaufhaltsam. Du wehrst dich, du schlägst wie wild mit den Fäusten auf ihn ein - vergeblich. Du siehst noch, wie er sich die Stufen zum Vordereingang des schwarzen Gebäudes hinaufschleppt, als dich ein harter Schlag ins Gesicht trifft und du besinnungslos wirst."
end the story finally saying "Du bist verloren."

When hunting ends pretty cruel:

say "Das Fleisch ist weg und der Mann wendet sich stattdessen dir zu. Er schleift dich mit sich, langsam, aber unaufhaltsam. Du wehrst dich, du schlägst wie wild mit den Fäusten auf ihn ein - vergeblich. Du siehst noch, wie er sich die Stufen zum Vordereingang des schwarzen Gebäudes hinaufschleppt, als dich ein harter Schlag ins Gesicht trifft und du besinnungslos wirst."
end the story finally saying "Du bist verloren."

When hunting ends early:

say "Der Koloss stürzt sich auf das Fleisch und verzehrt es in einem Atemzug. Du bleibst starr stehen. Der Koloss rümpft noch einmal die Nase, bemerkt aber nichts mehr und tritt langsam zum Tor zurück."
now distance is 3;
move flesh to dummy room;
move caretaker to dumping ground.

When hunting ends unsuccessfully:

say "Der Mann wankt weiter auf dich zu. Gerade als er den Holzdeckel betreten will, bemerkt er ihn und angelt sich das Stück Fleisch geschickt mit einem abgebrochenen Ast.
[paragraph break]Der Koloss verschlingt das Fleisch in einem Atemzug. Du bleibst starr stehen. Der Koloss rümpft noch einmal die Nase, bemerkt aber nichts mehr und tritt langsam zum Tor zurück."
now distance is 3;
move flesh to dummy room;
move caretaker to dumping ground.

Ein paar der Aktionen, die zum frühen Ende führen (Fütter den Mann mit dem Fleisch):

Instead of throwing the flesh at the caretaker:

say "Du wirfst das Stück Fleisch dem Koloss vor die Füße."
move flesh to location.

Instead of giving the flesh to the caretaker:

try throwing the flesh at the caretaker.

Instead of showing the flesh to the caretaker:

try throwing the flesh at the caretaker.

Instead of inserting the flesh into the caretaker:
try throwing the flesh at the caretaker.

So jetzt sind wir fast durch. Wie wir oben sehen, wird das optimale Ende nur erreicht, wenn der Spieler das Fleisch auf den Deckel in der Lichtung legt, der aber vorher durch Blätter getarnt werden muss.

Bitte, zuerst die Blätter:

The dying forest is a male room, south from the left path, in the outdoors.
"Unheimlich und bizarr zeichnen sich die Umrisse der kahlen und knorrigen Äste gegen den Nachthimmel ab. Obwohl es Hochsommer ist, wächst hier kein Grün. Die Bäume sind abgestorben und der Boden ist über und über mit welken Blättern bedeckt.[paragraph break]Im Norden liegt der Pfad zum Hinterhof, im Süden eine kleine Lichtung."

The printed name of the dying forest is "tot[^] Wald".

The ground-leaves are a female plural-named scenery in the dying forest. The printed name of the ground-leaves is "Blätter". Understand "Blaetter" or "Boden [m]" or "Laub [n]" as the ground-leaves. The description of the ground-leaves is "Die grauen Blätter bedecken den gesamten Boden hier."

Instead of taking the ground-leaves when the leaves are in the pit:
if the leaves are in the pit:
say "Du nimmst eine Handvoll Blätter mit.";
move the leaves to the player;
otherwise:
say "Du hast bereits eine Handvoll davon mitgenommen."

Instead of rubbing the ground-leaves:
say "Sie zerbröseln."

Instead of rubbing the leaves:
say "Sie zerbröseln in der Hand.";
move leaves to the pit.

Does the player mean doing something to the leaves: it is very likely.

Instead of dropping the leaves when location is dying forest:
say "Du wirfst die Blätter wieder auf den Boden.";
move the leaves to the pit.

Dann der Deckel:

Section - Clearing

The clearing is a female room, south from the dying forest, in the outdoors.
"Diese Lichtung ist ringsum mit kahlen, grauen und verdorrten Bäumen umgeben. In der Mitte der Lichtung befindet sich ein runder Holzdeckel, etwa zwei Meter im Durchmesser, umgeben von einem Rand aus gemauertem Stein.[paragraph break]Im Norden liegt der tote Wald, die einzige Möglichkeit die Lichtung zu verlassen."

The printed name of the clearing is "Lichtung".

A trapdoor is a kind of supporter. A trapdoor is either open or closed. A trapdoor is always enterable.

The wooden trapdoor is a male scenery trapdoor in the clearing. It is closed. The printed name of the wooden trapdoor is "Holzdeckel[-s]". understand "Falltuer [f]" or "Holzdeckel" or "Deckel" or "Fuge [f]" as the wooden trapdoor.

Und der Hebel zum Öffnen des Deckels:

The lever is male fixed in place in the clearing. "Ein Hebel ragt neben dem Deckel aus dem Boden." The printed name of the lever is "Hebel[-s]". Understand "hebel" as the lever. The description of the lever is "Ein Hebel aus rostigem Metall.". The description of the wooden trapdoor is "[if closed]Der Holzdeckel verdeckt irgendetwas. In der Mitte teilt eine schmale Fuge den Deckel in zwei Hälften.[otherwise]Der Holzdeckel ist geöffnet und gibt einen schwarzen Schacht frei, der in die Tiefe führt.[end if] Seitlich davon befindet sich ein Hebel.".

The lever is either pushed or pulled. The lever is pushed.

Und der Schacht darunter:

The shaft is a male scenery. The shaft is in the pit. The printed name of the shaft is "schwarz[^] Schacht[-s]". Understand "schwarz" and "schacht" as the shaft. The description of the shaft is "Der Schacht ist schwarz und du kannst keinen Boden sehen. Schatten flattern. Irgendetwas scheint sich dort unten zu bewegen.".

Und eine "customized" rule für das Beschreiben des Deckels, furchtbarer code:

For printing a locale paragraph about the wooden trapdoor:
set the locale priority of the wooden trapdoor to 0;
if the leaves are on the wooden trapdoor and the crowbar is on the wooden trapdoor and the screwdriver is on the wooden trapdoor:
say "Der hölzerne Deckel in der Mitte der Lichtung, die Brechstange und der Schraubendreher sind unter einer Schicht Laub perfekt verborgen.";
now the leaves are mentioned;
now the crowbar is mentioned;
now screwdriver is mentioned;
stop the action;
if the leaves are on the wooden trapdoor and the crowbar is on the wooden trapdoor:
say "Der hölzerne Deckel in der Mitte der Lichtung und die Brechstange sind unter einer Schicht Laub perfekt verborgen.";
now the leaves are mentioned;
now the crowbar is mentioned;
stop the action;
if the leaves are on the wooden trapdoor and the screwdriver is on the wooden trapdoor:
say "Der hölzerne Deckel in der Mitte der Lichtung und der Schraubendreher sind unter einer Schicht Laub perfekt verborgen.";
now the leaves are mentioned;
now the screwdriver is mentioned;
stop the action;
if the leaves are on the wooden trapdoor:
say "Der hölzerne Deckel in der Mitte der Lichtung ist unter einer Schicht Laub perfekt verborgen.";
now the leaves are mentioned;
stop the action;

```

if the crowbar is on the wooden trapdoor:
    say "Auf dem hölzernen Deckel in der Mitte der Lichtung liegt
        die Brechstange.";
    now the crowbar is mentioned;
    stop the action;
if the screwdriver is on the wooden trapdoor:
    say "Auf dem hölzernen Deckel in der Mitte der Lichtung liegt
        der Schraubendreher.";
    now the screwdriver is mentioned;
    stop the action;
if the shaft is in the clearing:
    say "Ein schwarzer Schacht führt in die Tiefe.";
    stop the action.

```

Open and Close wird abgefangen, da dies ja der Hebel für uns erledigt:

```

Instead of opening the wooden trapdoor:
    if the wooden trapdoor is closed:
        say "Du rüttelst an dem hölzernen Deckel. Er bewegt sich
            keinen Millimeter.";
    otherwise:
        say "Der Deckel ist doch bereits geöffnet.".

Instead of closing the wooden trapdoor:
    if the wooden trapdoor is open:
        say "Du ziehst mit aller Kraft an dem hölzernen Deckel. Er
            bewegt sich keinen Millimeter.";
    otherwise:
        say "Der Deckel ist doch bereits geschlossen.".

```

Und die Regeln für das Drauflegen von den Blättern und dem Fleisch:

```

Before putting the leaves on the wooden trapdoor:
    if the wooden trapdoor is open:
        say "Die Blätter fallen in den offenen Schacht.";
        move the leaves to the pit;
        stop the action.

Before putting the flesh on the wooden trapdoor:
    if the wooden trapdoor is open:
        say "Das Fleisch fällt in den offenen Schacht.";
        move the flesh to the pit;
        stop the action.

Report putting the leaves on the wooden trapdoor:
    say "Du verbirgst den Holzdeckel vollständig unter einer
        Blätterschicht.";
    stop the action.

```

Das Bewegen des Hebels, das auch mitunter das Problem des Kolosses löst:

```

Instead of pulling or pushing the lever:
    if the lever is pushed:
        now the lever is pulled;
        now the wooden trapdoor is open;
        say "Mit einem quietschenden Geräusch bewegst du den
            Hebel in eine Richtung - der hölzerne Deckel springt
            wie eine Falltür nach unten auf und gibt einen tiefen
            Schacht frei.";
        move the shaft to the clearing;

```

```

if the player is on the wooden trapdoor and the caretaker is on
the wooden trapdoor:
    say "[line break]Ihr stürzt beide in die Tiefe, absolute
    Dunkelheit um dich herum. Du landest
    verhältnismäßig weich, aber deine anfängliche
    Erleichterung wandelt sich in pures Entsetzen,
    als du bemerkst, dass zahlreiche Arme nach dir
    greifen und dich weiter in die Tiefe ziehen.";
    end the story finally saying "Du bist verloren.";
if the player is on the wooden trapdoor:
    say "[line break]Du stürzt in die Tiefe, absolute
    Dunkelheit um dich herum. Du landest
    verhältnismäßig weich, aber deine anfängliche
    Erleichterung wandelt sich in pures Entsetzen,
    als du bemerkst, dass zahlreiche Arme nach dir
    greifen und dich weiter in die Tiefe ziehen.";
    end the story finally saying "Du bist verloren.";
if the caretaker is on the wooden trapdoor:
    say "[line break]Der Koloss schreit auf und stürzt in
    die Tiefe - kurz darauf folgt ein donnernder
    Aufprall. Du schaust nach unten, kannst aber in
    der Dunkelheit nichts erkennen.";
    move the caretaker to the pit;
    move flesh to dummy room;
Let L be the number of things on the wooden trapdoor;
if L is greater than 0:
    repeat with item running through things on the
    wooden trapdoor:
        if item is flesh:
            move item to dummy room;
        otherwise:
            move item to pit;

    stop the action;
if the lever is pulled:
    now the lever is pushed;
    now the wooden trapdoor is closed;
    move the shaft to the pit;
    say "Mit einem quietschenden Geräusch bewegst du den
    Hebel wieder zurück - der hölzerne Deckel schließt
    sich wieder."

```

So weit so gut. Wir erinnern uns: Der Spieler erkundet den Außenbereich, legt die Falle für den "Hausmeister" und kann jetzt ohne weitere Schwierigkeiten das Gelände verlassen. Doch er ist noch nicht in Sicherheit. Er kommt zu einer kleinen Landstraße, wo glücklicherweise ein Polizeiwagen direkt vor seiner Nase hält. Glücklicherweise? Nun ja, "Mariel" wäre nicht "Mariel", wenn da nicht ein Haken dabei wäre ...

3.3 Der Schluss

Section - Policeman

The policeman is a man in the police car. The printed name of the policeman is "Polizeibeamt[^]". Understood "Polizei" or "Polizeibeamter" or "Beamte" or "Beamter" or "Mann" or "Bulle" or "Cop" or "Polizist" or "kommissar" or "john" or "rezznar" or "john rezznar" or "albert" or "albert lee" as the policeman. The description of the policeman is "[The policeman] hat braune lange Haare, einen Vollbart und eine dicke Zigarre im Mund. Seinen Ausweis hat er an seine Jacke geheftet.".

Rule for writing a paragraph about the policeman:

if the player is asked:
 now the policeman is mentioned;
 otherwise:
 if policeman is in rage:
 say "[Der policeman] blickt dich hasserfüllt an.";
 otherwise:
 say "[one of][Der policeman] schaut stur geradeaus.
 [or][Der policeman] wirkt irgendwie ungeduldig.
 [or][Der policeman] spielt nervös mit seinen
 Fingern.[or][Der policeman] wirft dir einen
 skeptischen Blick zu.[or]Erwartungsvoll blickt
 dich [der policeman] an.[cycling]".

The jacket is a female thing, worn by the policeman. The printed name of the jacket is "Jacke". Understand "Jacke" or "loecher [p]" as the jacket. the description of the jacket is "Eine schmutzige Jacke, die ziemlich abgetragen aussieht, bereits einige Löcher hat und obendrein viel zu klein für [den policeman] ist.".

The badge is a male part of the jacket. Understand "ausweis" or "polizeimarke" or "polizeiausweis" or "marke" as the badge. The printed name is "Ausweis". The description is "Auf dem Ausweis ist deutlich der Name 'Albert Lee' zu lesen."

Wie schon gesagt, die Jacke ist ihm irgendwie zu klein, seltsam, seltsam ... Schauen wir uns den Wagen mal näher an:

Section - Police car

The police car is a neuter room, inside from the street. "Eines dieser typischen Polizeiautos. Die Ausstattung sieht ziemlich schäbig aus, die Sitze, die Rücksitze, das Handschuhfach, das Lenkrad, der Rückspiegel - nicht gerade der neueste Wagen und es riecht hier stark nach Rauch."

The printed name of the police car is "Im Polizeiauto".

Und der Typ lässt dich nicht wieder aussteigen ...

Instead of going outside when location is police car:
 if policeman is in rage:
 say "[Der policeman] packt mich am Kragen und hält mich zurück.";
 otherwise:
 say "'Moment mal' sagt der Polizeibeamte. 'Wo wollen Sie denn hin in diesem Zustand? Erzählen Sie mir erst mehr, damit ich ihnen helfen kann.'".

Es gibt so gewisse Filme, da schlägt der Zuschauer die Hände über dem Kopf zusammen, dass der Protagonist so blöd ist, in den Wagen zu steigen, ohne auf den Rücksitz zu schauen ...

The mirror is a male scenery, here. Understand "Rueckspiegel" or "Spiegel" as the mirror. The printed name is "Rückspiegel". The description is "Im Rückspiegel siehst du eine Leiche auf dem Rücksitz liegen!".

Instead of searching the mirror:
 try examining the mirror.

The backseats are a plural-named blatant scenery, here. Understand "Ruecksitze" or "Ruecksitz [m]" as the backseats. The printed name is "Rücksitze". The description is "Eine Leiche liegt auf dem Rücksitz!"

Instead of searching the backseats:
try examining the backseats.

Instead of looking under the backseats:
try examining the backseats.

Tja, und was muss der Spieler erkennen?

The corpse is a female blatant scenery, here. Understand "tot" and "Koerper [m]" or "Leiche" or "Toten" as the corpse. The printed name is "Leiche". The description is "Der Körper des Toten ist deutlich kleiner als der Polizist. Aus seiner Brust ragt ein Messer."

After examining a blatant thing:
now the player is not naive.

Zumindest, dass er nicht mehr unwissend ist, nachdem er ein "blatant thing" untersucht hat.

The knife is a neuter faraway blatant scenery, here. Understand "Messer" or "Griff [m]" as the knife. The printed name is "Messer[-s]". The description is "Auf dem Griff sind Initialen eingeritzt: 'J.R.'"

Moment, die Initialen, da war doch ein gewisser JOHN REZZNAR ...
Am Lenkrad gibt es einen kleinen Hinweis, denn John ist stärker als der Spieler ...

The steering wheel is a neuter scenery, here. Understand "Lenkrad" or "Rad" as the steering wheel. The printed name is "Lenkrad[-s]". The description is "Das Lenkrad ist alt und abgegriffen, wie auch der Rest des Wagens. Seitlich ist ein unscheinbarer Aufkleber angebracht."

The tiny note is a male part of the steering wheel. Understand "unscheinbar" and "Aufkleber" as the tiny note. The printed name is "Aufkleber". The description is "Auf dem Aufkleber steht: 'Sicherheitshinweis: Bitte jegliche Erschütterung des Lenkrads vermeiden, sonst Gefahr der Auslösung des Airbags.'"

The airbag is a male thing. Understand "Bag" or "Airbag" as the airbag. The printed name is "Airbag[-s]". The description is "Der Airbag hängt schlaff am Lenkrad."

Instead of attacking or pushing or pulling the steering wheel:
if airbag is in police car:
say "Weitere Attacken sind nutzlos." instead;
say "Mit aller Macht hämmerst du auf die Lenksäule ein. Plötzlich
öffnet sich der Airbag mit einem lauten Zischen und drückt
[den policeman] halb aus dem Wagen.";
move airbag to police car;
now policeman is half-out.

Jetzt, nachdem der Airbag ausgelöst wurde, ist der vermeintliche Polizist schon halb draußen ...
Der Rest wird im Abschnitt "Scene investigating" abgehandelt.

Section - Scene investigating

Investigating is a scene.

Investigating begins when the player is in the police car.

Investigating ends when the the player is in the police car for the 15th turn.

When investigating begins:

say "Schnell steigst du ein. 'Ich bin aus dieser Klinik geflohen, die einst eine Kathedrale war', bringst du atemlos hervor.
'Beruhigen sie sich erst einmal' sagt der Polizeibeamte und wendet langsam den Kopf. 'Haben Sie sich verletzt?';
now the player is asked.

When investigating ends:

say "Bevor du noch irgendetwas tun kannst, tauchen seltsame Gestalten aus der Finsternis auf. Sie haben den Polizeiwagen umstellt, es gibt kein Entkommen mehr. [Der policeman] hat dich schon mit seinen eiskalten Fingern gepackt und zerrt dich nach draußen."
end the story finally saying "Du bist verloren."

The policeman can be in rage.
The policeman can be half-out.

Instead of attacking or pushing or pulling the policeman when the player is naive:

say "Dazu siehst du doch keinerlei Veranlassung."

Instead of pushing or pulling or attacking the policeman when policeman is not half-out and player is not naive:

now the policeman is in rage;
say "[One of]In diesem Moment erfasst dich eine innere Panik und du greifst [den policeman] verzweifelt an. Es entsteht ein wildes Handgemenge.[or]Du schlägst wild auf [den policeman] ein, es scheint ihm aber nicht das geringste auszumachen.[or][Der policeman] drückt dich mit eisernem Griff in den Sitz. Du kannst dich kaum mehr bewegen.[stopping]"

So, wir haben es tatsächlich geschafft. Das optimale Ende:

Instead of pushing or pulling or attacking the policeman when policeman is half-out:

say "Mit einer ruckartigen Bewegung stößt du [den policeman] aus dem Wagen, schließt die Tür und verriegelst sie von innen. Er klettert auf die Motorhaube und hämmert mit Faustschlägen wild auf die Windschutzscheibe ein. Du startest den Wagen und fährst mit quietschenden Reifen los. [Der Policeman] springt ab und starrt dem Wagen mit hasserfüllten Augen nach, als er schnell in der Dunkelheit verschwindet."
end the story finally saying "Du bist entkommen."

Somit sind wir auch am Ende des Kapitels "Gameplay" angelangt. Wie schon vorher erwähnt, sind aufgrund des großen Umfangs und der vielen Wiederholungen nicht alle Abschnitte hier abgehandelt worden, einige Räume und nur für die Atmosphäre und Vollständigkeit notwendige Gegenstände sind nicht extra erwähnt worden.


Hier könnt ihr entweder selbst ein wenig experimentieren, oder einfach den gesamten Quelltext nochmals in das Projekt kopieren, ganz wie ihr wollt.

Wir machen weiter mit dem Abschnitt "Testen und Ent-buggen". [->]

4. Das Testen und Ent-buggen

Inform 7 zeigt auch hier seine Stärken: Das Testen des eigenen Quellcodes passiert quasi "zeitgleich" neben dem Schreiben her. Hier war zu Inform 6 Zeiten ein aufwändiger Prozess notwendig. Man musste zum Compiler wechseln, das Quellfile compilieren lassen, evtl. Fehler in der Notation korrigieren (mühsam unter Angabe der Quelltextzeile, die dann erst mal gesucht werden musste), dann wieder speichern, erneut versuchen zu compilieren und wenn es dann erfolgreich war, mühsam wieder zu der Stelle spielen, an der man aufgehört hatte – bei längeren Spielen eine Qual.

Inform 7 macht dies alles mit einem Klick – ist ein Fehler drin, wird gleich die betreffende Zeile markiert:



4.1Replay und Skein

Inform 7 legt für jeden Spielversuch einen eigenen Zweig in der "Skein" an – das ist eine Tabelle mit Befehlen, die zu dem aktuellen Spielstand im Spielfenster führen.

Dabei ergibt eine Änderung innerhalb eines Skein-Knotens eine neue Abzweigung. Inform 7 merkt sich, welche Skein im Moment durchgespielt wurde und wird durch Klick auf "replay" auch wieder dorthin spielen – ganz automatisch.

Das "transcript" kann auch "blessed", also abgesegnet werden. Dann merkt sich Inform den aktuellen "Zustand" des Texts und vergleicht ihn auf Abweichungen (im Ausgabertext) zum aktuellen Spiel. Änderungen werden dabei farblich gekennzeichnet.

Section - Going to near locations

Going near is an action applying to one thing.

Understand "geh zu [directive]" or "geh nach [directive]" or "geh in [directive]" or "geh [directive]" or "[directive]" as going near.

Source

Du öffnest leise die Tür.

Vorsichtig schaust du um die Ecke - niemand zu sehen.

>

s

Gerade als du den Raum verlassen willst, steht urplötzlich eine Krankenschwester in der Tür. "Aber, aber, mein Bester. Wer wird denn schon aufstehen?", sagt sie mit einem kühlen Unterton und betritt das Zimmer. "Soll ich Ihnen etwas zu trinken bringen?"

Alles um dich dreht sich und du kannst dich kaum aufrecht halten.

>

leg dich hin

Du legst dich auf das alte Bett.

"Ich bin übrigens Mariel", stellt sie sich vor.

"So ist es brav, sie brauchen jetzt viel Ruhe", sagt Mariel, "ich komme später wieder zu Ihnen." Die Tür wird abrupt zugeschlagen.

>

steh

Du schielst von dem alten Bett

Blessed Transcript

Du öffnest leise die Tür.

Vorsichtig schaust du um die Ecke - niemand zu sehen.

>

s

Gerade als du den Raum verlassen willst, steht urplötzlich eine Krankenschwester in der Tür. "Aber, aber, mein Bester. Wer wird denn schon aufstehen?", sagt sie mit einem kühlen Unterton und betritt das Zimmer. "Soll ich Ihnen etwas zu trinken bringen?"

Alles um dich dreht sich und du kannst dich kaum aufrecht halten.

>

leg dich hin

Du legst dich auf das alte Bett.

"Ich bin übrigens Mariel", stellt sie sich vor.

"So ist es brav, sie brauchen jetzt viel Ruhe", sagt Mariel, "ich komme später wieder zu Ihnen." Die Tür wird abrupt zugeschlagen.

>

steh

Du schielst von dem alten Bett

Die grünen Bereiche rechts sind das "blessed transcript". Die Bereiche werden in der Skein gelb angelegt:

"Mariel" by Michael Baltes

Include Locksmith by Emily Short.
Include German by Team Gerx.
Include German Exit Lister by Team Gerx.
Include German Default Messages by Team Gerx.

Part - The prelude

Section - Release Options

Release along with cover art.
Release along with the source text.

Release along with a "german" website.
Release along with the "quixe" interpreter.

Section - General settings

Use no scoring.
Use explicit error messages.
Use DICT_WORD_SIZE of 15.

The story genre is "Survival Horror".
The story headline is "Ein interaktiver Horror".
The story description is "Ein Krankenhaus, in dem nichts so ist, wie es sein sollte. Nach einem schweren Unfall deckst du allmählich die Geschichte dieses Ortes auf, bis schließlich die Wahrheit ans Licht kommt."

The release number is 4.


When play begins:
say "Ein leiser immer wiederkehrender Piepton ist zu hören - sonst nichts. Du schlägst langsam die Augen auf und schüttelst die Benommenheit so gut es geht ab. Was ist bloß passiert? Wie kommst du hierher?[paragraph break]Das Einzige, woran du dich momentan erinnern kannst, ist ein blendendes Licht, ein Aufprall. Du hattest keine Chance mehr zu reagieren. Die Kurve war unübersichtlich und das entgegenkommende Auto hatte gerade an dieser Stelle zum

Flowchart

```

graph TD
    start([start]) --> steh1([steh])
    start -.-> schau([schau])
    steh1 --> oeffne_tuer1([öffne tür])
    oeffne_tuer1 --> s1([s])
    s1 --> leg_dich_hin([leg dich hin])
    leg_dich_hin --> steh2([steh])
    steh2 --> oeffne_tuer2([öffne tür])
    oeffne_tuer2 --> s2([s])
    schau -.-> steh3([steh])
    steh3 -.-> oeffne_tuer3([öffne tür])
    oeffne_tuer3 -.-> s3([s])
    s3 -.-> red_mit_mariel([red mit mariel])
    red_mit_mariel -.-> sag_ja([sag ja])
    sag_ja -.-> leg_dich_hin2([leg dich hin])
  
```

Wenn etwas im Text geändert wird (hier "ständig" statt "immer" im Einleitungstext), wird das "transcript" im nächsten Replay rot markiert und Inform 7 wartet erneut auf ein "Bless":



Außerdem kann eine Verzweigung des Skein, die zum erfolgreichen Ende führt, als "Musterlösung" gekennzeichnet werden. Das passiert mit der Eingabe von "***" beim letzten Befehl oberhalb des Knotens in der Skein. (Rechtsklick auf Knoten – Kontextmenu – "Add Label")

Hinweis: dies ist sehr zu empfehlen, da nach kleinen Änderungen immer getestet werden kann, ob das Spiel noch "gewinnbar" ist, oder ob sich doch irgendwo ein kleiner Fehler eingeschlichen hat.

Für das Testen auf Fehler gibt es außerdem die

4.2 Debug-Befehle

Inform 7 kennt folgende Debug – Befehle (diese Befehle sind möglich während der Test- und Schreibphase, nicht aber nach dem Release)

Definiert durch Inform 7:

Abstract (Gegenstand) to (etwas):

Bewegt irgendeinen beliebigen Gegenstand in etwas hinein, egal ob etwas offen oder geschlossen ist.

Actions (on/off):

Gibt nach jeder Eingabe die verstandene, bzw. geparste Aktion dazu im Klammern aus.

Gklist: (Glukx only)

Gibt eine Liste der verwendeten Windows, Streams und Soundchannels aus.

Gonear (Gegenstand):

Bewegt den Spieler in denselben Raum wie der angegebene Gegenstand

Nouns (Fürwort):

Gibt eine Liste der gesetzten Pronomen aus und zeigt an, auf welche Gegenstände sich diese momentan beziehen

Purloin:

"Klaut" irgendeinen beliebigen Gegenstand aus dem Spiel und legt ihn ins Inventar

Random:

macht den Zufall vorhersehbar. Diese Option kann auch unter dem "settings" panel – Randomness aktiviert werden. Hilft beim Testen, da durch Zufall generierte Dinge in diesem Durchlauf immer gleich ablaufen.

Relations:

Zeigt die gesetzten Relations an.

Rules (on/off):

Schaltet die Anzeige der abgearbeiteten Rules an. Das kann u.U. eine ganz schön lange Liste pro Zug werden.

Scenes (on/off):

Schaltet die Anzeige der gerade ablaufenden Szenen an und aus.

Scope:

Zeigt alle Dinge an, die sich in Reichweite des Spielers befinden.

Showheap:

Zeigt die Speicherbelegung an.

Showme:

Zeigt alle Dinge an, die sich in Reichweite des Spielers befinden, mit Angabe des Typs.

Showverb (Verb):

Zeigt zu dem Verb alle möglichen Synonyme und Syntaxregeln an.

Test:

Führt ein Testskript aus (muss im Quelltext vorher definiert sein)

Trace (on/off):

Verfolgt die Aktion des Spielers in die Tiefen der Parserfunktionen hinein, die zusätzliche Angabe einer Ziffer (1 bis einschl. 6) legt den Tiefe des Vorgangs fest.

Tree:

Zeigt alle Objekte in Baumstruktur auf.

Definiert durch GerX:**Libcheck:**

Führt einen Wörterbuchcheck durch – sind alle Einträge korrekt, sind welche mit Umlauten dabei, sind Endungen auf -e bei Verben vorhanden, etc. Wird normalerweise im Hintergrund bei Spielstart durchgeführt und meldet sich dann bereits zu Wort, falls was nicht passt.

Deklinier (Gegenstand):

Dekliniert einen Gegenstand durch und deckt evtl. Fehler auf.

Das war es auch schon. Wir machen weiter mit der ...

4.3 Die Musterlösung

The screenshot shows the MarielR5 Inform 7 editor. The left pane displays the 'Release Options' and 'General settings' sections. The right pane shows a flowchart with two columns of nodes connected by vertical lines. The left column nodes are: 'öffne tür', 's', 'leg dich hin', 'steh', 'öffne tür', 's', 'e', '***', 'lies liste'. The right column nodes are: 'steh', 'öffne tür', 's', 'red mit mariel', 'sag ja', 'leg dich hin', 'schlafe', 'spuck'.

Wenn wir in den Release Optionen ergänzen: Release along with a solution wird dieser Teil der Skein als reine Textdatei mit ausgegeben.


5. Das Veröffentlichen

So, wir sind jetzt (fast) am Ende unserer Arbeit angelangt. Wir kompilieren das Ganze nochmal, indem wir auf "Release" klicken:


The screenshot shows the MarielR5 Inform 7 editor with the 'Packaging up for Release - Succeeded' dialog box open. The dialog box displays a folder structure for 'MarielR5 Materials' containing 'Release' and 'Mariel.gblorb'. It lists the contents of the release: a blorb file (1760K), a website (75 pages), a play-in-browser page (73 web pages), and source text (1047K). It also lists additional files: 'Cover.png' (1047K) and 'Small Cover.png' (28K). The dialog box provides instructions on how to use the release materials and offers suggestions for additional content.

5.1. Eine Webseite

Nachdem wir die entsprechenden Einstellungen schon vorher im ersten Kapitel vorgenommen haben, ist die Webseite bereits fix und fertig im Release Ordner zu finden:
Die Index Seite:



Und die "Play"-Seite:



Quixe for Inform 7 (v. 1.1.1)

5.2. IFDB, Ifwiki und Ifwizz

Die Einträge in der IFDB und im Ifwiki müssen per Hand vorgenommen werden. Meldet Euch an und Ihr könnt die entsprechenden Seiten "online" editieren. Am Schluß könnte es folgendermaßen aussehen:

The screenshot shows the Ifwiki page for 'Mariel'. The page layout includes a sidebar on the left with navigation links (Main Page, Community portal, Current events, Recent changes, Random page, Help) and categories (Basics, Comics, People, Competitions, Authoring Systems, Interpreters, The Glossary, Tutorials). The main content area has a title 'Mariel' with a small image of a skull and the text 'XYZZY Awards 2010 Finalist - Best Individual NPC'. Below this is a 'Contents' table of contents, a 'How It Begins' section (which is empty), a 'Trivia and Comments' section (with one comment), a 'Versions' section (with one release), and a 'Links' section. On the right side, there is a detailed information table for 'Mariel'.

Mariel	
File:Mariel small cover .jpg	
Author(s)	Michael Baltes
Publisher(s)	n/a
Release date(s)	21-Aug-2010
Authoring system	Inform 7
Platform(s)	Z-code
Language(s)	German
License(s)	Freeware
Multimedia	
Color	none
Graphics	none
Sound/Music	none
Ratings	
Cruelty scale	Cruel

The screenshot shows the IFDB page for 'Mariel'. The page layout includes a header with the IFDB logo and navigation links. The main content area has a title 'Mariel' by Michael Baltes, a small image of the game cover, and a description. Below this is a 'Game Details' section with various attributes (Language, Current Version, License, Development System, Forgiveness Rating, IFID, TUID). On the right side, there is a 'Download' section with a 'Play On-line' button, a 'Download Page' section, a 'Have you played this game?' section, a 'Playlists and Wishlists' section, and an 'RSS Feeds' section.

Mariel
by Michael Baltes [profile](#)

Horror
2010

[Web Site](#)

☆☆☆☆ (based on 2 ratings)
[1 member review](#)

About the Story
Nothing in this hospital is as usual. After a serious accident you have to figure out what happened and how to escape. Written to showcase the German library extension for Inform 7.

Game Details


Language: German (de)
Current Version: 4
License: Freeware
Development System: Inform 7
Forgiveness Rating: Cruel
IFID: 91603C65-B20D-4D64-B7CA-721A95C620E3
TUID: dvkh0cmqxa1ee14

Awards

Nominee, Best Individual NPC - [2010 XYZZY Awards](#)

Gratulation!

Wir haben den Einstieg in ein modernes "Interactive Fiction Authoring Tool" erfolgreich gemeistert und können das Ergebnis z.B. in "Wingit" ansehen:



Nachwort:

Dieses kleine Handbuch kann natürlich nur einen beschränkten Einblick in das komplexe System Inform 7 geben. Die hier aufgezeigten Texte haben bestenfalls Beispielcharakter und sind keinesfalls als zwingende oder ausschließliche Lösung zu Problemen der Implementierung zu sehen. Das Ergebnis lässt sich auf vielfältige Weise erreichen.

Im Lauf der Zeit wird wohl jeder Autor auch seinen eigenen "Schreibstil" in Inform 7 entwickeln, also eine ganz eigene Art und Logik, je nachdem worauf er besonders Wert legt oder aus welcher Sicht er die Dinge betrachtet.

Möglicherweise wird der Eine oder Andere einen einfacheren oder effizienteren Weg finden, um zu dem gewünschten Ergebnis zu gelangen. Möglicherweise werde ich selbst in ein, zwei Jahren die zwingende Notwendigkeit sehen, dieses Handbuch zu überarbeiten, weil die Methoden vollkommen umständlich oder schlicht veraltet sind.

Als kleiner Einblick in das Schreiben mit Inform 7 und GerX und als Einführung in die technischen und erzählerischen Möglichkeiten moderner IF möchte dieses Handbuch stattdessen dazu anregen, kreative Ideen in Textform umzusetzen.

Viel Spaß beim Experimentieren!
Michael Baltes, im Juni 2011